

P.BEAUFILS/M.LAMARCHE/Y.MUGGIANU

# PROGRAMMES DE PHYSIQUE SUR AMSTRAD

**E**  
EYROLLES









**PROGRAMMES  
DE PHYSIQUE  
SUR  
AMSTRAD**

## CHEZ LE MEME EDITEUR

### Des mêmes auteurs :

- *Programmes de mathématiques sur AMSTRAD.*

### Autres ouvrages :

- C. DELANNOY — *Je débute en BASIC AMSTRAD.*  
— *Faites vos jeux avec AMSTRAD.*
- P. BIHAN — *Programmation sur AMSTRAD PCW 8256 et 8512. BASIC et fichiers.*
- J. MILSANT — *Lexique d'informatique et de micro-informatique avec index alphabétique anglais-français.*
- C. DELANNOY — *Initiation à la programmation.*
- O. LEPAPE — *L'assembleur facile du Z 80.*
- J.-P. DELAHAYE — *Dessins géométriques et artistiques avec votre micro-ordinateur.*  
— *Nouveaux dessins géométriques et artistiques avec votre micro-ordinateur.*
- Ph. DAX — *CP/M et sa famille. Guide d'utilisation.*
- P. STUMM — *Jeux de robotique en BASIC.*
- M.G. MONTEIL, R. SCHOMBERG — *Programmes d'intelligence artificielle en BASIC.*
- M. JAMES — *Introduction à l'intelligence artificielle sur micro-ordinateur.*
- M. ROUSSELET — *Calcul numérique sur AMSTRAD.*

# PROGRAMMES DE PHYSIQUE SUR AMSTRAD

par

Pierre BEAUFILS, Michel LAMARCHE  
et Yves MUGGIANU

  
EYROLLES

61, boulevard Saint-Germain — 75005 PARIS  
1986

Si vous désirez être tenu au courant de nos publications, il vous suffit d'adresser votre carte de visite au :

Service « Presse », Editions EYROLLES,  
61, Boulevard Saint-Germain  
75240 PARIS CEDEX 05,

en précisant les domaines qui vous intéressent.  
Vous recevrez régulièrement un avis de parution des nouveautés en vente chez votre libraire habituel.

« La loi du 11 mars 1957 n'autorisant, aux termes des alinéas 2 et 3 de l'article 41, d'une part, que les « copies ou reproductions strictement réservées à l'usage privé du copiste et non destinées à une utilisation collective » et, d'autre part, que les analyses et les courtes citations dans un but d'exemple et d'illustration, « toute représentation ou reproduction intégrale, ou partielle, faite sans le consentement de l'auteur ou de ses ayants droit ou ayants cause, est illicite » (alinéa 1<sup>er</sup> de l'article 40). »

« Cette représentation ou reproduction, par quelque procédé que ce soit, constituerait une contrefaçon sanctionnée par les articles 425 et suivants du Code pénal. ».

© ÉDITIONS EYROLLES 1986.

# Préface

*Les lois de la physique sont des êtres bien étranges. Nous les « cotoyons » à chaque instant et pourtant elles nous semblent parfois aussi inaccessibles que les galaxies des confins de l'univers.*

*Partez à la découverte de ces étoiles aux commandes de votre micro-ordinateur. Le temps de charger le programme et vous êtes à pied d'œuvre. Les concepteurs de votre vaisseau et de son logiciel ont tout fait pour que vous franchissiez sans problèmes la barrière épineuse de mathématiques qui a anéanti les tentatives de tant de vos prédécesseurs.*

*Une fois cet obstacle passé, l'immense domaine des lois s'étend devant vous. Approchez de l'une d'elles : l'écran de votre moniteur vous en donne des images. C'est à travers de ces images que vous allez avoir la révélation de la loi : de même que vous imaginez la carrosserie d'une voiture à partir d'une série de photos prises sous des angles différents, de même vous allez vous familiariser avec la loi à travers les applications numériques qu'elle gouverne.*

*Nous avons tout fait pour que votre mission réussisse mais nous vous laissons seul à la fin du voyage. Nous espérons que vous reviendrez ébloui de cette rencontre avec la loi. Désormais, vous saurez mettre un visage derrière son nom, vous saurez la reconnaître sous son masque de frottements et d'imprécision des appareils de mesure lorsque vous la croiserez dans la vie de tous les jours.*

*Après quelques unes de ces aventures, n'hésitez pas à venir nous rejoindre et concevoir des programmes de voyages vers d'autres lois.*





# Sommaire

## MECANIQUE

— Centre de gravité .....	1
— Loi de Newton .....	7
— Mouvement d'un corps dans le champ de pesanteur	19
— Effet Magnus .....	21
— Interactions gravitationnelles — Application à l'étude du mouvement des satellites terrestres .....	28
— Conservation de l'énergie .....	35
— Plan incliné .....	41
— Dynamique .....	48
— Composition de mouvements .....	53
— Le pendule simple .....	60
— Le pendule de Foucault .....	65
— Oscillateurs couplés .....	72

## ELECTRICITE

— Champ et potentiel créés par un système de charges ponctuelles .....	77
— Circuit électrique .....	86
— Circuits électriques : droites de charge .....	94
— Résonance .....	102
— Filtrage RC simple ou double alternance .....	107

## OPTIQUE

— Lentilles minces .....	115
— Lentille .....	125
— Dioptre plan .....	133

## CHIMIE

— Propriétés d'une solution chimique . . . . .	137
— Dosages volumétriques . . . . .	143
— Dissociation d'une solution chimique . . . . .	151
— Couches électroniques . . . . .	153

## PROGRAMMES GÉNÉRAUX

— Equation de Van der Maals . . . . .	161
— Méthode des moindres carrés — Application aux équations non linéaires . . . . .	165

# 1

# Mécanique

## **1. Centre de gravité**

On charge une caisse sur le plateau d'un camion. Suivant la charge et sa position, la répartition du poids sur les essieux est différente. Le camion risque même de basculer si le centre de gravité de l'ensemble se situe en arrière de l'essieu arrière.

Ce programme vous montre de quelle façon les réactions du sol sur les essieux équilibrent le poids du camion et il vous familiarise avec la notion de centre de gravité.

### **Pourquoi ce programme ?**

La notion de centre de gravité est importante car elle permet de simplifier considérablement la résolution des problèmes. Chaque tôle, chaque écrou du camion est soumis à la force d'attraction de la terre : son poids. Pour étudier le comportement du camion, il faudrait tenir compte de ces millions de petits poids ! Non, car lorsque vous étudiez le mouvement global du camion (par exemple : va-t-il basculer ?) vous pouvez remplacer tous ces petits poids par une seule force, le poids du camion, appliquée au centre de gravité du camion.

### **Utilisation du programme**

L'ordinateur vous indique la position du centre de gravité du camion vide (\*), puis il vous demande des renseignements concernant la charge. Il vous indique alors le centre de gravité de la charge

(\*) et celui de l'ensemble camion + charge (X). Enfin il vous indique l'intensité des réactions du sol sur les essieux.

Pour faire avancer le programme, taper →

Les longueurs sont exprimées en mètres, les masses en kilogrammes, les forces en Newton. Pour introduire une donnée, taper le nombre correspondant (sans indiquer l'unité), puis faites ENTER.

Pour sortir du programme, faites ESC ESC.

## Le cœur du programme

### La caisse

On suppose que le volume de la caisse est proportionnel à sa masse et que sa largeur est imposée. La surface du rectangle représentant la caisse sur l'écran est donc proportionnelle à sa masse  $M''$

$$M'' = h \cdot c / 1,6 \text{ (ligne 230)}$$

On suppose d'autre part que le centre de gravité de la caisse est au milieu de celle-ci.

### Le centre de gravité (fig. 1)

Soit  $G'$  le centre de gravité du camion vide de la masse  $M'$  et soit  $G''$  celui de la caisse de masse  $M''$ . La position du centre de gravité  $G$  de l'ensemble est donnée par :

$$(M' + M'') \cdot \vec{OG} = M' \cdot \vec{OG'} + M'' \cdot \vec{OG''}$$

soit en projetant sur deux axes  $Ox$  et  $Oy$  :

$$(M' + M'') \cdot y = M' \cdot y' + M'' \cdot y''$$

$$(M' + M'') \cdot z = M' \cdot z' + M'' \cdot z''$$

On retrouve ceci en lignes 270 et 280 avec

$$M' = 5000 \quad y' = 232 \quad z' = 136$$

$$y'' = 180 \text{ (cabine)} + d \text{ (espace)} + c/2 \text{ (demi longueur de la caisse)}$$

$$z'' = 146 \text{ (plateau)} + h/2 \text{ (demi hauteur de la caisse)}.$$




Fig. 1. — Position des centres de gravité

### Réactions du sol (fig. 2)

$\vec{R}_1$  est la réaction du sol sur l'ensemble des deux roues, avant,  
 $\vec{R}_2$  sur l'ensemble des deux roues arrière.

Ces forces équilibrent le poids  $\vec{P}$  du camion chargé

donc :  $\vec{R}_1 + \vec{R}_2 + \vec{P} = \vec{O}$




Fig. 2. — Forces exercées sur le camion

projetons sur un axe vertical vers le haut :

$$R1 + R2 - P = 0$$

$$\text{avec } P = (M'' + 5000) \cdot 9,81 \text{ (ligne 320)}$$

donc : la somme des moments des forces par rapport à un point quelconque est nulle. On a choisi ce point sous les roues avant :

$$R2 \cdot (432 - 88) - P \cdot (y - 88) = 0 \text{ (ligne 310)}$$

Ces deux équations permettent de calculer R1 et R2.

Une valeur négative de R1 signifierait que le sol tire l'essieu avant vers le bas. Ceci étant impossible en l'absence de boue gluante, l'essieu avant quitte le sol : le camion bascule.

## Liste des variables

c	longueur de la caisse
d	distance caisse-cabine
m	masse de la caisse
h	hauteur de la caisse
y,z	coordonnées du centre de gravité de l'ensemble
r1,r2	réactions du sol sur les essieux


## Déroulement du programme

10	présentation
90,500	dessin du camion
100	introduction des données
220,500	dessin du camion
240	dessin de la caisse
270	calcul
330,700	si le camion bascule
410	fin

## Illustrations

L'illustration 3a a été obtenue avec les valeurs 15, 2, 20 000 ; l'illustration 3b avec 5, 15, 12, 500.

Dans ce dernier cas, le centre de gravité de l'ensemble est alors en arrière de l'essieu arrière : le camion bascule.



le camion bascule

Fig. 3a →




Fig. 3b

→

```

10 CLS
20 LOCATE 12,8:PRINT"Centre de gravite"
30 LOCATE 15,10:PRINT"d'un camion"
40 LOCATE 1,22:PRINT"pour faire avancer le program
me,appuyer sur "CHR$(243)
50 REM
60 REM
70 q=INKEY(1):IF q<>0 THEN 70
80 CLS
90 GOSUB 500
95 q=INKEY(1):IF q<>0 THEN 95
100 PRINT"masse du camion: 5000"
110 PRINT:PRINT"longueur de la caisse ? (entre 1
et 20)"
120 INPUT c
130 PRINT:PRINT"intervalle entre la caisse et la c
abine ?"
140 PRINT"(entre 0 et"20-c")"
150 INPUT d
160 PRINT:PRINT"masse de la caisse ?"
170 PRINT"(entre 0 et"2500*c")"
180 INPUT m
190 c=20*c:d=20*d

```

```

200 REM
210 CLS
220 GOSUB 500
230 h=1.6*m/c
240 PLOT 180+d,146
250 DRAWR c,0:DRAWR 0,h:DRAWR -c,0:DRAWR 0,-h

255 TAG:MOVE 176+d+c/2,150+h/2:PRINT CHR$(238);:TAGOFF
260 REM
270 y=(232*5000+(180+d+(c/2))*m)/(m+5000)
280 z=(136*5000+(146+(h/2))*m)/(m+5000)
290 TAG:MOVE y-4,z+4:PRINT"x";:TAGOFF
300 REM
310 r2=(m+5000)*9.81*(y-88)/344
320 r1=((m+5000)*9.81)-r2
330 IF r1<0 THEN GOTO 700
340 LOCATE 6,21:PRINT CHR$(240)
350 LOCATE 3,23:PRINT INT(r1)
360 LOCATE 28,21:PRINT CHR$(240)
370 LOCATE 25,23:PRINT INT(r2)
390 REM
400 q=INKEY(1):IF q<>0 THEN 400
410 CLS:GOTO 100
420 REM
500 REM"camion"
510 MOVE 180,144:DRAW 580,144:DRAWR 0,-2
520 DRAW 40,142:DRAWR 0,-24:DRAWR 20,0
530 MOVE 116,120:DRAW 404,120
540 MOVE 70,144:DRAWR 0,42:DRAWR 30,0
550 DRAWR 20,40:DRAWR 60,0:DRAW 180,144
560 MOVE 100,186:DRAWR 60,0:DRAWR 0,35
570 DRAWR -40,0:DRAWR 0,-35
580 FOR a=0 TO 2*PI STEP 0.1
590 PLOT 88+22*COS(a),118+22*SIN(a)
600 PLOT 432+22*COS(a),118+22*SIN(a)
610 PLOT 88+26*COS(a/2),118+26*SIN(a/2)
620 NEXT a
630 TAG:MOVE 228,140:PRINT CHR$(238);:TAGOFF

640 LOCATE 39,25:PRINT CHR$(243)
650 RETURN
700 REM
710 LOCATE 1,1:PRINT"le camion bascule"
720 q=INKEY(1):IF q<>0 THEN 720
730 GOTO 100

```




## 2. Loi de Newton

Cette loi dit que deux corps s'attirent proportionnellement à leur masse et en raison inverse du carré de leur distance. C'est cette loi qui régit le mouvement des corps célestes.

Mathématiquement, elle s'exprime de la manière suivante :

Le corps de masse  $M_2$  subit, de la part de  $M_1$ , une force  $F$  dirigée vers  $M_1$  et égale à

$$\vec{F} = k \frac{M_1 M_2}{r^3} \vec{r}$$



Inversement pour  $M_1$ .


Par ailleurs, le principe fondamental de la Dynamique dit que le produit de la masse  $M$  d'un corps par son accélération  $a$  est égal à la somme (vectorielle) des forces qui lui sont appliquées.

Considérons alors deux corps de masses  $M_1$  et  $M_2$ . Ils se soumettent mutuellement à des forces attractives. (Il pourrait s'agir de deux charges électriques de signes opposés, dont on néglige la masse : les équations sont du même type).

On peut écrire :

$$M_1 \vec{a}_1 = -k \frac{\vec{r}}{r^3} M_2$$

$$M_2 \vec{a}_2 = -k \frac{\vec{r}}{r^3} M_1$$



### Liste des variables

X1	abscisse du mobile $M_1$
Y1	ordonnée du mobile $M_1$
X2	abscisse du mobile $M_2$
Y2	ordonnée du mobile $M_2$
V1	composante sur OX de la vitesse de $M_1$

- W1 composante sur OY de la vitesse de  $M_1$
- V2 composante sur OX de la vitesse de  $M_2$
- W2 composante sur OY de la vitesse de  $M_2$
- A1 composante sur OX de l'accélération de  $M_1$
- B1 composante sur OY de l'accélération de  $M_1$
- A2 composante sur OX de l'accélération de  $M_2$
- B2 composante sur OY de l'accélération de  $M_2$

Ces deux équations vectorielles conduisent aux relations suivantes :

$$\frac{d^2x_1}{dt^2} = -k \frac{(x_1 - x_2)}{r^3} \qquad \frac{d^2y_1}{dt^2} = -k \frac{(y_1 - y_2)}{r^3}$$

$$\frac{d^2x_2}{dt^2} = +k \frac{(x_1 - x_2)}{r^3} \qquad \frac{d^2y_2}{dt^2} = +k \frac{(y_1 - y_2)}{r^3}$$

avec  $r = (x_1 - x_2)^2 + (y_1 - y_2)^2$ .

Elles s'intègrent de façon classique :

Soit DT l'intervalle de calcul choisi ; la variation de vitesse d'un ensemble sur OX par exemple est égale à  $a_x \cdot DT$ . De même, la variation de l'abscisse x pendant le temps DT vaut  $v_x \cdot DT$ . La nouvelle coordonnée permet de déterminer r, donc l'accélération, et ainsi d'effectuer un nouveau pas de calcul.

Nous proposons plusieurs variantes du programme de base, montrant les différentes manières dont on peut exploiter cette loi.

— Afin d'obtenir des courbes intéressantes dans un premier temps, il a fallu choisir des valeurs numériques convenables ; pour cela, si l'on s'en tient aux phénomènes de gravitation universelle, nous posons :

$$K = \frac{Gm}{(\text{unité de } r)^2} \qquad G = 6,67 \times 10^{-11} \text{ MKSA}$$

— Pour les phénomènes de l'infiniment petit, m sera de l'ordre de  $1,67 \times 10^{-27}$  kg (masse du neutron et l'unité de distance sera de l'ordre de l'Angström ( $10^{-10}$  m)).

Ces raisons nous ont amené à prendre  $K = 10$  pour les applications numériques.

Le choix de DT est délicat : il faut prendre cette quantité assez petite, de façon à ce que les accroissements A1.DT, ..., VA.DT, ... sont petits devant V1, ..., X1, ..., mais pas trop pour ne pas rendre le pro-

gramme trop lent. Un raffinement consisterait à asservir la valeur de DT à ces accroissements, c'est-à-dire par exemple à s'imposer des variations de X1, ... constantes ou inférieures à une valeur déterminée.

## Programme n° 1

Il montre le cas de deux mobiles, lancés l'un vers l'autre, de masses égales. On remarquera la courbure des trajectoires à proximité du point de rencontre.

## Programme n° 2

Il permet d'étudier le cas où les forces entre mobiles sont toujours en  $1/R^2$ , mais répulsives (c'est le cas en électricité, pour deux charges de même signe). La courbure des trajectoires est moindre que précédemment, car nous avons pris des vitesses initiales plus grandes. Le passage du programme 1 au programme 2 se fait simplement en changeant les signes des accélérations (lignes 50/60).

## Programme n° 3

Il permet l'étude des chocs entre mobiles. Pour cela, les forces newtoniennes en  $1/R^2$  sont remplacées par des forces en  $1/R^7$ . Cela signifie que ces forces sont négligeables dans tous les cas, sauf quand les deux mobiles sont à proximité immédiate. On retrouve ici les phénomènes observés sur une table à coussin d'air ou sur un plateau de billard.

## Programme n° 4

Il retourne à l'étude des forces newtoniennes en  $1/R^2$ , mais en envisageant cette fois-ci des mobiles de masses différentes.

On peut écrire :

$$\vec{m}_1 \vec{a}_1 = G \frac{m_1 m_2}{r^3} \vec{r} \quad \text{d'où} \quad \vec{a}_1 = \frac{G m_2}{r^3} \vec{r}$$

$$\vec{m}_2 \vec{a}_2 = -G \frac{m_1 m_2}{r^3} \vec{r} \quad \text{d'où} \quad \vec{a}_2 = \frac{G m_1}{-r^3} \vec{r}$$

Le coefficient de proportionnalité de  $a_1$ , soit  $Gm_2$  est proportionnel à  $m_2$  et inversement. En introduisant ainsi un facteur 5 dans les expressions des composantes de  $a_1$ , nous rendons compte d'une masse  $m_2$  cinq fois plus grande que  $m_1$ . Il est alors aisé de voir que la particule (1), plus légère donc, est nettement plus déviée que la particule (2).

## Programme n° 5

Il permet d'introduire l'influence d'un champ de gravitation uniforme sur le phénomène d'interaction entre mobiles. Cela revient à incliner la table à coussin d'air. Nous avons gardé le rapport de masses (5) du programme n° 4.

## Programme n° 6

Il permet d'extraire de l'observation de ces phénomènes l'une des conséquences du principe fondamental de la Dynamique : le centre de gravité du système des deux mobiles se déplace d'un mouvement uniforme et est indifférent aux interactions entre ceux-ci.

## Programme n° 7


Il permet de poursuivre l'étude précédente. Nous visualisons cette fois-ci le mouvement des deux mobiles dans le repère du "laboratoire", comme nous l'avons fait jusqu'ici, mais également dans le repère du centre de gravité. Le mouvement (complexe) des deux mobiles dans le premier repère prend une forme simple dans le second : une nouvelle loi de la Mécanique en découle ; la quantité de mouvement globale du système y est toujours nulle.

```
10 REM *****
20 REM * attraction *
30 REM * universelle1 *
40 REM *
50 REM *****
60 REM
70 MODE 2 : CLS
```

```

80 k=10 : dt =0.5
90 x1=20 : x2=20:y1=100:y2=300
100 v1=2:w1=0.45:v2=2:w2=-0.5
110 r=SQR(((ABS(x1-x2))^2)+((ABS(y1-y2))^2))
120 a1=-k*(x1-x2)/r^3:a2=-a1
130 b1=-k*(y1-y2)/r^3:b2=-b1
140 v1=v1+a1*dt:w1=w1+b1*dt
150 v2=v2+a2*dt:w2=w2+b2*dt
160 x1=x1+v1*dt:y1=y1+w1*dt
170 x2=x2+v2*dt:y2=y2+w2*dt
180 IF x1 < 0 OR x1 > 639 THEN STOP
190 IF x2 < 0 OR x2 > 639 THEN STOP
200 IF y2 < 0 OR y2 > 399 THEN STOP
210 IF y1 < 0 OR y1 > 399 THEN STOP
220 PLOT x1,y1:PLOT x2,y2
230 GOTO 110


```



```

10 REM *****
20 REM *   attraction   *
30 REM * universelle2 *
40 REM *               *
50 REM *****
60 REM
70 MODE 2 : CLS
80 k=10 : dt =0.5
90 x1=20 : x2=20:y1=100:y2=300
100 v1=2:w1=0.45:v2=2:w2=-0.5
110 r=SQR(((ABS(x1-x2))^2)+((ABS(y1-y2))^2))
120 a1=k*(x1-x2)/r^3:a2=-a1
130 b1=k*(y1-y2)/r^3:b2=-b1
140 v1=v1+a1*dt:w1=w1+b1*dt
150 v2=v2+a2*dt:w2=w2+b2*dt
160 x1=x1+v1*dt:y1=y1+w1*dt
170 x2=x2+v2*dt:y2=y2+w2*dt
180 IF x1 < 0 OR x1 > 639 THEN GOTO 240
190 IF x2 < 0 OR x2 > 639 THEN GOTO 240
200 IF y2 < 0 OR y2 > 399 THEN GOTO 240
210 IF y1 < 0 OR y1 > 399 THEN GOTO 240
220 PLOT x1,y1:PLOT x2,y2
230 GOTO 110
240 STOP

```




```

10 REM *****
20 REM *   attraction   *
30 REM * universelle3 *
40 REM *               *
50 REM *****
60 REM
70 MODE 2 : CLS
80 k=10 : dt =0.5
90 x1=20 : x2=20:y1=100:y2=300
100 v1=2:w1=0.45:v2=2:w2=-0.5
110 r=SQR(((ABS(x1-x2))^2)+((ABS(y1-y2))^2))
120 a1=k*(x1-x2)/r^8:a2=-a1
130 b1=k*(y1-y2)/r^8:b2=-b1
140 v1=v1+a1*dt:w1=w1+b1*dt
150 v2=v2+a2*dt:w2=w2+b2*dt
160 x1=x1+v1*dt:y1=y1+w1*dt
170 x2=x2+v2*dt:y2=y2+w2*dt
180 IF x1 < 0 OR x1 > 639 THEN GOTO 240
190 IF x2 < 0 OR x2 > 639 THEN GOTO 240
200 IF y2 < 0 OR y2 > 399 THEN GOTO 240
210 IF y1 < 0 OR y1 > 399 THEN GOTO 240
220 PLOT x1,y1:PLOT x2,y2
230 GOTO 110
240 STOP

```

y2= 243.672874


y1= 140.827135




```

10 REM *****
20 REM * attraction *
30 REM * universelle4 *
40 REM *
50 REM *****
60 REM
70 MODE 2 : CLS
80 k=10 : dt =0.5
90 x1=20 : x2=20:y1=100:y2=300
100 v1=2:w1=0.45:v2=2:w2=-0.48
110 r=SQR(((ABS(x1-x2))^2)+((ABS(y1-y2))^2))
120 a1=-5*k*(x1-x2)/r^3:a2=-a1/5
130 b1=-5*k*(y1-y2)/r^3:b2=-b1/5
140 v1=v1+a1*dt:w1=w1+b1*dt
150 v2=v2+a2*dt:w2=w2+b2*dt
160 x1=x1+v1*dt:y1=y1+w1*dt
170 x2=x2+v2*dt:y2=y2+w2*dt
180 IF x1 < 0 OR x1 > 639 THEN GOTO 240
190 IF x2 < 0 OR x2 > 639 THEN GOTO 240
200 IF y2 < 0 OR y2 > 399 THEN GOTO 240
210 IF y1 < 0 OR y1 > 399 THEN GOTO 240
220 PLOT x1,y1:PLOT x2,y2
230 GOTO 110
240 STOP

```







Programme Attraction universelle 4  
(paramètre : 5 des lignes 120 et 130 remplacé par 50)

```


10 REM *****
20 REM * attraction *
30 REM * universelle5 *
40 REM *           *
50 REM *****
60 REM
70 MODE 2 : CLS
80 k=10 : dt =0.5
90 x1=20 : x2=20:y1=200:y2=399
100 v1=2:w1=0.45:v2=2:w2=-0.48
110 r=SQR(((ABS(x1-x2))^2)+((ABS(y1-y2))^2))
120 a1=-5*k*(x1-x2)/r^3:a2=-a1/5
130 b1=-5*k*(y1-y2)/r^3:b2=-b1/5
140 v1=v1+a1*dt:w1=w1+(b1-0.02)*dt
150 v2=v2+a2*dt:w2=w2+(b2-0.02)*dt
160 x1=x1+v1*dt:y1=y1+w1*dt
170 x2=x2+v2*dt:y2=y2+w2*dt
180 IF x1 < 0 OR x1 > 639 THEN GOTO 240
190 IF x2 < 0 OR x2 > 639 THEN GOTO 240
200 IF y2 < 0 OR y2 > 399 THEN GOTO 240

```

```

210 IF y1 < 0 OR y1 > 399 THEN GOTO 240
220 PLOT x1,y1:PLOT x2,y2
230 GOTO 110
240 STOP

```



```


10 REM *****
20 REM * attraction *
30 REM * universelle6 *
40 REM * *
50 REM *****
60 REM
70 MODE 2 : CLS
80 k=10 : dt =0.5
90 x1=20 : x2=20:y1=100:y2=300
100 v1=2:w1=0.45:v2=2:w2=-0.48
110 r=SQR(((ABS(x1-x2))^2)+((ABS(y1-y2))^2))
120 a1=-k*(x1-x2)/r^3:a2=-a1
130 b1=-k*(y1-y2)/r^3:b2=-b1
140 v1=v1+a1*dt:w1=w1+b1*dt
150 v2=v2+a2*dt:w2=w2+b2*dt
160 x1=x1+v1*dt:y1=y1+w1*dt
170 x2=x2+v2*dt:y2=y2+w2*dt
180 IF x1 < 0 OR x1 > 639 THEN GOTO 240
190 IF x2 < 0 OR x2 > 639 THEN GOTO 240

```

```

200 IF y2 < 0 OR y2 > 399 THEN GOTO 240
210 IF y1 < 0 OR y1 > 399 THEN GOTO 240
220 PLOT x1,y1:PLOT x2,y2
225 PLOT (x1+x2)/2,(y1+y2)/2
230 GOTO 110
240 STOP
250 STOP

```



```


10 REM *****
20 REM * attraction *
30 REM * universelle7 *
40 REM * *
50 REM *****
60 REM
70 MODE 2 : CLS
80 LOCATE 40,4:PRINT "centre de gravite"
90 LOCATE 20,20 :PRINT "laboratoire"
100 k=10 : dt =0.5
110 x1=100 : x2=0:y1=0:y2=20
120 v1=-0.4:w1=1:v2=0.6:w2=0.7
130 r=SQR(((ABS(x1-x2))^2)+((ABS(y1-y2))^2))

```

```

140 a1=-k*(x1-x2)/r^3:a2=-a1
150 b1=-k*(y1-y2)/r^3:b2=-b1
160 v1=v1+a1*dt:w1=w1+b1*dt
170 v2=v2+a2*dt:w2=w2+b2*dt
180 x1=x1+v1*dt:y1=y1+w1*dt
190 x2=x2+v2*dt:y2=y2+w2*dt
200 IF x1 < 0 OR x1 > 639 THEN GOTO 280
210 IF x2 < 0 OR x2 > 639 THEN GOTO 280
220 IF y2 < 0 OR y2 > 399 THEN GOTO 280
230 IF y1 < 0 OR y1 > 399 THEN GOTO 280
240 PLOT 200+x1-(x1+x2)/2,320+y1-(y1+y2)/2
250 PLOT 200+x2-(x1+x2)/2,320+y2-(y1+y2)/2
260 PLOT x1,y1:PLOT x2,y2
270 GOTO 130
280 STOP
290 STOP

```



### **3. Mouvement d'un corps dans le champ de pesanteur**

Ce problème est classique en Terminale. L'exemple cité est souvent celui de la bombe lâchée par un avion. La décomposition de la force (unique : c'est le poids de l'objet) sur deux axes orthogonaux OX et OY conduit, après intégration des équations différentielles linéaires, d'une part à un mouvement rectiligne à vitesse constante sur OX, d'autre part à un mouvement rectiligne uniformément varié sur OY.

Le programme propose de montrer, dans l'ordre, le mouvement de l'objet suivant OX, puis celui suivant OY, enfin la composition de ces deux mouvements qui donne l'aspect de la trajectoire pour un observateur terrestre. Dans le référentiel lié à l'avion, c'est la seconde observation qu'il faut prendre en compte. Dans un dernier temps, nous tentons de montrer l'influence de la résistance de l'air par l'intermédiaire d'une force de frottement proportionnelle au carré de la vitesse de l'objet qui tombe.

Pour rendre plus générale l'utilisation de ce programme, l'utilisateur doit introduire la vitesse initiale  $V$  (en mètres par seconde), l'angle de départ  $T$  (en degrés), les coordonnées de départ ( $X$ ,  $Y$ ) et éventuellement le coefficient de la force de frottement  $K$  (qui peut être compris entre  $-0,001$ , force très faible, et  $-0,1$ , force très élevée),

On retrouve facilement les cas classiques :

— obus lancé d'un avion

$$\begin{array}{ll} V = \text{quelconque} & ; X = 0 \\ T = 0 \text{ à } -90^\circ & ; Y = 300 \end{array}$$

lignes 80 à 110

— obus lancé par un canon


$$\begin{array}{ll} V = \text{quelconque} & ; X = 0 \\ T = \text{compris entre } 0 \text{ et } +90^\circ & ; Y = 150 \end{array}$$

```


10 REM *****
20 REM *
30 REM * chute libre *
40 REM *
50 REM *****
60 REM
70 MODE 2 : g=9.81
80 INPUT"vitesse initiale:";v:vo=v
90 INPUT"angle de depart:";t:T0=t
100 INPUT"abscisse de depart:";x:xo=x
110 INPUT"ordonnee de depart:";y:yo=y
120 PLOT x,y:DRAWR 639-x,0
130 PLOT x,y:DRAWR 0,399-y
140 FOR n=1 TO 5:READ k
150 v=vo:x=xo:y=yo:t=T0
160 t=t*PI/180:dt=0.01
170 vx=v*COS(t):vy=v*SIN(t)
180 v=(vx^2+vy^2)^0.5
190 fx=k*v*vx:fy=k*v*vy
200 ax=-fx:ay=-fy-g
210 vx=vx+ax*dt:vy=vy+ay*dt
220 x=x+vx*dt:y=y+vy*dt
230 IF x>640 OR y>400 OR y<0 THEN 270
240 PLOT x,y
250 DATA 0,0.001,0.005,0.01,0.1
260 GOTO 180
270 NEXT n

```


vitesse initiale:? 65  
angle de depart:? 45  
abscisse de depart:? 0  
ordonnee de depart:? 150



vitesse initiale: ? 80  
angle de depart: ? 10  
abscisse de depart: ? 0  
ordonnee de depart: ? 200




vitesse initiale: ? 90  
angle de depart: ? 0  
abscisse de depart: ? 0  
ordonnee de depart: ? 300



#### **4. Mouvement d'un corps lancé** **Effet Magnus**

Les programmes suivants permettent d'étudier les trajectoires de corps lancés avec une vitesse initiale oblique dans le vide ou dans

l'air. Dans le cas particulier où le corps en mouvement est sphérique (cas d'une balle de tennis, de golf ou d'un ballon de football), on peut faire intervenir la force Magnus. L'effet Magnus étant dû à la différence de pression qui apparaît entre deux points diamétralement opposés d'une sphère lorsque celle-ci tourne.



Rappelons que la résistance de l'air est de la forme :

$$\vec{F}_a = -K S V^2 \frac{\vec{V}}{\|\vec{V}\|}$$

Pour une sphère  $S = \pi R^2$  (section droite).

Dans le cas d'une balle, la force Magnus est de la forme

$$\vec{F}_M = A \vec{W} \wedge \vec{V}$$

K et A sont des coefficients qui dépendent de la balle et aussi du domaine de la vitesse. Ce sont des grandeurs déterminées expérimentalement.

Dans le cas des forces  $\vec{F}_a$  et  $\vec{F}_M$ , c'est la vitesse relative de l'air et de la balle qui intervient. Dans le cas où la vitesse du vent n'est pas négligeable devant celle de la balle, il faut calculer cette vitesse relative.

Si le vent s'oppose au mouvement de la balle, il faut donc augmenter  $\vec{F}_a$  et  $\vec{F}_M$  à vitesse égale de la balle par rapport à l'observateur ( $\vec{V}_R = \vec{V}_{\text{balle}} - \vec{V}_{\text{vent}}$ ).

Appliquons la relation fondamentale de la Dynamique.

$$M \vec{\gamma} = M \vec{g} - K S V^2 \frac{\vec{V}}{\|\vec{V}\|} + A \vec{W} \wedge \vec{V}$$

Cette équation différentielle sera calculée numériquement dans les programmes.



Pour une balle de golf, par exemple

$$\frac{KS}{M} = 8.10^{-3} \quad \text{et} \quad \frac{2\pi A}{M} = 2.10^{-3}$$

## Liste des variables

V, VO et VF	vitesse de la balle (on fait, dans une boucle, varier la vitesse V de VO à VF)
K	coefficient égal à (KS/M) (Résistance de l'air)
L	coefficient égal à $2\pi \times A/M$ (effet Magnus)
A	valeur de l'angle que fait la vitesse VO avec l'horizontale
W, WO, WF	vitesse de rotation de la balle en tours/seconde
G	accélération de la pesanteur
TM	Temps de l'observation (durée maximale prévue).

## Déroulement du programme 1

Dans le programme 1, on étudie les trajectoires d'une balle de golf liftée, l'angle de la vitesse initiale avec l'horizontale étant égal à 5 degrés. On calcule 600 points pour 5 valeurs de W différentes (ligne 145).

lignes 10 à 125	titre et entrées des données
ligne 140	introduction des constantes pas du calcul (pris égal à 1/100 s)
ligne 200	on fait varier la vitesse de rotation de la balle W de 0 à - 100 tours/seconde (balle liftée) avec une vitesse initiale constante de 80 m/s faisant un angle A avec l'horizontale égale à 5 degrés
lignes 200 à 400	résolution numérique de l'équation différentielle du mouvement (projection sur l'axe OX et projection sur l'axe OY), les paramètres S et R permettent de tracer les courbes à l'échelle prévue par l'utilisateur (ligne 470)
lignes 410 à 560	on trace les courbes sur l'écran et on imprime les différentes données.

## Déroulement du programme 2

Dans le programme 2, on étudie les trajectoires fictives d'une balle de golf liftée (drive) à laquelle on aurait communiqué une vitesse initiale de  $-300$  tours/seconde, la vitesse initiale de celle-ci variant de  $10$  à  $80$  m/s.



L'angle de la vitesse avec l'horizontale est pris égal à  $45^\circ$ .

```
10 REM EFFET MAGNUS
20 REM *****
30 REM
40 REM ENTREE DES DONNEES
50 REM *****
60 CLS
70 INPUT "VALEUR DE V=";V
80 INPUT "VALEUR DE ALPHA A=";A
90 INPUT "VALEUR DE K=";K
100 INPUT "VALEUR DE L=";L
110 INPUT "VALEUR DE WO=";WO
120 INPUT "VALEUR DE WF=";WF
125 INPUT "TEMPS DE L'OBS.TM=";TM
130 CLS
135 S=0:R=0
140 G=9.81:Z=A*PI/180:DT=0.01
142 LOCATE 12,12:PRINT "CALCUL EN COURS"
145 DIM A(5,600):DIM B(5,600)
150 REM ETUDE DES TRAJECTOIRES
160 REM *****
165 N=0
200 FOR W=WO TO WF STEP -25
205 N=N+1:LOCATE 3,15:PRINT N
206 IF N>5 THEN GOTO 410
210 X=0:Y=0:TX=V*COS(Z):TY=V*SIN(Z)
215 M=0
220 FOR T=0 TO TM STEP DT
225 M=M+1
226 IF M=600 THEN GOTO 400
230 ATX=-K*SQR(TX*TX+TY*TY)*TX+L*W*TY
240 ACTX=ATX*DT:TX=TX+ACTX:ACX=TX*DT
250 X=X+ACX:A(N,M)=X
255 IF ABS(X)>S THEN S=ABS(X)
260 ATY=-K*SQR(TX*TX+TY*TY)*TY-L*W*TX-G
270 ACTY=ATY*DT:TY=TY+ACTY:ACY=TY*DT
280 Y=Y+ACY:B(N,M)=Y
285 IF ABS(Y)>R THEN R=ABS(Y)
290 IF Y<0 THEN GOTO 400
320 NEXT T
400 NEXT W
```

```

410 CLS
420 MODE 2
430 PLOT 0,0:DRAW 640,0
440 PLOT 0,0:DRAW 0,400
445 LOCATE 3,2:PRINT "Y":LOCATE 78,24:PRINT "X"
450 FOR I=1 TO N
460 FOR J=1 TO M
470 PLOT A(I,J)/S*600,B(I,J)/R*300
480 NEXT J
490 NEXT I
500 LOCATE 55,3:PRINT "V=";V;"M/S"
510 LOCATE 55,4:PRINT "A=";A;"DEG."
520 LOCATE 55,5:PRINT "K=";K
530 LOCATE 55,6:PRINT "L=";L
540 LOCATE 55,7:PRINT "WO=";WO;"T/S"
550 LOCATE 55,8:PRINT "WF=";WF;"T/S"
560 LOCATE 55,9:PRINT "TM=";TM

```



```



10 REM EFFET MAGNUS
20 REM *****
30 REM
40 REM ENTREE DES DONNEES
50 REM *****
60 CLS
70 INPUT "VALEUR DE V0=";V0
75 INPUT "VALEUR DE VF=";VF
80 INPUT "VALEUR DE ALPHA A=";A
90 INPUT "VALEUR DE K=";K
100 INPUT "VALEUR DE L=";L
110 INPUT "VALEUR DE W=";W
125 INPUT "TEMPS DE L'OBS.TM=";TM
130 CLS
135 S=0:R=0
140 G=9.81:Z=A*PI/180:DT=0.01
142 LOCATE 12,12:PRINT "CALCUL EN COURS"
145 DIM A(5,600):DIM B(5,600)
150 REM ETUDE DES TRAJECTOIRES
160 REM *****
165 N=0
200 FOR V=V0 TO VF STEP 20
205 N=N+1:LOCATE 3,15:PRINT N
206 IF N>5 THEN GOTO 410
210 X=0:Y=0:TX=V*COS(Z):TY=V*SIN(Z)
215 M=0
220 FOR T=0 TO TM STEP DT
225 M=M+1
226 IF M=600 THEN GOTO 400
230 ATX=-K*SQR(TX*TX+TY*TY)*TX+L*W*TY
240 ACTX=ATX*DT:TX=TX+ACTX:ACX=TX*DT
250 X=X+ACX:A(N,M)=X
255 IF ABS(X)>S THEN S=ABS(X)
260 ATY=-K*SQR(TX*TX+TY*TY)*TY-L*W*TX-G
270 ACTY=ATY*DT:TY=TY+ACTY:ACY=TY*DT
280 Y=Y+ACY:B(N,M)=Y
285 IF ABS(Y)>R THEN R=ABS(Y)
290 IF Y<0 THEN GOTO 400
320 NEXT T
400 NEXT V
410 CLS
420 MODE 2
430 PLOT 0,0:DRAWR 640,0
440 PLOT 0,0:DRAWR 0,400
445 LOCATE 3,2:PRINT "Y":LOCATE 78,24:PRINT "X"
450 FOR I=1 TO N
460 FOR J=1 TO M
470 PLOT A(I,J)/S*600,B(I,J)/R*300
480 NEXT J
490 NEXT I

```

```

500 LOCATE 55,2:PRINT "VO=";VO;"M/S"
505 LOCATE 55,3:PRINT "VF=";VF;"M/S"
510 LOCATE 55,4:PRINT "A=";A;"DEG."
520 LOCATE 55,5:PRINT "K=";K
530 LOCATE 55,6:PRINT "L=";L
540 LOCATE 55,7:PRINT "W=";W;"T/S"
550 LOCATE 55,9:PRINT "TM=";TM

```



## **5. Interactions gravitationnelles application à l'étude du mouvement des satellites terrestres**


La loi de gravitation exprime que deux masses  $M_1$  et  $M_2$  situées à une distance  $d$  l'une de l'autre s'attirent mutuellement avec une force d'intensité :

$$F = G \frac{M_1 M_2}{d^2}$$

où  $G$  est la constante de gravitation universelle ( $G = 6.67 \times 10^{-11}$  SI).

Etant donnée la faible valeur de  $G$ , l'intensité  $F$  ne devient appréciable que lorsque les masses mises en jeu sont importantes.

Ce programme permet de déterminer l'orbite parcourue par un satellite terrestre de masse  $m$  lancé à partir de l'altitude  $H_0$  avec une vitesse initiale  $V_0$  faisant un angle donné  $A$ , ni nul, ni plat avec la droite  $OM$  qui joint le centre de la Terre au centre de gravité du satellite.




### **Remarque :**

La lune sera toujours considérée dans une position telle que la perturbation introduite par celle-ci sur l'orbite du satellite soit considérée comme négligeable.

### **Etude de l'orbite du satellite**

Expression de la force  $\vec{F}$  qui agit sur le satellite :

$$\vec{F} = - G \frac{MTm}{R^2} \vec{p} \quad \text{avec } R = R_T + H$$



Energie d'interaction de gravitation :  $W(R) = - G \frac{MTm}{R}$

Les intégrales premières du mouvement traduisant la conservation du moment cinétique et celle de l'énergie s'écrivent respectivement :

$$mR^2 \frac{dB}{dt} = J \quad (J/m = \text{constante des aires}),$$

$$- G \frac{MTm}{R} + \frac{1}{2} m \left[ \left( \frac{dR}{dt} \right)^2 + R^2 \left( \frac{dB}{dt} \right)^2 \right] = E$$

En éliminant le temps entre elles, on obtient l'équation de la trajectoire du satellite en coordonnées polaires.

$$R^2 \left( \frac{dB}{dt} \right)^2 = \frac{J^2}{m^2 R^2} \left( \frac{dR}{dt} \right)^2 = \left( \frac{dR}{dB} \right)^2 \left( \frac{dB}{dt} \right)^2 = \frac{J^2}{m^2 R^4} \left( \frac{dR}{dB} \right)^2.$$

d'où

$$- G \frac{MTm}{R} + \frac{J^2}{2mR^2} \left[ \frac{1}{R^2} \left( \frac{dR}{dB} \right)^2 + 1 \right] = E$$

Effectuons le changement de variable

$$u = \frac{1}{R} \quad \text{donc} \quad \frac{du}{dB} = - \frac{1}{R^2} \frac{dR}{dB}$$

il vient

$$- GMTmu + \frac{J^2}{2m} \left[ \left( \frac{du}{dB} \right)^2 + u^2 \right] = E$$

décrivons par rapport à B,

$$\frac{du}{dB} \left[ -GMTm + \frac{J^2}{m} \left( \frac{d^2u}{dB^2} + u \right) \right] = 0$$

La solution  $\frac{du}{dB} = 0$  conduit à u indépendant de B, c'est-à-dire à des trajectoires circulaires ; la deuxième solution est celle de l'équation différentielle :

$$\frac{d^2u}{dB^2} + u = G \frac{MTm^2}{J^2}$$

On rappelle que l'orbite du satellite est une conique. La nature de cette conique dépend uniquement de la grandeur (et non de la direction) de la vitesse initiale.

Pour que le satellite puisse s'éloigner indéfiniment de la Terre, il faut que l'on ait  $E \geq 0$ , soit, d'après les conditions initiales

$$-G \frac{MTm}{(RT + HO)} + \frac{1}{2} mVO^2 \geq 0$$

l'orbite ne pourra comporter une branche infinie que si

$$\frac{1}{2} VO^2 \geq G \frac{MT}{(RT + HO)}$$

La vitesse minimale à communiquer au satellite appelée vitesse de libération VL est donnée par

$$VL = \sqrt{\frac{2GMT}{(RT + HO)}}$$

- Si  $VO < VL$ , l'orbite est elliptique,
- Si  $VO = VL$ , l'orbite est parabolique,
- Si  $VO > VL$ , l'orbite est hyperbolique.

Dans les deux derniers cas, l'orbite ne comprend que la branche de parabole ou d'hyperbole dont la concavité contient le foyer O, à la fois centre de la terre et centre d'attraction.

## Liste des variables

- G     constante universelle de gravitation
- MT    masse de la Terre (5.98E24 kg)



- RT rayon terrestre (6 370 km)  
 HO altitude initiale du satellite  
 VO vitesse initiale  
 A angle de  $\overrightarrow{VO}$  et du rayon vecteur initial  $\overrightarrow{OM}$  (en radian)  
 B angle du rayon vecteur initial et du rayon vecteur  $\overrightarrow{OM}$  à l'instant t (varie de 0 à  $2\pi$  au cours d'une révolution)  
 C constante des aires ( $C = J/m$ )  
 (on rappelle que  $C = VO (RT + HO)\sin A$ )  

$$U = \frac{1}{R} : \text{avec } R = RT + H$$

## Déroulement du programme

Le programme permet de tracer sur l'écran l'orbite du satellite et de faire apparaître à l'échelle de celle-ci la Terre ; dans le cas où les orbites sont elliptiques, on calcule la durée de révolution, l'altitude et la vitesse à l'apogée.

lignes 10 à 60	introduction
lignes 65 à 90	introduction des conditions initiales nécessaires à la résolution numérique de l'équation différentielle de mouvement
ligne 180	dans le cas d'une orbite parabolique ou hyperbolique, on arrête le calcul lorsque la distance $R = RT + H$ est supérieure à 150 fois le rayon terrestre
ligne 185	dans le cas d'une orbite elliptique selon les valeurs de A et de VO, le satellite peut retourner sur Terre, on arrête donc le calcul lorsque $R \leq RT$
ligne 186	on calcule, dans le cas des orbites elliptiques, la durée d'une révolution ou le temps que met le satellite à revenir sur Terre.
ligne 190	calcul du paramètre M qui permet de tracer l'orbite à l'aide d'une échelle calculée automatiquement.
ligne 210	résolution numérique de l'équation différentielle du mouvement
lignes 240 à 310	tracé des axes et d'un cercle correspondant à la Terre à l'échelle de l'orbite du satellite.
lignes 320 à 350	tracé de l'orbite du satellite.

lignes 400 à 430 impression sur écran des différents paramètres.

```

10 REM GRAVITATION
20 REM *****
30 REM
35 CLS
40 REM INTROD.DES CONSTANTES
50 REM *****
60 G=6.67E-11:RT=6370000:MT=5.98E+24
65 INPUT "PAS DU CALCUL DB=";DB
70 INPUT "ALTITUDE INITIALE H0=";H0
80 INPUT "VITESSE INITIALE V0=";V0
90 INPUT "VALEUR DE ALPHA=";A
100 LOCATE 12,15:PRINT "CALCUL EN COURS"
110 C=(RT+H0)*V0*SIN(A)
120 K=G*MT/(C^2)
130 L=2*PI/DB+1:DIM A(L):DIM B(L)
140 U=1/(RT+H0):TU=-(1/C)*V0*COS(A)
150 M=0:N=0:P=0:Q=0:T=0
160 FOR B=0 TO 2*PI STEP DB
170 N=N+1:A(N)=1/U
180 IF A(N)>150*RT THEN A(N)=0:P=1
185 IF A(N)<=RT THEN Q=1
186 IF Q=0 AND P=0 THEN B(N)=(1/U)^2/C*DB
187 IF Q=1 AND P=0 THEN B(N)=0
188 T=T+B(N)
190 IF ABS(A(N))>M THEN M=ABS(A(N)):W=N
200 LOCATE 1,17:PRINT N
210 DU=K-U:AD=DU*DB:TU=TU+AD:U=U+TU*DB
220 NEXT B
225 IF Q=0 OR Q=1 AND P=0 THEN K1=((A(W+1)-A(W))/B
(W)):K2=C/A(W):V=SQR(K1*K1+K2*K2)
230 CLS
240 MODE 2
245 LOCATE 50,25:PRINT INT((M/19.9*0.001)+0.5); "KM
"
246 PLOT 400,32:DRAWR 10,0
250 PLOT 320,0:DRAWR 0,400
260 PLOT 0,200:DRAWR 640,0
270 FOR Z=0 TO 2*PI STEP 2*PI/300
280 X=RT/M*199*COS(Z)+320
290 Y=RT/M*199*SIN(Z)+200
300 PLOT X,Y
310 NEXT Z
320 N=0
325 FOR B=0 TO 2*PI STEP DB
326 N=N+1

```


```

327 IF A(N)=0 OR A(N)<0 THEN GOTO 400
330 IF A(N)<=RT THEN GOTO 400
340 PLOT 320+A(N)/M*199*COS(B),200+A(N)/M*199*SIN(
B)
350 NEXT B
400 LOCATE 1,19:PRINT "HO=";HO*0.001;"KM":LOCATE 1
,20:PRINT "VO=";VO*0.001;"KM/S":LOCATE 1,21:PRINT
"A=";INT(A*180/PI+0.5);"DEG."
405 IF A(N)=0 OR A(N)<0 THEN STOP
410 IF Q=0 THEN LOCATE 1,1:PRINT "DUREE DE REVOLUT
ION=";T/3600;"H"
420 IF Q=1 AND P=0 THEN LOCATE 1,1:PRINT "DUREE RE
TOUR=";T/3600;"H"
430 IF Q=0 OR Q=1 AND P=0 THEN LOCATE 1,2:PRINT "A
LTITUDE APOGEE=";INT((M-RT)*0.001+0.5);"KM":LOCATE
1,3:PRINT "VITESSE APOGEE=";V*0.001;"KM/S"
440 STOP


```

DUREE DE REVOLUTION= 1.85183443 H  
ALTITUDE APOGEE= 2866 KM  
VITESSE APOGEE= 6.51471662 KM/S

MO= 588 KM  
VO= 8 KM/S  
A= 98 DEG.




DUREE DE REVOLUTION= 1.49982703 H  
 ALTITUDE APOGEE= 373 KM  
 VITESSE APOGEE= 7.63485738 KM/S




HD= 188 KM  
 VD= 7.86 KM/S  
 A= 90 DEG.

DUREE RETOUR= 7.36866955 H  
 ALTITUDE APOGEE= 31675 KM  
 VITESSE APOGEE= 0.876939287 KM/S




HD= 388 KM  
 VD= 10 KM/S  
 A= 30 DEG.

DUREE RETOUR= 16.5231487 H  
 ALTITUDE APOGEE= 55330 KM  
 VITESSE APOGEE= 0.982881055 KM/S



HD= 588 KM  
 VD= 10.2 KM/S  
 A= 60 DEG.



## 6. Conservation de l'énergie

Le wagon du grand-huit va-t-il réussir à monter jusqu'au sommet suivant ? Ce programme est destiné à montrer comment on peut résoudre des problèmes de physique à partir du principe de conservation de l'énergie. Il étudie le mouvement d'un objet que vous lancez sur une route dont vous donnez le profil.

### Pourquoi ce programme ?

Le principe de conservation de l'énergie est une « règle du jeu » que les physiciens tiennent absolument à respecter.

Ils définissent pour cela de nouvelles formes d'énergie chaque fois qu'ils découvrent un nouveau domaine. Il faut dire que ce principe est très efficace pour résoudre de nombreux problèmes.

### Utilisation du programme

Les données sont introduites sans unité, suivies de ENTER.

Pour faire avancer le programme, taper →

Pour sortir du programme, faire ESC ESC puis RUN ou RUN 200 ou MODE1.

L'objet a pour masse 1 kg. Les longueurs sont en mètres, les temps en secondes. Vous indiquez le profil de la route au moyen d'une équation en ligne 1 000 et la loi de frottement en ligne 1100. Vous pouvez éviter de recopier les lignes 10 à 190 qui ne sont qu'un préambule.

L'objet a, au départ, l'énergie E. L'ordinateur indique à chaque instant la valeur des énergies cinétique et potentielle sous la forme d'un pourcentage de E. Quand il y a frottement, de l'énergie mécanique disparaît et la somme EC + EP est différente de 100 %.

L'ordinateur augmente l'abscisse de l'objet de 4 mètres entre chaque image et indique le temps écoulé depuis le départ. Il ne s'écoule donc généralement pas un temps fixe entre deux images.

Les résultats peuvent être erronés, on observe par exemple des oscillations, lorsque la vitesse de l'objet est très faible.

Ne donnez pas une vitesse nulle au départ de l'objet.

## Le cœur du programme

Y est l'altitude de l'objet. E est l'énergie initiale. L'énergie cinétique est donnée par  $EC = (1/2).M.V^2$ , ici :  $0,5.V^2$ . L'énergie potentielle est donnée par  $EP = M.g.Y$ , ici :  $9,8.Y$ . En l'absence de frottement,  $EC + EP = E = \text{constante}$ .

Lorsque le point avance de dX en abscisse, son altitude varie de dY. Ici, dX n'est pas infiniment petit mais égal à 4, on veillera donc à ce que le profil de la route n'ait pas d'accident inférieur à une douzaine de mètres en abscisse, sinon on corrigera la ligne 270.

L'ordinateur calcule la nouvelle énergie potentielle du point à partir de la nouvelle valeur de Y et en déduit la nouvelle valeur de l'énergie cinétique par  $EC + EP = E$ . Il en déduit alors la nouvelle valeur de la vitesse.

S'il y a frottement, l'énergie de l'objet a diminué du travail de la force de frottement, soit

$$F. \sqrt{(dx)^2 + (dY)^2}$$

Tout se passe bien tant que la différence  $E - EP$  n'est pas négative, car cela aboutit à une énergie cinétique négative ce qui est impossible : l'objet fait alors demi-tour (lignes 560 et 550).

On a daté la position de l'objet ; ce travail, pénible lors d'une résolution par le calcul, est chose facile pour l'ordinateur. L'objet avance de la distance  $dL = \sqrt{(dx)^2 + (dY)^2}$  à la vitesse  $V$  pendant l'intervalle de temps  $dT = dL / V$ . La valeur de  $dT$  s'ajoute au temps précédent chaque fois que l'objet avance ou recule de  $dX = 4$ .

## Liste des variables

X,Y	abscisse et altitude de l'objet
YY	valeur de Y précédente
V	vitesse
F	force de frottement
E	énergie totale au départ
H	= $100 / E$
AC	= $EC / E$ en %, $AP = EP / E$ en %
S	= 1 : l'objet va de gauche à droite ; - 1, de droite à gauche
EC	énergie cinétique, EP énergie potentielle
W	= $E - EP$
dX	progression de l'objet en abscisse
dY	variation d'altitude quand l'objet avance de dX
dT	variation de temps quand l'objet avance de dX
Z	compteur de boucle pour l'affichage.

## Déroulement du programme

10	préambule
200	introduction de la vitesse initiale
220	tracé de la route
230, 1000	valeur de Y
240	tracé du profil
290	calcul de E, EC, EP initiaux, préparation de l'affichage
360	début de la boucle
	calcul de EC, EP, effacement et affichage des valeurs
430	effacement du mobile
440	le mobile avance
450	l'objet sort de l'écran, fin
460, 1000	calcul de la nouvelle altitude

470	marquage de la nouvelle position
490	calcul de l'augmentation de temps
530, 1100	calcul de la nouvelle énergie
La ligne 580 évite une division par zéro	
600	fin de la boucle

## Commentaire des illustrations

On a utilisé les lignes 1000 et 1100 du programme ci-joint (la force de frottement est constante) avec au départ  $v = 1$ .

Fig. 1 : l'énergie cinétique au départ est très faible, l'ordinateur l'arrondit à 0 %. L'énergie potentielle est grande.

Fig. 2 : l'objet descend la pente, arrive en bas, son énergie potentielle est alors nulle, son énergie cinétique est grande.

Fig. 3 : l'objet a perdu de l'énergie en raison des frottements. Son énergie mécanique est égale à  $11 + 74 = 85$  % de l'énergie initiale, aussi il n'a pas pu franchir la bosse. Après avoir fait demi-tour, (il a laissé une trace) il commence à redescendre.

Fig. 4 : l'objet s'immobilise sous l'action des frottements sur la pente droite du vallon. Il lui reste encore un peu d'énergie mécanique.

DANS LE PROGRAMME, LES LIGNES 1000 ET 1100 SONT DONNÉES A TITRE D'EXEMPLE.

```

10 CLS:MODE 1
20 LOCATE 6,8:PRINT"Conservation de l'énergie"
30 LOCATE 1,22:PRINT"pour faire avancer le program
me, appuyer sur "CHR$(243)
40 LOCATE 39,25:PRINT CHR$(243)
50 REM
60 REM
70 q=INKEY(1):IF q<>0 THEN 70
80 CLS
90 PRINT:PRINT"donner le profil du parcours,"
100 PRINT"y en fonction de x,"
110 PRINT"en lignes 1000 a 1090"
120 PRINT"(0<x<600      0<y<300)"
130 PRINT:PRINT"donner la force de frottement,"

```



```

140 PRINT"f en fonction de v,"
150 PRINT"en lignes 1100 a 1190"
160 PRINT:PRINT"faire run 200 pour lancer le progr
amme"
170 PRINT:LIST 1000-1090:LIST 1100-1190
180 REM
190 REM
200 MODE 1:CLS:INPUT"vitesse initiale ";v
210 CLS
220 FOR X=0 TO 600
230 GOSUB 1000
240 PLOT 10+X,Y,1
250 NEXT X
260 WINDOW 1,40,1,3
270 x=0:t=0:s=1:dx=4
280 GOSUB 1000
290 yy=y: ec=0.5*v^2:ep=9.8*yy :e=ec+ep
300 h=100/e
310 LOCATE 1,3:PRINT"%EC="
320 LOCATE 32,3:PRINT"%EP="
330 LOCATE 1,2:PRINT"t="
340 REM
350 REM
360 ec=0.5*v^2:ep=9.8*yy
370 ac=INT(ec*h):ap=INT(ep*h)
380 tt=0.001*INT(1000*t)
390 FOR z=3 TO 13:LOCATE z,2:PRINT" ":NEXT z

400 LOCATE 3,2:PRINT tt
410 FOR z=1 TO 3:LOCATE 4+z,3:PRINT" ":LOCATE 35+z
,3:PRINT" ":NEXT z
420 LOCATE 5,3:PRINT ac:LOCATE 36,3:PRINT ap
430 MOVE 10+x,yy+2:DRAWR 0,8,0
440 x=x+dx*s
450 IF x>600 OR x<0 THEN GOTO 1200
460 GOSUB 1000
470 MOVE 10+x,y+2:DRAWR 0,8,1
480 REM
490 dy=y-yy
500 dt=SQR(dx^2+dy^2)/v
510 t=t+dt
520 REM
530 GOSUB 1100
540 e=e-f*SQR(dx^2+dy^2)
550 w=e-9.8*y
560 IF w<0 THEN s=-s:GOTO 360
570 v=SQR(2*ABS(W))
580 IF v<1*10^-4 THEN v=1*10^-4
590 YY=Y
600 GOTO 360

```


```

610 REM
990 REM
1000 y=150+150*COS(0.012*x)
1095 RETURN
1096 REM
1100 f=0.5
1195 RETURN
1196 REM
1200 LOCATE 39,1:PRINT CHR$(243)
1210 q=INKEY(1):IF q<>0 THEN GOTO 1210
1220 GOTO 200

```


**t= 35.661**  
**%EC= 11**

**%EP= 74**



**t= 218.796**  
**%EC= 0**


**%EP= 8**



$$t = 0$$

$$\%EC = 0$$


$$\%EP = 99$$



$$t = 23.954$$

$$\%EC = 92$$

$$\%EP = 0$$



## 7. Plan incliné

Un objet placé sur un plan incliné est soumis à son poids à la réaction du plan et à une force  $\vec{F}$ . Est-il en équilibre ? Sinon quel est son mouvement ?

L'utilisateur indique à l'ordinateur des valeurs de tous les paramètres. L'ordinateur dessine l'objet sur le plan, les vecteurs représen-

tent les forces et il indique les valeurs de ces forces. Eventuellement, l'ordinateur indique le sens de déplacement de l'objet.

## Pourquoi ce programme ?

Un objet ponctuel soumis à plusieurs forces est en équilibre si la somme vectorielle de ces forces est nulle. Ce programme montre que le principe permet de résoudre des cas complexes comme celui-ci.

D'autre part nous modélisons le phénomène du frottement sec en une ligne de calcul (300), est-ce conforme à la réalité ? En essayant le programme dans de nombreux cas, vous conviendrez que ce modèle très simple donne des résultats satisfaisants.

On distinguera bien le principe qui régit l'équilibre que l'on ne peut remettre en question sans bouleverser la physique et l'équation qui représente le frottement qui n'est qu'une proposition intuitive, approximative et qui peut être modifiée au besoin.

## Utilisation du programme

Les masses sont en kg, les forces en N, les angles en degrés mais les données doivent être introduites sans unités.

Taper ENTER après l'introduction d'une donnée.

Taper → pour faire avancer le programme.

Taper ESC ESC pour sortir du programme.

Pour se familiariser avec le programme, on étudiera d'abord l'objet sur un plan horizontal ( $A = 0$ ) et sans frottement ( $K = 0$ ).

## Le cœur du programme

L'objet est soumis à son poids  $\vec{P}$ , à la force  $\vec{F}$ , à la réaction du sol que l'on a décomposé en  $N$  et  $FF$  (fig. 1).




Fig. 1. — Bilan des forces sur l'objet

Supposons l'objet en équilibre :

$$\vec{P} + \vec{F} + \vec{N} + \vec{FF} = \vec{0}$$

Projetons sur Oy :  $N = P \cos A + F \cdot \cos (AF - A + 90^\circ)$

sur Ox :  $FF + P \cdot \sin A = F \cdot \cos (AF - A)$

(lignes 260 et 270)

A, AF, P, F, étant donnés, on en déduit N et FF.

Une solution qui conduit à  $N < 0$  est à rejeter, elle signifierait que le plan tire sur l'objet pour le maintenir en équilibre. Comme il ne peut pas le faire, l'objet décolle du plan.

D'autre part, FF ne peut pas dépasser la valeur  $K.N$  (K : coefficient de frottement) (ligne 300).

Si la valeur de FF ne provient pas de l'équation (projection sur Ox), mais de  $FF = K \cdot N$ , la somme des forces appliquées ne sera pas nulle et l'objet se mettra en mouvement en glissant sur le plan.

## Liste des variables

A	angle du plan avec l'horizontale
AF	angle de la force F avec l'horizontale
F	valeur de la force
M	masse de l'objet
P	poids de l'objet
N	composante normale de la réaction du plan
FF	composante tangentielle de la réaction du plan
K	coefficient de frottement
FM	valeur de la plus grande force
ECH	coefficient d'échelle
C =	- 1 l'objet glisse vers la gauche
C =	1 l'objet glisse vers la droite
C =	2 l'objet est immobile.

## Déroulement du programme

10	présentation
90	introduction des données
250	calcul
280, 800	si l'objet décolle
300	l'objet glisse
360	dessin des forces
590, 1600	rappel des données et résultats
670	fin
680, 1600	rappel
700	retour à 90

```
10 CLS:DEG
20 LOCATE 14,8:PRINT"Plan incline"
30 LOCATE 1,21:PRINT"pour faire avancer le program
me, appuyer sur "CHR$(243)
40 LOCATE 39,25:PRINT CHR$(243)
50 REM
60 REM
70 q=INKEY(1):IF q<>0 THEN 70
80 CLS
90 PRINT:PRINT"masse de l'objet (kg) ?"
```

```

100 INPUT m
110 PRINT:PRINT"angle du plan incline avec l'horiz
ontale"
120 PRINT"(entre 0 et 89 deg) ?"
130 INPUT a
140 PRINT:PRINT"coefficient de frottement ?"
150 INPUT k
160 PRINT:PRINT"angle de la force f avec l'horizon
tale"
170 PRINT"(entre -90 et 90 deg) ?"
180 INPUT af
190 PRINT:PRINT"module de la force ? (en N) "
200 INPUT f
210 LOCATE 39,25:PRINT CHR$(243)
220 q=INKEY(1):IF q<>0 THEN 220
230 CLS
240 REM
250 p=9.8*m
260 n=p*COS(a)+f*COS(af-a+90)
270 ff=f*COS(af-a)-p*SIN(a)
280 IF n<0 THEN GOTO 800
290 c=2
300 IF ABS(ff)>k*n THEN c=SGN(ff):ff=k*n*c
310 fm=MAX(p,f,n,ff)
340 ech=160/fm
350 REM
360 MOVE 320-180*COS(a),200-180*SIN(a)
370 DRAW 360*COS(a),360*SIN(a)
380 TAG:MOVE 316,204:PRINT CHR$(238);
390 ORIGIN 320,200
400 DRAW ech*f*COS(af),ech*f*SIN(af)
410 MOVER 10,10:PRINT"F";
415 ORIGIN 320,200
420 DRAW ech*n*COS(a+90),ech*n*SIN(a+90)
430 MOVER 10,10:PRINT"N";
440 IF ff=0 THEN GOTO 500
445 ORIGIN 310,210
450 DRAW ech*ff*COS(a+180),ech*ff*SIN(a+180)


460 MOVER -c*10,-c*10:PRINT"FF";
500 ORIGIN 320,200
510 DRAW 0,-ech*p:MOVER 0,-10:PRINT"P";
520 TAGOFF:ORIGIN 0,0
530 REM
590 GOSUB 1600
640 REM
650 LOCATE 39,25:PRINT CHR$(243)
660 q=INKEY(1):IF q<>0 THEN 660
670 CLS
680 PRINT"essai precedent"

```


```

690 GOSUB 1600
700 GOTO 90
710 REM
720 REM
800 PRINT:PRINT"l'objet s'envole"
810 LOCATE 39,25:PRINT CHR$(243)
820 q=INKEY(1):IF q<>0 THEN 820
830 REM
1590 REM
1600 PRINT"m="m:PRINT"a="a:PRINT"k="k:PRINT"af="af
:PRINT"f="f
1610 PRINT:PRINT"p="INT(p):PRINT"f="INT(f)
1620 PRINT"n="INT(n):PRINT"ff="INT(ff)
1630 IF c=-1 THEN PRINT"l'objet glisse":PRINT" vers
la gauche"
1640 IF c=1 THEN PRINT"l'objet glisse":PRINT"vers
la droite"
1650 RETURN

```



→




→



```

m= 10
a= 45
k= 0.5
af= 0
f= 100
p= 98
f= 100
n= 140
ff= 1

```



→

```


m= 10
a= 45
k= 0.5
af= 0
f= 10

```

```

p= 98
f= 10
n= 76
ff= -39
l'objet glisse
vers la gauche

```



→

```


m= 10
a= 45
k= 0.5
af= 0
f= 300

```

```

p= 98
f= 300
n= 281
ff= 140
l'objet glisse
vers la droite

```



→

## 8. Dynamique

Votre astronef, dont les moteurs sont en panne, réussira-t-il à échapper à l'attraction de la planète Maudite ? Quelle relation y a-t-il entre la force appliquée à un point matériel et son mouvement ? C'est ce que vous montre ce programme où vous pouvez à tout instant modifier la force.

### Pourquoi ce programme ?

On utilise ici de la façon la plus simple qui soit le principe fondamental de la dynamique, clé de voûte de toute la mécanique.

### Utilisation du programme

Les masses sont exprimées en kg, les distances en mètres, les temps en secondes, les forces de Newton, mais les données doivent être introduites sans unité.

Après introduction d'une donnée, taper ENTER

Pour faire avancer le programme, taper —

Pour sortir du programme, faire ESC ESC, puis RUN ou *MODE 1*.

Différents types de force peuvent être imposés au point ; soit des forces pré-programmées (cas 2, 3, 4), soit une force déterminée par l'utilisateur (cas 1).

Après avoir choisi le cas, taper ENTER.

*Cas 2* : la force est constante de module 20 N et dirigée vers le bas. On pourra simuler ainsi une chute libre.

*Cas 3* : la force est dirigée vers le centre de l'écran, son module est constant égal à 100 N. La ligne 625 est destinée à éviter une division par zéro.

*Cas 4* : la force est centrale et inversement proportionnelle au carré de la distance du point au centre de l'écran. On pourra simuler ainsi le mouvement d'un objet sous l'action d'une force de gravitation.

Dans les cas 2, 3 et 4, on fera avancer le temps d'une seconde en appuyant sur la touche.

*Cas 1* : tous les dix tours de boucle, l'utilisateur redéfinit la force en donnant d'abord son module puis l'angle (en degrés) de sa direc-

tion avec l'horizontale (taper ENTER après l'introduction de chaque donnée). Essayez donc de faire suivre au point un trajet fixé à l'avance, ce n'est pas toujours facile.

## Le cœur du programme

Le point évolue sous l'action de la force  $\vec{F}$ . Cette évolution est déterminée la relation

$$\vec{F} = M \cdot \vec{A} \text{ (ligne 320)}$$

L'ordinateur détermine les composantes AX et AY de l'accélération et en déduit l'augmentation de vitesse par

$$d\vec{V} = \vec{A} \cdot dt$$

dt a été choisi, non pas infiniment petit, mais égal à 0,1 s. L'ordinateur calcul la nouvelle vitesse égale à  $V + dV$  par ses composantes  $VX + AX \cdot dt$  et  $VY + AY \cdot dt$  (ligne 330).

De la même façon, l'ordinateur détermine la nouvelle position du point par :  $X + VX \cdot dt$  et  $Y + VY \cdot dt$  (ligne 340).

Dans le cas (1), l'ordinateur effectue dix fois cette boucle de calcul avant de demander à l'utilisateur de redéfinir la force appliquée au point.

## Liste des variables

t	temps
z	compteur de boucle
x,y	position du point
dd	distance du point au centre de l'écran
xx	différence entre l'abscisse du point et l'abscisse du point et l'abscisse du centre de l'écran
v	module de la vitesse
an	angle de la vitesse avec l'horizontale
vx,vy	composantes de la vitesse
ax,ay	composantes de l'accélération
c	type de force
f	module de la force
s	angle de la force avec l'horizontale
fx,fy	composantes de la force

## Déroulement du programme

10	présentation
90	introduction des données
240	début de la boucle
280	si $c = 1$ , on va en 400 : introduction de la force
300	dessin de la force
310	sous boucle de $z = 1$ à 10
312	vers 500, 600 ou 700 : calcul de la force
	si $c = 1$ , c'est toujours la force introduite en 400
315	calcul du mouvement
350	fin de la sous-boucle
360	fin de la boucle

## Illustrations

Fig. 1 : l'utilisateur choisit la force (taper ENTER après chaque valeur). Départ 10, 390, vitesse 40, angle 0, masse 10.

Fig. 2 : force constante (taper → pour faire avancer le point).  
Départ 10,390, vitesse 20, angle 0, masse 10.

Fig. 3 : force centrale de module constant.  
Départ 320, 400, vitesse 16, angle 0, masse 50.

Fig. 4 : force centrale en  $1/d^2$ .  
Départ 320, 400, vitesse 6, angle 0, masse 50.  
Plus l'objet est près du centre de l'écran, plus la force est grande.

```
10 CLS:DEG:MODE 1
20 LOCATE 12,8:PRINT"Dynamique"
30 LOCATE 1,22:PRINT"pour faire avancer le program
me,appuyer sur"CHR$(243)
40 LOCATE 39,25:PRINT CHR$(243)
50 REM
60 REM
70 q=INKEY(1):IF q<>0 THEN 70
80 CLS
90 PRINT:PRINT"choix de la force"
100 PRINT:PRINT"1 quelconque","2 constante"
110 PRINT:PRINT"3 centrale","4 centrale en 1/d^2"
120 PRINT: INPUT c
130 PRINT:PRINT"position initiale ?"
```

```

140 INPUT"x=";x
150 INPUT"y=";y
160 PRINT:PRINT"vitesse initale ?"
170 INPUT"v=";v
180 INPUT"angle=";an
185 vx=v*COS(an):vy=v*SIN(an)
190 PRINT:INPUT"masse=";m
200 REM
210 LOCATE 39,25:PRINT CHR$(243)
220 q=INKEY(1):IF q<>0 THEN 220
230 CLS:TAG
235 IF c=1 THEN WINDOW 1,6,1,24
240 MOVE x-4,y+4:PRINT CHR$(203);
250 t=t+1
255 IF c=1 THEN GOTO 280
260 MOVE 620,10:PRINT CHR$(243);
270 q=INKEY(1):IF q<>0 THEN 270
280 IF c=1 THEN GOSUB 400
300 MOVE x,y:DRAWR fx,fy
310 FOR z=1 TO 10
312 IF c<>1 THEN ON (c-1) GOSUB 500,600,700
315 fx=f*COS(s):fy=f*SIN(s)
320 ax=fx/m:ay=fy/m
330 vx=vx+ax*0.1:vy=vy+ay*0.1
340 x=x+vx*0.1:y=y+vy*0.1
350 NEXT z
360 GOTO 240
370 REM
400 REM"quelconque"
430 TAGOFF: LOCATE 1,5+2*t:INPUT f
440 LOCATE 1,6+2*t:INPUT s:TAG
450 fx=f*COS(s):fy=f*SIN(s)
460 RETURN
500 REM"constante"
510 f=20:s=-90
520 RETURN
600 REM"centrale"
610 f=100
625 xx=x-320:IF xx=0 THEN xx=0.01
630 s=ATN((y-200)/xx)
640 IF x>320 THEN s=s+180
650 RETURN
700 REM"centrale en 1/d^2"
710 dd=(x-320)^2+(y-200)^2
720 f=(1*10^6)/dd
725 xx=x-320:IF xx=0 THEN xx=0.01
730 s=ATN((y-200)/xx)
740 IF x>320 THEN s=s+180
750 RETURN

```




Fig. 1




Fig. 2




Fig. 3




Fig. 4



## 9. Composition de mouvements

Un autobus parcourt les rues tortueuses de la ville, un passager se déplace d'un siège à l'autre dans l'autobus. Quel est le mouvement du passager par rapport au sol ? C'est la composition du mouvement de l'autobus par rapport au sol et du mouvement du passager par rapport à l'autobus. Ce programme d'animation vous propose quelques exemples de composition de mouvements simples.

### Pourquoi ce programme ?

N'avez-vous pas observé qu'une bille lâchée sur le plancher de l'autobus qui prend des virages a un comportement curieux ? Qu'il est difficile de suivre une trajectoire simple lorsque vous marchez sur le plateau d'un manège qui tourne ?

On s'intéresse au mouvement par rapport au sol parce que les lois de la physique permettent d'expliquer plus facilement le mouvement d'un objet par rapport à un repère fixe.

### Utilisation du programme

Vous choisissez d'abord le mouvement de l'objet par rapport au rectangle (mouvement relatif), puis le mouvement du rectangle par rap-

port au sol (mouvement d'entraînement), enfin vous observez le mouvement de l'objet par rapport au sol.

Pour faire avancer le programme (quand apparaît par exemple le mot "fin"), taper →

Pour choisir un mouvement, taper le chiffre correspondant puis ENTER.

Pour sortir du programme, faire ESC ESC.

Remarquons que ce programme permet également d'étudier la loi de composition des vitesses. La vitesse moyenne entre deux marques est égale à la distance entre ces points divisée par l'unité de temps de l'horloge. Mesurez les vitesses absolue, d'entraînement, et relative pour la même valeur de  $h$  ; dans le cas où le rectangle ne tourne pas vous pourrez vérifier que

$$\vec{V}_a = \vec{V}_e + \vec{V}_r$$

## Le cœur du programme

Connaissant la position du rectangle par rapport au sol (sommet A en  $x_a, y_a$ , le côté AB fait un angle  $a$  avec l'horizontale) et la position de l'objet par rapport au rectangle ( $x_r, y_r$ ), il faut calculer la position de l'objet par rapport au sol ( $x, y$ ). Ceci est réalisé en lignes 4020,4030 comme le montre la figure suivante (fig. 1).




Fig. 1. — Composition des mouvements



## Liste des variables

M = 1	lors du mouvement absolu
h	horloge
r	choix du mouvement relatif
d	choix du mouvement absolu
a, xa, ya	position du rectangle
xr, yr	position du point dans le rectangle
x, y	position du point

## Déroulement du programme

10	présentation
90	choix du mouvement relatif
150	boucle de h = 1 à 25
3000	calcul de la position relative
4000	position du point
210	choix du mouvement d'entraînement
280	boucle de h = 1 à 25
1000	position du rectangle
2000	dessin du rectangle
350	mouvement absolu
370	boucle de h = 1 à 25
1000	position du rectangle
3000	calcul de la position relative
4000	position du point
1470	le rectangle est effacé
999	fin

## Quelques exemples d'utilisation

La combinaison de deux mouvements rectilignes uniformes conduit à un mouvement rectiligne uniforme (fig. 2).

La combinaison d'un mouvement rectiligne uniforme et d'un mouvement rectiligne uniformément accéléré conduit à un mouvement parabolique (fig. 3a, 3b, 3c).

La combinaison d'un mouvement rectiligne uniforme et d'un mouvement circulaire conduit à un mouvement cycloïdal (fig. 4a, 4b, 4c). Ce mouvement est analogue à celui du point d'une roue d'une voiture se déplaçant à vitesse constante.

```

10 CLS:INK 0,0:INK 1,26
20 LOCATE 14,8:PRINT"COMPOSITION"
30 LOCATE 13,12:PRINT"DE MOUVEMENTS"
40 LOCATE 1,22:PRINT"pour faire avancer le program
me,appuyer sur "CHR$(243)
50 REM
60 REM
70 q=INKEY(1):IF q<>0 THEN 70
80 CLS
90 PRINT"Choix du mouvement relatif"
100 m=0:xa=0:ya=0:a=0 :e=1
110 PRINT:PRINT"1:immobile","2:uniforme"
120 PRINT:PRINT"3:accelere","4:circulaire"
130 PRINT:INPUT r
140 CLS:GOSUB 2000
150 FOR h=1 TO 25
155 FOR z=0 TO 500:NEXT z
160 LOCATE 35,1:PRINT"h="h
170 ON r GOSUB 3100,3200,3300,3400
180 NEXT h
190 LOCATE 1,1:PRINT"fin"
195 REM
200 q=INKEY(1):IF q<>0 THEN 200
210 CLS:PRINT"Choix du mouvement d'entrainement"

220 m=0:xr=0:yr=0
230 PRINT:PRINT"1:uniforme horizontal lent"
240 PRINT:PRINT"2:uniforme horizontal rapide"
250 PRINT:PRINT"3:uniforme vertical"
260 PRINT:PRINT"4:circulaire"
270 PRINT:INPUT d
280 CLS:FOR h=1 TO 25
285 FOR z=0 TO 500:NEXT z
290 LOCATE 35,1:PRINT"h="h
300 ON d GOSUB 1100,1200,1300,1400
310 NEXT h
315 LOCATE 1,1 :PRINT"fin"
320 REM
330 REM
340 q=INKEY(1):IF q<>0 THEN 340
350 CLS:PRINT"mouvement absolu"
360 m=1
370 FOR h=1 TO 25
380 LOCATE 35,1:PRINT"h="h
390 ON d GOSUB 1100,1200,1300,1400
400 NEXT h
405 LOCATE 1,3:PRINT"fin"
410 REM
420 REM
430 q=INKEY(1):IF q<>0 THEN 430

```

```

440 CLS:GOTO 90
450 REM
999 END
1000 REM
1010 REM"entrainement"
1100 a=0:ya=120
1110 xa=7*h +10
1120 GOTO 1435
1200 a=0:ya=120
1210 xa=14*h
1220 GOTO 1435
1300 a=0:xa=170
1310 ya=8*h+10
1320 GOTO 1435
1400 a=0.25*h
1410 xa=340-170*COS(a+0.49)
1420 ya=200-170*SIN(a+0.49)
1435 e=1:GOSUB 2000
1440 IF m=0 THEN RETURN
1450 ON r GOSUB 3100,3200,3300,3400
1460 FOR z=0 TO 500:NEXT z
1470 e=0:GOSUB 2000
1480 RETURN
2000 REM
2010 REM"rectangle"
2020 ORIGIN xa, ya
2030 s=SIN(a):c=COS(a)
2040 DRAWR 300*c,300*s,e
2050 DRAWR -160*s,160*c,e
2060 DRAWR -300*c,-300*s,e
2070 DRAWR 160*s,-160*c,e
2080 RETURN
3000 REM
3010 REM"relatif"
3100 xr=30:yr=30:GOTO 4000
3200 xr=12*h:yr=100:GOTO 4000
3300 xr=0.48*h*h:yr=100::GOTO 4000
3400 xr=80+60*COS(0.252*h):yr=80+60*SIN(0.252*h):G
OTO 4000
4000 REM
4010 REM"marquage du point"
4015 ORIGIN 0,0
4020 x=xa+xr*COS(a)-yr*SIN(a)
4030 y=ya+xr*SIN(a)+yr*COS(a)
4040 PLOT x,y
4050 RETURN

```

**mouvement absolu**  
**fin**

**h= 25**




Fig. 2

**h= 25**




Fig. 3a

**h= 25**




Fig. 3b

**mouvement absolu**  
**fin**

**h= 25**




Fig. 3c

**h= 25**




Fig. 4a

**h= 25**




Fig. 4b

**mouvement absolu**  
**fin**

**h= 25**



Fig. 4c

## 10. Le pendule simple

Il est à la base de l'étude des mouvements sinusoïdaux. Rappelons l'équation différentielle qui régit son mouvement :

$$l \frac{d^2\theta}{dt^2} + g \sin \theta = 0$$

Si l'on se limite à l'étude des "petits" mouvements,  $\sin \theta \approx \theta$  et donc l'équation se simplifie :

$$l \frac{d^2\theta}{dt^2} + g \theta = 0$$

dont on connaît la solution analytique :

$$\theta = \theta_0 \cos (\omega t + \sigma_0)$$

pour laquelle :

$$\omega = \frac{2\pi}{T} \text{ et } T = 2\pi \sqrt{\frac{l}{g}}$$

Nous proposons de simuler une telle expérience, et de comparer les résultats obtenus à ceux fournis par l'intégration directe de l'équation différentielle non simplifiée. Le pendule sera lâché dans une certaine position, sans vitesse initiale.

### Programme 1

Il permet de visualiser le mouvement de deux pendules, l'un réel, obéissant à l'équation différentielle de départ, l'autre théorique, satisfaisant à l'équation simplifiée. Les dessins (1) à (5) montrent divers aspects de l'écran à des instants consécutifs. Il s'agit là d'une simulation du phénomène.


### Programme 2

Il permet de représenter les élongations angulaires des deux pendules précédents en fonction du temps. On constate que, si les deux pendules sont bien synchrones pour les petits angles initiaux : (6), ils ne le sont plus dès que ces mêmes angles ont de grandes valeurs : (7) et (8).

```


10 REM *****
20 REM *
30 REM * pendule *
40 REM *
50 REM *****
60 REM
70 INPUT "longueur du pendule:";l
80 PRINT "position initiale en degres:";INPUT "(+ a
droite,- a gauche)";t0
90 g=9.81
100 CLS:x=320:y=350
110 t=2*PI*(l/g)^0.5:ti=t0
120 dt=t/100:t1=t0
130 t0=2*PI*t0/360
140 t1=2*PI*t1/360
150 a=g/l:p0=0:p1=0:GOSUB 260
160 t=0
170 FOR n = 0 TO 478
180 d0=-a*t0:d1=-a*SIN(t1)
190 t=t+dt:LOCATE 2,23:PRINT "t=";t
200 p0=p0+d0*dt:P1=P1+d1*dt
210 GOSUB 310:GOSUB 330
220 t0=t0+p0*dt:t1=t1+P1*dt
230 GOSUB 300:GOSUB 320
240 NEXT n
250 STOP
260 LOCATE 2,20:PRINT "l= ";l;"m."
270 LOCATE 2,21:PRINT "periode theorique=";t;"s."
280 LOCATE 2,22:PRINT "angle de depart=";ti;"degres
."
290 RETURN
300 PLOT x,y:DRAW 1*190*SIN(t0),-1*190*COS(t0),1
:RETURN
310 PLOT x,y:DRAW 1*190*SIN(t0),-1*190*COS(t0),0:
RETURN
320 PLOT x,y:DRAW 1*190*SIN(t1),-1*190*COS(t1),1:
RETURN
330 PLOT x,y:DRAW 1*190*SIN(t1),-1*190*COS(t1),0:
RETURN

```




l= 1 m.  
periode theorique= 2.00606668 s.  
angle de depart=-20 degrees.  
t= 0.521577338

Fig. 1



l= 1 m.  
periode theorique= 2.00606668 s.  
angle de depart=-20 degrees.  
t= 2.52764402

Fig. 2



l= 1 m.  
periode theorique= 2.00606668 s.  
angle de depart=-20 degrees.  
t= 4.53371068

Fig. 3





l= 1 m.  
 periode theorique= 2.00606668 s.  
 angle de depart=-20 degrees.  
 t= 6.53977727

Fig. 4



l= 1 m.  
 periode theorique= 2.00606668 s.  
 angle de depart=-20 degrees.  
 t= 8.54584386

Fig. 5

```

10 REM *****
20 REM *
30 REM * pendule *
40 REM *
50 REM *****
60 REM
70 INPUT"longueur du pendule:";l
80 PRINT"position initiale en degres:";INPUT"(+ a
droite,- a gauche)";t0
90 g=9.81
100 MODE 2
110 CLS:x=320:y=350

```

```

120 t=2*PI*(1/g)^0.5:ti=t0
130 dt=t/100:t1=t0
140 t0=2*PI*t0/360
150 t1=2*PI*t1/360
160 a=g/l:p0=0:p1=0:GOSUB 300
170 t=0
180 PLOT 0,200:DRAW 639,200
190 PLOT 0,30:DRAW 0,380
200 FOR n=0 TO 5 :PLOT 5,200+(PI/2-n*PI/4)*100:DRA
WR -5,0 : NEXT
210 FOR n = 0 TO 639
220 d0=-a*t0:d1=-a*SIN(t1)
230 t=t+dt:LOCATE 2,23:PRINT "t=";t
240 p0=p0+d0*dt:P1=P1+d1*dt
250 t0=t0+p0*dt:t1=t1+P1*dt
260 PLOT n,200 + T0*100
270 PLOT n,200 + T1*100
280 NEXT n
290 STOP
300 LOCATE 2,20:PRINT"l= ";l;"m."
310 LOCATE 2,21:PRINT"periode theorique=";t;"s."
320 LOCATE 2,22:PRINT"angle de depart=";ti;"degres
."
330 RETURN

```




Fig. 6




Fig. 7




Fig. 8

## 11. Le pendule de Foucault


L'expérience du pendule de Foucault est l'une des plus célèbres du dix-neuvième siècle.

Foucault attribua le phénomène observé à la rotation de la Terre autour de l'axe des pôles.

Le plan d'oscillation d'un pendule simple abandonné sans vitesse initiale tourne lentement autour de la verticale du point d'attache du fil.

## Mise en équations du phénomène

On se propose d'étudier l'influence de la rotation de la Terre sur le mouvement, rapporté à un référentiel lié à la Terre, d'un pendule pesant supposé réduit à une particule M de masse m suspendue à un fil de longueur l fixé en F.



Z est la latitude du lieu considéré.

Z est la latitude du lieu considéré.

Dans ce référentiel (non galiléen), la masse m est soumise :

- à son poids  $-mg\vec{k}$  (ce qui revient à négliger la force d'inertie d'entraînement par rapport à la force d'attraction terrestre) ;
- à la tension  $\vec{T}$  du fil ;
- à la force de Coriolis (force d'inertie complémentaire) :  
 $\vec{f}_c \equiv -2m\vec{A} \wedge \vec{V}$  ;
- $\vec{A}$  est le vecteur rotation de la Terre.

Appliquons la relation fondamentale de la Dynamique.

$$m\vec{\gamma} = m\vec{g} + \vec{T} - 2m\vec{A} \wedge \vec{v}$$

Soit en projetant sur les axes :

$$\frac{d^2x}{dt^2} = -\frac{T}{m} \frac{x}{l} - 2A (\cos Z \frac{dz}{dt} - \sin Z \frac{dy}{dt})$$

$$\frac{d^2y}{dt^2} = - \frac{T}{m} \frac{y}{l} - 2A \sin Z \frac{dx}{dt}$$

$$\frac{d^2z}{dt^2} = -g + \frac{T}{m} \left(1 - \frac{z}{l}\right) + 2A \cos Z \frac{dx}{dt}$$

Si le mouvement est supposé plan, on prend  $z = 0$ ,

$$\frac{d^2z}{dt^2} = 0 \text{ et } \frac{dz}{dt} = 0.$$

$$\text{On a donc } 0 = -g + \frac{T}{m} + 2A \cos Z \frac{dx}{dt}$$

Le dernier terme étant négligeable devant  $g$ , on peut écrire  $T \simeq mg$ .  
Le système d'équation devient :

$$\frac{d^2x}{dt^2} = - \frac{g}{l} x + 2A \sin Z \frac{dy}{dt}$$

$$\frac{d^2y}{dt^2} = - \frac{g}{l} y - 2A \sin Z \frac{dx}{dt}$$

Les caractéristiques du pendule monté par Foucault au Panthéon à Paris étaient les suivantes :  $l = 67 \text{ m}$ , période d'oscillation ( $T = 2\pi \sqrt{l/g} = 16,421 \text{ s}$ ,  $g = 9,81 \text{ m/s}^2$  et  $Z = 48^\circ 51'$ ).

**Remarque :** Pour la Terre  $A = 0,73 \cdot 10^{-4} \text{ rad/s}$ .

Foucault énonça dans les termes suivants la loi qui relie la durée du plan de révolution du plan d'oscillation à la latitude du lieu :

“A mesure que l'on approche de l'équateur, le plan de l'horizon prend sur l'axe de la Terre une position de plus en plus inclinée, et la verticale, au lieu de tourner sur elle-même comme au pôle, décrit un cône de plus en plus ouvert ; il en résulte un ralentissement dans le mouvement du plan d'oscillation, mouvement qui annule à l'équateur pour changer de sens dans l'autre hémisphère”.

Le programme suivant permet la résolution numérique des équations du mouvement du pendule et calcule la déviation au cours de chaque oscillation.

Le mouvement est simulé à la latitude de Paris, en un lieu de même latitude en valeur absolue dans l'hémisphère sud et enfin à l'équateur où l'on observe aucune déviation.

## Liste des variables

A	vitesse angulaire de la Terre ( $73\text{E-}4$ rad/s)
G	accélération de la pesanteur à la latitude de Paris ( $9.81$ m/s <sup>2</sup> )
Z	latitude de Paris $48.85$ (exprimée en degré décimal) à l'équateur $Z = 0$
L	longueur du pendule ( $67$ m)

## Déroulement du programme

lignes 10 à 50	titre et introduction des données
ligne 70	pas de calcul ( $dT = 1/100$ s)
lignes 90 et 100	tracé des axes
lignes 110 à 200	résolution numérique des équations différentielles du mouvement et tracé sur l'écran du mouvement de l'extrémité du pendule sur le plan horizontal.
ligne 300	l'ordinateur imprime sur l'écran la déviation observée au cours d'une oscillation.

## Remarque :


Nous retrouvons bien les  $2.7$  mm observés par Foucault au cours de son expérience au Panthéon. La rotation du plan d'oscillation à la latitude de Paris est d'environ  $12$  degrés par heure.

```
10 REM PENDULE DE FOUCAULT
20 REM *****
30 REM
40 REM MOUVEMENT DANS L'HEMISPHERE NORD LATITUDE D
E PARIS 48 51'
50 A=0.000073:G=9.81:Z=48.85*PI/180:L=67
60 CLS
70 DT=0.01
80 MODE 2
85 LOCATE 32,1:PRINT "PENDULE DE FOUCAULT"
90 PLOT 320,0:DRAWR 0,400
100 PLOT 0,200:DRAWR 640,0
110 REM CONDITIONS INITIALES
120 REM *****
130 X=3:Y=0:TX=0:TY=0
140 REM EQUATION DE LA TRAJECTOIRE
```

```

150 REM *****
160 FOR F=0 TO 16.421 STEP DT
170 ATX=-G/L*X+2*A*SIN(Z)*TY:ACTX=ATX*DT:TX=TX+ACT
X:ACX=TX*DT:X=X+ACX
180 ATY=-G/L*Y-2*A*SIN(Z)*TX:ACTY=ATY*DT:TY=TY+ACT
Y:ACY=TY*DT:Y=Y+ACY
190 PLOT X*100+320,Y*20000+200
200 NEXT F
300 LOCATE 2,3:PRINT "DEVIATION=";Y*1000;"MM"

```



```


10 REM PENDULE DE FOUCAULT
20 REM *****
30 REM
40 REM MOUVEMENT DANS L'HEMISPHERE SUD LATITUDE -
48 51'
50 A=0.000073:G=9.81:Z=-48.85*PI/180:L=67
60 CLS
70 DT=0.01
80 MODE 2
85 LOCATE 32,1:PRINT "PENDULE DE FOUCAULT"

```

```

90 PLOT 320,0:DRAWR 0,400
100 PLOT 0,200:DRAWR 640,0
110 REM CONDITIONS INITIALES
120 REM *****
130 X=3:Y=0:TX=0:TY=0
140 REM EQUATION DE LA TRAJECTOIRE
150 REM *****
160 FOR F=0 TO 16.421 STEP DT
170 ATX=-G/L*X+2*A*SIN(Z)*TY:ACTX=ATX*DT:TX=TX+ACT
X:ACX=TX*DT:X=X+ACX
180 ATY=-G/L*Y-2*A*SIN(Z)*TX:ACTY=ATY*DT:TY=TY+ACT
Y:ACY=TY*DT:Y=Y+ACY
190 PLOT X*100+320,Y*20000+200
200 NEXT F
300 LOCATE 2,3:PRINT "DEVIATION=";Y*1000;"MM"

```



```

10 REM PENDULE DE FOUCAULT
20 REM *****
30 REM
40 REM MOUVEMENT A L'EQUATEUR LATITUDE 0
50 A=0.000073:G=9.81:Z=0:L=67

```




```

60 CLS
70 DT=0.01
80 MODE 2
85 LOCATE 32,1:PRINT "PENDULE DE FOUCAUT"
90 PLOT 320,0:DRAW 0,400
100 PLOT 0,200:DRAW 640,0
110 REM CONDITIONS INITIALES
120 REM *****
130 X=3:Y=0:TX=0:TY=0
140 REM EQUATION DE LA TRAJECTOIRE
150 REM *****
160 FOR F=0 TO 16.421 STEP DT
170 ATX=-G/L*X+2*A*SIN(Z)*TY:ACTX=ATX*DT:TX=TX+ACT
X:ACX=TX*DT:X=X+ACX
180 ATY=-G/L*Y-2*A*SIN(Z)*TX:ACTY=ATY*DT:TY=TY+ACT
Y:ACY=TY*DT:Y=Y+ACY
190 PLOT X*100+320,Y*20000+200
200 NEXT F
300 LOCATE 2,3:PRINT "DEVIATION=";Y*1000;"MM"

```

# PENDULE DE FOUCAUT

DEVIATION= 0 MM




## 12. Oscillateurs couplés

La production d'oscillations forcées dans un système oscillant met en évidence des forces dont les modules sont des fonctions sinusoïdales du temps.

Ces forces sont produites par un système oscillant appelé "excitateur" que l'on fait agir sur le système à étudier appelé "résonateur".

Si la transmission des forces se fait entre les deux systèmes par une liaison élastique, on parle alors d'**oscillateurs couplés**.



### Mise en équations

Envisageons le cas où les deux oscillateurs sont des ressorts R1 et R2 liés par un troisième ressort R. Les constantes de raideur sont respectivement K1, K2 et K. Les ressorts R1 et R sont liés par une masse M1 en A1. Les ressorts R2 et R sont liés par une masse M2 en A2.

- appelons  $l$  la distance  $O_1 O_2$
- appelons  $a_1$ ,  $a_2$  et  $a$  les longueurs respectives des ressorts R1, R2 et R hors tension.

Soit  $Z_1$  et  $Z_2$  les cotes de A1 et A2 en mouvement et  $Z_{10}$  et  $Z_{20}$  à l'équilibre.

Les équations d'équilibre s'écrivent :

$$- M_1 g + K_1 (l - a_1 - Z_{10}) - K (Z_{10} - Z_{20} - a) = 0,$$

$$- M_2 g - K_2 (Z_{20} - a_2) + K (Z_{10} - Z_{20} - a) = 0$$

Si les forces de frottement fluide sont supposées négligeables, c'est-à-dire si l'on considère les ressorts comme des oscillateurs non amortis, les équations du mouvement s'écrivent :

$$- M_1 g + K_1 (l - Z_1 - a_1) - K (Z_1 - Z_2 - a) = M_1 \frac{d^2 Z_1}{dt^2},$$

$$- M_2 g - K_2 (Z_2 - a_2) + K (Z_1 - Z_2 - a) = M_2 \frac{d^2 Z_2}{dt^2}$$

Posons  $X_1 = Z_1 - Z_{10}$

et  $X_2 = Z_2 - Z_{20}$

On peut écrire :

$$M_1 \frac{d^2 X_1}{dt^2} = - (K_1 + K) X_1 + K X_2$$

$$M_2 \frac{d^2 X_2}{dt^2} = - (K_2 + K) X_2 + K X_1$$

Les exemples présentés traitent les cas concernant un couplage faible entre deux oscillateurs identiques ( $K_1 = K_2$ ) liés par un ressort de constante de raideur  $K$ . ( $M_1 = M_2$ )

C'est le cas par exemple de deux pendules identiques faiblement couplés.

### Remarque :

L'utilisateur pourra étudier à loisir ses exemples personnes. Nous avons aussi introduit dans le programme une force de frottement fluide de la forme  $B \, dx/dt$  (ligne 270 et ligne 310).

### Liste des variables

$K, K_1, K_2$	constantes de raideur des ressorts $R, R_1$ et $R_2$
$M_1, M_2$	valeurs des masses $M_1$ et $M_2$
$D$	durée de l'observation (maximale prévue)
$B_1$ et $B_2$	coefficients correspondant à des forces de frottement fluide
$X_0$ et $Y_0$	valeurs de $Z_{10}$ et $Z_{20}$ à l'instant initial
$TX_0$ et $TY_0$	valeurs de $dZ_1/dt$ et $dZ_2/dt$ à l'instant initial.

## Déroulement du programme


- lignes 10 à 200 titre et entrées des données (ne pas oublier B1 et B2 si vous désirez observer le mouvement avec un frottement fluide)
- ligne 210 on efface l'écran
- ligne 230 conditions initiales
- lignes 240 à 360 résolution numérique des deux équations différentielles : l'ordinateur calcule 600 points pour chaque équation (ligne 220). Les variables S pour X et R pour Y permettent le calcul d'une échelle automatique pour tracer les courbes.
- lignes 400 à 600 on trace les axes, les courbes  $X = f(T)$  et  $Y = f(T)$  et on imprime les données.


```
10 REM OSCILLATEURS COUPLES
20 REM *****
30 REM
40 REM ENTREE DES DONNEES
50 REM *****
55 CLS
60 INPUT "VALEUR DE K=";K
70 INPUT "VALEUR DE K1=";K1
80 INPUT "VALEUR DE M1=";M1
90 INPUT "VALEUR DE K2=";K2
100 INPUT "VALEUR DE M2=";M2
110 INPUT "DUREE OBSERVATION D=";D
150 REM CONDITIONS INITIALES
160 REM *****
170 INPUT "VALEUR DE X0=";X0
180 INPUT "VALEUR DE TX0=";TX0
190 INPUT "VALEUR DE Y0=";Y0
200 INPUT "VALEUR DE TY0=";TY0
210 CLS:DT=D/600
215 LOCATE 12,12:PRINT "CALCUL EN COURS"
220 DIM A(600):DIM B(600)
230 X=X0:S=ABS(X0):Y=Y0:R=ABS(Y0):A(0)=X0:B(0)=Y0
240 FOR T=0 TO D STEP DT
250 N=N+1:LOCATE 3,15:PRINT N
260 IF N=600 THEN GOTO 355
270 ATX=(-(K1+K)*X+K*Y)/M1+B1*TX
280 ACTX=ATX*DT:TX=TX+ACTX
290 X=X+TX*DT:A(N)=X
300 IF ABS(X)>S THEN S=ABS(X)
310 ATY=(-(K2+K)*Y+K*X)/M2+B2*TY
```

```

320 ACTY=ATY*DT:TY=TY+ACTY
330 Y=Y+TY*DT:B(N)=Y
340 IF ABS(Y)>R THEN R=ABS(Y)
350 NEXT T
355 IF S>=R THEN F=S
360 IF R>S THEN F=R
400 MODE 2
410 PLOT 0,0:DRAWR 0,400
420 PLOT 0,200:DRAWR 640,0
430 FOR I=0 TO N
440 PLOT I,A(I)/F*199+200:PLOT I,B(I)/F*199+200
450 NEXT I
500 LOCATE 15,17:PRINT "K=";K
510 LOCATE 15,18:PRINT "K1=";K1
520 LOCATE 15,19:PRINT "K2=";K2
530 LOCATE 15,20:PRINT "M1=";M1
540 LOCATE 15,21:PRINT "M2=";M2
550 LOCATE 15,22:PRINT "D=";D
560 LOCATE 15,23:PRINT "DT=";DT
570 LOCATE 35,17:PRINT "X0=";X0
580 LOCATE 35,18:PRINT "TX0=";TX0
590 LOCATE 35,19:PRINT "Y0=";Y0
600 LOCATE 35,20:PRINT "TY0=";TY0

```





# 2

## Electricité

### ***1. Champ et potentiel créés par un système de charges ponctuelles***

I. Ce programme intéresse aussi bien la mécanique que l'électricité. En effet, les équations fournissant les composantes du champ électrique produit par une charge ponctuelle sont de même nature que celles fournissant les composantes du champ gravitationnel produit par une masse ponctuelle. Le traitement de ce genre de problème est donc le même dans les deux cas : cependant, ce n'est qu'en électricité qu'il existe des charges négatives ; en mécanique, toutes les masses sont positives.

Ce programme propose donc d'esquisser les lignes de champ et les équipotentiels du champ produit par un système de charges (ou de masses) ponctuelles placées sur l'écran. Seules sont esquissées en chaque point du plan (l'écran) les tangentes à ces courbes.

Rappelons que :

- la force produite par une charge électrique est proportionnelle au champ électrique qu'elle crée. Les lignes de champ sont tangentes en tout point au champ électrique ;

- une équipotentielle est une ligne perpendiculaire en chaque point à la ligne de champ qui y passe.

II. Le programme est simple : une fois définies les positions et les valeurs des charges sur l'écran, il faut calculer la distance du point

courant à chacune des charges, évaluer les composantes sur OX et OY des champs électriques produits par les charges en ce point, en déduire les composantes du champ résultant, calculer le module du champ résultant, puis la tangente de la ligne de champ en ce même point (d'où la longueur des calculs). On trace alors un petit segment de droite, dont la pente est donnée par la valeur de la tangente ainsi calculée. Un certain nombre de points sont ainsi examinés sur l'écran. Le tracé des équipotentiellles se fait de la même manière, dans la mesure où l'on sait que, par définition, en chaque point, une équipotentielle est perpendiculaire à la ligne de champ.

III. Le programme ne présente pas de difficultés majeures (jusqu'à la ligne 240). A partir de là intervient un problème lié à l'esthétique du dessin. Nous nous sommes imposé de tracer des petits segments (symbolisant des petits morceaux de ligne de champ) de longueur constante égale à 9 pixels. En effet, pour une pente donnée, si l'on impose  $\Delta X$ , il est possible que, suivant la valeur de la tangente, Y devienne très grand puisque  $\Delta Y = \Delta X \cdot \text{tg}\alpha$  (dessin 1) s'imposant, on a donc :

$$A^2 = (\Delta X)^2 + (\Delta Y)^2, \text{ on connaît } B = \frac{\Delta Y}{\Delta X} = \text{tg}\alpha$$

D'où, pour un point de coordonnées X et Y, les composantes du petit segment que l'on y trace :

$$\Delta X^2 = \frac{A^2}{B^2 + 1} \quad \Delta Y = B \cdot \Delta X$$

On voit qu'il n'y a plus de problème même si B tend vers l'infini.

En outre, pour le tracé des équipotentiellles, le calcul reste le même, sauf lignes 260 à 270. En effet (dessin 2), on peut tracer un segment de droite perpendiculaire au précédent en utilisant comme composantes :

$$\Delta X' = \Delta Y \quad \text{et} \quad \Delta Y' = - \Delta X$$




Fig. 1






Fig. 2

## Liste des variables

$X1, X2, Y1, Y2$	positions des deux charges
$A2$	carré de la longueur d'un segment ( $= 9 \times 9$ ici)
$Q1$ et $Q2$	valeurs des charges électriques
$Y$ et $X$	position du point courant
$R1$ et $R2$	distances du point courant aux deux charges

EX, EY	composantes du champ produit en X, Y par la première charge
FX, FY	composantes du champ produit en X, Y par la seconde charge
EX, EY	composantes du champ résultant
E	module du champ résultant
B	tangente de l'angle entre l'horizontale et le champ résultant
DX, DY	composantes ( $\Delta X$ , $\Delta Y$ ) du petit segment.

## Détail du programme

80	position des deux charges
90	valeur des deux charges et longueur des petits segments
170 - 180	définissent les points qui vont être examinés
190 - 200	évaluation de la distance du point courant à la première charge ; calcul du champ produit par celle-ci
210 - 220	idem pour la seconde charge
230	calcul des coordonnées du champ total
240	calcul de la tangente du champ par rapport à l'horizontale
250 - 260	calcul de $\Delta X$ et $\Delta Y$ (voir plus haut)
270	positionnement du point et tracé du petit segment de droite

```

10 REM *****
20 REM *
30 REM * champ electrique *
40 REM *
50 REM *****
60 REM
70 MODE 2
80 x1=200:x2=400:y1=200:y2=200
90 a2=81:q1=1:q2=1
100 PRINT #8 > "x1="; x1
110 PRINT #8 > "y1="; y1
120 PRINT #8 > "x2="; x2

```

```


130 PRINT #8 , "y2="; y2
140 PRINT #8 , "q1="; q1
150 PRINT #8 , "q2="; q2
160 CLS
170 FOR y = 5 TO 395 STEP 8
180 FOR x= 5 TO 635 STEP 8
190 r1= (SQR((x-x1)^2+(y-y1)^2))^3
200 ex=q1*(x-x1)/R1:ey=q1*(y-y1)/R1
210 r2= (SQR((x-x2)^2+(y-y2)^2))^3
220 fx=q2*(x-x2)/R2:fy=q2*(y-y2)/R2
230 ex=ex+fx:ey=ey+fy
240 b=ey/ex
250 dx=(a2/(b^2+1))^0.5
260 dy=b*dx
270 PLOT x-dx/2,y-dy/2:DRAWR dx,dy
280 NEXT x
290 NEXT y
300 STOP

```


```

x1= 200
y1= 200
x2= 400
y2= 200
q1= 1
q2= 1


```



```
x1= 200  
y1= 200  
x2= 400  
y2= 200  
q1= 1  
q2=-1
```




```
x1= 200  
y1= 200  
x2= 400  
y2= 200  
q1= 4  
q2=-1
```



```

x1= 200
y1= 200
x2= 400
y2= 200
q1= 0
q2= 1
x1= 200
y1= 200
x2= 400
y2= 200
q1= 0
q2= 1

```



```

10 REM *****
20 REM *
30 REM * equipotentielle *
40 REM *
50 REM *****
60 REM
70 MODE 2
80 x1=200:x2=400:y1=200:y2=200
90 a2=81:q1=1:q2=-1
100 PRINT #8 ,"x1=";x1
110 PRINT #8 ,"y1=";y1
120 PRINT #8 ,"x2=";x2
130 PRINT #8 ,"y2=";y2

```

```


140 PRINT #8 , "q1="; q1
150 PRINT #8 , "q2="; q2
160 CLS
170 FOR y = 5 TO 395 STEP 8
180 FOR x = 5 TO 635 STEP 8
190 r1= (SQR((x-x1)^2+(y-y1)^2))^3
200 ex=q1*(x-x1)/R1:ey=q1*(y-y1)/R1
210 r2= (SQR((x-x2)^2+(y-y2)^2))^3
220 fx=q2*(x-x2)/R2:fy=q2*(y-y2)/R2
230 ex=ex+fx:ey=ey+fy
240 b=ey/ex
250 dx=(a2/(b^2+1))^0.5
260 dy=-b*dx
270 PLOT x-dy/2,y-dx/2:DRAWR dy,dx
280 NEXT x
290 NEXT y

```


```

x1= 200
y1= 200
x2= 400
y2= 200
q1= 0
q2= 1


```



x1= 200  
y1= 200  
x2= 400  
y2= 200  
q1= 1  
q2= 1



x1= 200  
y1= 200  
x2= 400  
y2= 200  
q1= 4  
q2=-1



x1= 200


y1= 200

x2= 400

y2= 200

q1= 1

q2=-1



## **2. Circuit électrique**

Soulevez le coin d'une nappe, quelle forme prend-elle ? Chauffez le bord d'une plaque, quelle est la température en ses différents points ? Appliquez une tension électrique entre deux points d'un réseau de résistances électriques, quel est le potentiel en chacun des nœuds du réseau ? Ces problèmes sont voisins, c'est le dernier exemple cité que nous avons choisi de traiter ici.

### **Pourquoi ce programme ?**

... pour vous présenter une méthode de résolution par approximations successives. Cette méthode est applicable immédiatement



à des réseaux très étendus. Si nous nous sommes contentés ici d'un réseau à six nœuds, c'est pour que la résolution soit rapide, facilement vérifiable "à la main", pour pouvoir dessiner facilement le réseau et indiquer les valeurs des résistances et des potentiels. Mais le programme ne serait pas plus long à écrire pour un réseau de six cents nœuds. Enfin elle présente l'avantage de ne comporter qu'une ou deux lignes de calcul.

Cette méthode pose le problème du compromis rapidité-précision : si vous désirez des résultats dix fois plus précis, les calculs seront dix fois plus longs.

## Utilisation du programme

Vous donnerez les valeurs des résistances en Ohms, sans introduire l'unité, par un nombre entier compris entre 1 et 999. Vous pouvez également envisager des résistances nulles ou des circuits ouverts. Le programme assimile les court-circuits à des résistances de un micro-Ohm.

La tension appliquée entre les bornes du circuit est de 100 Volts. Les résultats sont donnés à un volt près. Pour obtenir une plus grande précision, modifiez les lignes 380 et 520.

## Le cœur du programme

Deux relations très simples sont utilisées (fig. 1) :

La loi d'Ohm :  $U_1 - U = R_1 \cdot I_1$

ou en posant  $G_1 = 1/R_1$

$I_1 = G_1 \cdot (U_1 - U)$

La loi des nœuds :  $I_1 + I_2 + I_3 + I_4 = 0$

soit :

$G_1 \cdot (U_1 - U) + G_2 \cdot (U_2 - U) + G_3 \cdot (U_3 - U) + G_4 \cdot (U_4 - U) = 0$

d'où :

$G_1 \cdot U_1 + G_2 \cdot U_2 + G_3 \cdot U_3 + G_4 \cdot U_4 = U \cdot (G_1 + G_2 + G_3 + G_4)$

Dans le programme, on a noté :  $G = G_1 + G_2 + G_3 + G_4$  (ligne 360)

d'où :  $U = (G_1 \cdot U_1 + G_2 \cdot U_2 + G_3 \cdot U_3 + G_4 \cdot U_4) / G$  (ligne 370)




Fig. 1

Au départ, tous les potentiels sont fixés arbitrairement à 50 V. Puis on applique les potentiels 0 et 100 V aux bornes de gauche. On se propose de calculer le potentiel de chaque nœud au moyen de la relation précédente, mais on ne peut pas les calculer simultanément. Le calcul du potentiel en un point fait intervenir les potentiels voisins qui sont erronés. Les premiers résultats obtenus sont donc erronés.

Après avoir fait un tour de réseau, on en refait un autre avec des valeurs un peu moins erronées que les valeurs arbitraires de départ... et ainsi de suite jusqu'à ce que deux tours de réseau fournissent des résultats pratiquement identiques (ligne 380).

Voici ce que cela donne sur un exemple simple (fig. 2) :





Fig. 2

Il est très agréable de n'avoir qu'une ligne de calcul, malheureusement tous les nœuds du réseau ne sont pas dans la même situation. Il y a des points marginaux (fig. 3) et plus le réseau est petit, plus la proportion de points marginaux est importante.



O nœud marginal  
X nœud normal (4 voisins)

Circuit électrique

Fig. 3

Nous avons donc été amenés à créer des nœuds fantômes et des résistances fantômes. Ces résistances sont infinies (ligne 260), afin que le potentiel des nœuds fantômes n'aient aucun effet sur le réseau réel. La figure 4 vous montre la notation utilisée.




Fig. 4

## Liste des variables

z	compteur de boucle
x,y	indices
R\$	résistance
G	conductance
C = 1	si la conductance est nulle
U	nouveau potentiel
V	ancien potentiel
T = 0	si le calcul est terminé

## Déroulement du programme

10	présentation
90	dessin du circuit
120	introduction et affichage des valeurs des résistances
260	conductances fantômes
290	initialisation des potentiels
340	calcul des potentiels
380	le calcul est-il terminé ?
430	les potentiels calculés deviennent "anciens"
680	en route pour un nouveau tour de piste
500	affichage des résultats

## Illustrations

Fig. 5 : toutes les résistances sont identiques.

Fig. 6 : un cas très simple...

Fig. 7 : le potentiel des nœuds supérieurs est divisé par deux à chaque cellule

Fig. 8 : le potentiel des nœuds supérieurs est diminué de vingt volts à chaque cellule

Fig. 9 : les nœuds de droite sont isolés du circuit. Ils sont restés au potentiel arbitraire de 50 volts.

```

10 CLS
20 LOCATE 12,6:PRINT"CIRCUIT ELECTRIQUE"
30 LOCATE 10,12:PRINT"court-circuit,taper: -"
40 LOCATE 10,14:PRINT"circuit ouvert,taper: /"
50 LOCATE 1,22:PRINT"pour faire avancer le program
me, taper"CHR$(243)
60 LOCATE 39,25: PRINT CHR$(243)
70 q=INKEY(1):IF q<>0 THEN 70
80 CLS:TAG
90 MOVE 48,54:DRAWR 568,0:MOVE 48,328:DRAWR 568,0
100 FOR z=0 TO 2:MOVE 232+192*z,54:DRAWR 0,274:NEX
T z
110 DIM g(9,9):DIM u(9,9):DIM v(9,9)
120 FOR X=2 TO 6 STEP 2
130 FOR Y=3 TO 5
140 MOVE -64+96*X,576-96*Y:PRINT"?";
150 MOVE 0,400:INPUT r$
160 IF r$="/" THEN G(X,Y)=0:C=1:GOTO 200
170 IF R$="-" THEN G(X,Y)=1000000:GOTO 225
180 C=0:G(X,Y)=1/VAL(R$)
200 FOR Z=0 TO 2
210 IF Y<>4 THEN MOVE -64+96*x+16*z,747-137*y:PRIN
T CHR$(143-15*C); ELSE MOVE 32+96*x,202-16*z:PRINT
CHR$(143-15*C);
220 NEXT Z
225 MOVE -64+96*X,576-96*Y:PRINT r$;
230 NEXT Y
240 NEXT X
245 FOR z=0 TO 624 STEP 16:MOVE z,400:PRINT" ";:NE
XT z
250 REM
260 FOR X=3 TO 7 STEP 2:G(X,2)=0:G(X,6)=0:NEXT X
270 G(8,3)=0:G(8,5)=0
280 REM
290 FOR X=1 TO 9:FOR Y=1 TO 9:U(X,Y)=50:V(X,Y)=50:
NEXT Y:NEXT X
300 U(1,3)=100:U(1,5)=0:MOVE 16,384:PRINT"100V";:M
OVE 16,32:PRINT"0 V";
310 REM
330 T=1
340 FOR X=3 TO 7 STEP 2
350 FOR Y=3 TO 5 STEP 2
360 G=G(X-1,Y)+G(X-1,Y+1)+G(X-1,Y-1)+G(X+1,Y)
370 U(X,Y)=(U(X-2,Y)*G(X-1,Y)+U(X,Y+2)*G(X-1,Y+1)+
U(X,Y-2)*G(X-1,Y-1)+U(X+2,Y)*G(X+1,Y))/G
380 IF ABS(U(X,Y)-V(X,Y))>0.2 THEN T=0
390 NEXT Y
400 NEXT X
410 REM
420 IF T=1 THEN GOTO 500


```




```


430 FOR X=3 TO 7 STEP 2
440 FOR Y=3 TO 5 STEP 2
450 V(X,Y)=U(X,Y)
460 NEXT Y
470 NEXT X
480 GOTO 330
490 REM
500 FOR X=3 TO 7 STEP 2
510 FOR Y=3 TO 5 STEP 2
520 MOVE -96+96*x,902-172*y:PRINT STR$(INT(u(x,y))
)+ "v";
530 NEXT Y
540 NEXT X
550 MOVE 16,620:PRINT CHR$(243);
560 q=INKEY(1):IF q<>0 THEN 560
565 CALL &A000
570 END

```

la ligne 565 est destinée à l'imprimante







### 3. Circuits électriques : droites de charge

Ce programme propose l'étude de circuits électriques simples, contenant un générateur et un récepteur. Celui-ci peut être, au choix de l'utilisateur : un résistor, une batterie ou une diode Zener.

Le problème suivant est posé : les éléments du générateur (force électromotrice et résistance interne) étant connus, ainsi que ceux du récepteur, quel est le point de fonctionnement de l'ensemble ? Le programme teste alors les résultats introduits au clavier, vérifie leur validité (à  $\pm 10\%$  près) et en cas d'erreur, fournit les bonnes valeurs de  $U$  et de  $I$ .

Puis, à titre de vérification, les caractéristiques des deux dipôles sont tracées ; leur intersection fournit le point de fonctionnement de l'ensemble.

Toutes les quantités numériques proposées sont absolument aléatoires, ce qui assure des problèmes perpétuellement nouveaux.

#### Programme

Il se compose de plusieurs parties :

- le tableau de présentation (80 - 170),



- le choix aléatoire des éléments (190 - 260),
- le calcul des éléments (280 - 800) et de la solution U, I :
  - générateur (280 - 370),
  - résistor (420 - 470),
  - diode Zener (480 - 550),
  - batterie (560 - 680).
- l'introduction de la solution (690 - 710) et sa vérification (720 - 730),
  - l'évaluation de son exactitude et l'affichage de la solution proposée par l'ordinateur (770 - 790),
  - le dessin des axes sur l'écran et leurs graduations (820 - 930),
  - le dessin de la caractéristique du générateur (940),
  - le dessin de la caractéristique du résistor (960 - 1010),
  - le dessin de la caractéristique de la diode Zener (1020 - 1080),
  - le dessin de la caractéristique de la batterie (1090 - 1150) et sa prolongation côté  $I < 0$  si par hasard  $V > E$  (1170 - 1260).

## Quelques remarques concernant les calculs

- Choix aléatoire des éléments ; on a ;

$$10 < E < 50 \quad \frac{E}{3} < V_Z < \frac{2E}{3}$$

$$100 < R < 1000 \quad 10 < R_Z < 20$$

$$\frac{R}{3} < R_1 < \frac{R}{3} + 8/3 * RND \quad 10 < V < 50$$

$$100 < R_P < 1000$$

Toutes ces valeurs sont bien sûr modifiables. A ce propos, il est prévu une sécurité pour  $V_Z > E$  (ligne 520).

- Les échelles sont telles que :
  - sur OX, axe des courants I : 88 pixels, 0,1 ampère ;
  - sur OY, axe des tensions U : 60 pixels, 10 volts.
- Une droite de pente R se construit alors de la manière suivante :

$$R = \frac{dU}{dI} = \frac{dY/6}{dX/800}$$


Donc, pour  $dX = 1$  pixel, il faut avoir

$$dY \text{ (en pixels)} = \frac{6}{880} R$$

D'où le tracé (lignes 980, 1050...).

- Dans le cas où  $V < E$ , le point d'intersection des caractéristiques est à gauche de l'axe des ordonnées ( $I < 0$ ) ; celles-ci sont prolongées en conséquence (lignes 1170 à 1260).

Définition des variables introduites dans le texte et dans le programme.



```

10 REM *****
20 REM *
30 REM * point de fonctionnement *
40 REM *
50 REM *****
60 REM
70 CLS
80 PRINT:PRINT"point de fonctionnement"
90 PRINT:PRINT"-generateur "
100 PRINT:PRINT"il est defini par sa resistance in
terne r et sa f.e.m e"
110 PRINT:PRINT"-recepteur"
120 PRINT:PRINT"vous avez le choix entre:"
130 PRINT:PRINT"resistor,de resistance r.....(1)"
140 PRINT:PRINT"diode zener,de tension vz,r...(2)"
150 PRINT:PRINT"batterie,f.e.m e,resistance r.(3)"

```

```

160 PRINT:PRINT"quel recepteur choisissiez vous?"
170 INPUT a
180 CLS
190 REM choix des elements
200 PRINT"appuyez 4 fois sur la touche a"
210 FOR n=1 TO 4
220 x=RND(1)
230 a$=INKEY$
240 IF a$<>"a" AND a$ <> "a" THEN 220
250 a(n)=x
260 NEXT n
270 CLS
280 REM generateur
290 e=a(1)*50
300 IF e < 10 THEN e=e+10
310 e=INT(e)
320 r=a(2)*1000
330 IF r < 100 THEN r=r+100
340 r=INT(r)
350 PRINT:PRINT"generateur
360 PRINT:PRINT" f.e.m=";e;"v."
370 PRINT"resistance interne=";r;"ohm."
380 REM recepteurs
390 PRINT:PRINT"recepteur"
400 ON a GOSUB 420,480,560,670
410 GOTO 690
420 REM resistor
430 r1=r/3+8*a(3)/3:r1=INT (r1)
440 u=r1*e/(r+r1)
450 i=e/(r+r1)
460 PRINT:PRINT"resistor de resistance :";r1;"ohm."
"
470 RETURN
480 REM diode zener
490 vz=e/3+e*a(3)/3:vz=INT(vz)
500 rz =10+10*a(4):rz=INT(rz)
510 u=vz+rz*(e-vz)/(r+rz)
520 i=(e-vz)/(r+rz):IF i< 0 THEN i=0:u=e
525 PRINT:PRINT"diode zener"
530 PRINT"tension de seuil:";vz;"v,"
540 PRINT"resistance:";rz;"ohm."
550 RETURN
560 REM batterie
570 v=a(3)*50
580 IF v <10 THEN v=v+10
590 v=INT (v)
600 rp=a(3)*1000
610 IF rp<100 THEN rp=rp+100
620 rp=INT (rp)
630 i=(e-v)/(r+rp)

```

```

640 u=e* rp/(r+rp)+v*r/(r+rp)
650 PRINT:PRINT"batterie"
660 PRINT" f.e.m: ";v;"v."
670 PRINT" resistance interne :";rp;"ohm."
680 RETURN
690 PRINT:PRINT"donnez moi votre solution"
700 PRINT:INPUT"u=";us
710 INPUT"i=";is:PRINT
720 IF us<1.1*u AND us>0.9*u THEN GOTO 730 ELSE GO
TO 750
730 IF ABS (is)<1.1*ABS(i)AND ABS(is)>0.9*ABS(i) T
HEN GOTO 740 ELSE GOTO 750
740 PRINT"bonne reponse":GOTO 800
750 PRINT"mauvaise reponse"
760 PRINT
770 PRINT"il fallait repondre:":PRINT
780 PRINT"u=";:PRINT USING"##.##";u;:PRINT"v."
790 PRINT"i=";:PRINT USING"##.##";i;:PRINT" a."
800 z$=INKEY$:IF z$="" THEN 800
810 MODE 2
820 PLOT 40,40:DRAWR 0,320
830 PLOT 40,40:DRAWR 560,0
840 PLOT 40,320:DRAWR -4,-4:PLOT 560,40 :DRAWR -4,
4
850 PLOT 40,320 :DRAWR 4,-4:PLOT 560,40 :DRAWR -4,
-4
860 REM graduation oy : 1 graduation/ 10 volts .
870 FOR j = 1 TO 4 : PLOT 38 , 60 * j + 40 : DRAWR
4,0 : NEXT j
880 LOCATE 7,5 : PRINT"U"
890 LOCATE 7,8:PRINT"40 V"
900 REM graduation ox : 1 graduation/ 0.1 ampere .
910 FOR j=1 TO 5 : PLOT 42 + 880 * 0.1 * j ,40 : D
RAWR 0,-4 : NEXT j
920 LOCATE 69,22:PRINT"I"
930 LOCATE 60,24:PRINT"0.5 A "
940 PLOT 40 , 6*e +40:DRAW 40+880*e/r,40
950 ON a GOTO 960,1020,1090
960 REM resistor
970 PLOT 40,40
980 x=x+1 : y= y+3*r1/440
990 IF x > 560 OR y > 320 THEN 1270
1000 DRAWR x,y
1010 GOTO 980
1020 REM diode zener
1030 x=0:y=0
1040 PLOT 40 , 40+6*vz
1050 x=x+1 : y = y+3*rz/440
1060 IF x > 560 OR y > 320 THEN 1270
1070 DRAWR x,y

```

```

1080 GOTO 1050
1090 REM batterie
1100 x=0 : y=0
1110 PLOT 40 , 40 + 6*v
1120 x=x+1: y=y+3*rp/440
1130 IF x > 560 OR y > 320 THEN 1160
1140 DRAWR x,y
1150 GOTO 1120
1160 IF e > v THEN GOTO 1270
1170 x=0:y=0:PLOT 40 , 40 + 6*v
1180 x=x-1: y=y-3*rp/440
1190 IF 40+x < 0 OR y+6*v < 0 THEN 1220
1200 DRAWR -1 , -3*rp/440
1210 GOTO 1180
1220 x=0:y=0 : PLOT 40,40+6*e
1230 x=x-1:y=y+3*r/440
1240 IF 40+x < 0 OR 40+y > 399 THEN 1270
1250 DRAWR -1,3*r/440
1260 GOTO 1230
1270 STOP

```

point de fonctionnement

-generateur

il est defini par sa resistance interne r et sa f.e.m e

-recepteur

vous avez le choix entre:

resistor,de resistance r.....(1)  
 diode zener,de tension vz,r...(2)  
 batterie,f.e.m e,resistance r.(3)

quel recepteur choisissez vous?

? 1

generateur

f.e.m= 19 v.  
 resistance interne= 178 ohm.

recepteur

resistor de resistance : 68 ohm.


donnez moi votre solution

u=? 4  
 i=? 0.08

mauvaise reponse

il fallait repondre:

u= 4.8v.  
 i= 0.08 a.



point de fonctionnement

-generateur

il est defini par sa resistance interne  $r$  et sa f.e.m  $e$

-recepteur

vous avez le choix entre:

resistor, de resistance  $r$ .....(1)

diode zener, de tension  $v_z, r$ ...(2)

batterie, f.e.m  $e$ , resistance  $r$ .(3)

quel recepteur choisissez vous?

? 2

generateur

f.e.m= 41 v.

resistance interne= 664 ohm.

recepteur

tension de seuil: 25 v,


resistance: 19 ohm.

donnez moi votre solution

$u=?$  25

$i=?$  0.023

bonne reponse



point de fonctionnement

-generateur

il est defini par sa resistance interne  $r$  et sa f.e.m  $e$

-recepteur

vous avez le choix entre:

resistor, de resistance  $r$ .....(1)

diode zener, de tension  $v_z, r$ ...(2)

batterie, f.e.m  $e$ , resistance  $r$ .(3)

quel recepteur choisissez vous?

$\begin{smallmatrix} 4 \\ 3 \end{smallmatrix}$

generateur

f.e.m= 30 v.

resistance interne= 478 ohm.

recepteur

batterie

f.e.m: 29 v.


resistance interne : 587 ohm.

donnez moi votre solution

$u=?$  29.5

$i=?$  0.001

bonne reponse



point de fonctionnement

-generateur

il est defini par sa resistance interne  $r$  et sa f.e.m  $e$

-recepteur

vous avez le choix entre:

resistor, de resistance  $r$ .....(1)

diode zener, de tension  $v_z, r$ ...(2)

batterie, f.e.m  $e$ , resistance  $r$ .(3)

quel recepteur choisissez vous?

$\begin{smallmatrix} 4 \\ 3 \end{smallmatrix}$

generateur

f.e.m= 16 v.

resistance interne= 544 ohm.

recepteur

batterie

f.e.m: 25 v.


resistance interne : 514 ohm.

donnez moi votre solution

u=? 20.6

i=? 0.0085

bonne reponse



## 4. Résonance

On étudie la résonance en tension aux bornes du condensateur d'un circuit R L C soumis à une tension sinusoïdale de un volt. L'ordinateur trace les courbes donnant la tension aux bornes du condensateur en fonction de la fréquence, et cela pour diverses valeurs de la résistance, ainsi que le lieu des points correspondant à la fréquence propre du circuit. Après quoi l'utilisateur donne les valeurs des éléments du circuit et la fréquence de l'excitation. L'ordinateur place alors le point correspondant sur le graphique.

### Pourquoi ce programme ?

Pour la forme : la représentation d'une famille de courbes en perspective est toujours attractive et parle mieux à l'imagination.



Pour le fond : la résonance est un phénomène très courant en mécanique, en électricité : pourquoi les vitres du vieil autobus tremblent-elles quand le moteur tourne au ralenti ? Pourquoi la bouteille dans laquelle vous soufflez émet-elle un son ? Pourquoi pouvez-vous sélectionner avec votre poste un émetteur de radio ? Pourquoi faites-vous déborder le café de la tasse quand vous agitez la cuillère à une certaine vitesse ?... Parce qu'il y a résonance.

Nous étudions ici un système électrique, mais les résultats sont valables pour un système mécanique avec la correspondance suivante :





Fig. 1



## Utilisation du programme

Les données sont introduites en Ohm, Henry, microFarad, Hertz, mais sans indiquer l'unité, suivies de ENTER.

Pour faire avancer le programme, taper —

Pour sortir du programme, faites ESC ESC puis RUN ou MODE I

Pour voir sur l'écran le point correspondant au circuit que vous proposez, il faut que la fréquence soit inférieure à deux fois la fréquence propre  $F_0$  et que la résistance soit comprise entre 3,75 et 0,023 fois la résistance caractéristique  $R_0$  du circuit.

## Le cœur du programme

L'ordinateur trace les courbes en coordonnées réduites, ainsi quelle que soit la fréquence de résonance, la courbe représentant le circuit se situe à la même place.

La valeur remarquable de la fréquence est la fréquence de résonance en courant  $F_0$ , qui correspond à la pulsation  $\omega_0$  telle que :

$$L.C\omega_0^2 = 1$$

La valeur remarquable de la résistance est la résistance critique  $R_o$  telle que :

$$R_o = 2.F. \sqrt{L/C}$$

La fréquence réduite est donc  $F / F_o = \omega/\omega_o = fr$

La résistance réduite est donc  $R / R_o = rr$

Soit  $I$  l'intensité complexe dans le circuit,  $Z_c$  l'impédance complexe du condensateur et  $Z$  l'impédance complexe totale du circuit :

$$\underline{U_e} = \underline{Z} \cdot \underline{I} \quad \underline{U_s} = \underline{Z_c} \cdot \underline{I} \quad \underline{U_s} = \underline{U_e} \cdot \underline{Z_c} / \underline{Z}$$

$$\underline{Z} = R + j(L\omega - 1/C\omega) \quad \text{et} \quad \underline{Z_c} = -j/C\omega$$

En multipliant haut et bas par  $jC\omega$ , on obtient :

$$\underline{U_s} = \underline{U_e} \cdot 1 / (jRC\omega - LC\omega^2 + 1)$$

$$\underline{U_s} = \underline{U_e} \cdot 1 (2j.rr.fr + 1 - fr^2)$$

En module :

$$U_s = U_e \cdot 1/F. \sqrt{4.rr^2.fr^2 + (1 - fr^2)^2} \quad (\text{ligne 1010})$$

## Liste des variables

n	numéro de la courbe
rr	résistance réduite
fr	fréquence réduite
L	inductance de la bobine
c	capacité du condensateur
Fo	fréquence de résonance en courant
Ro	résistance critique
x,y	coordonnées du point
u	tension aux bornes du condensateur
a,b	coordonnées de la case texte effacée

## Déroulement du programme

10	présentation
82	préparation du graphique
95	boucle des 20 courbes
1000	calcul des coordonnées du point
1020 et 1030	effet de perspective
180	tracé de la courbe

1000	calcul des coordonnées du point
250	introduction des données
280	calcul de $F_0$ et $R_0$
340	calcul de $f_r$ et $r_r$
1000	calcul des coordonnées du point
360	affichage des résultats
400	effacement des résultats

## Commentaire des illustrations

Dans le cas de la figure 2, la fréquence imposée est égale à la fréquence propre du circuit mais l'amortissement est élevé si bien que la tension aux bornes du condensateur est inférieure à la tension appliquée. Il en est de même pour la figure 3 bien que l'amortissement soit faible parce que la fréquence est trop élevée. Par contre, dans la figure 4, la fréquence est voisine de la fréquence propre, l'amortissement est faible, il y a surtension aux bornes du condensateur.

```

10 CLS:MODE 1
20 LOCATE 14,8:PRINT"resonance"
30 LOCATE 1,22:PRINT"pour faire avancer le program
me, appuyer sur"CHR$(243)
40 LOCATE 39,25:PRINT CHR$(243)
50 REM
60 REM
70 q=INKEY(1):IF q<>0 THEN 70
80 CLS
82 MOVE 14,94:DRAW 14,34:DRAW 494,-26:DRAW 550,60

84 WINDOW 1,6,1,5
86 LOCATE 2,1:PRINT"u":LOCATE 2,2:PRINT CHR$(209):
LOCATE 1,3:PRINT"r"CHR$(47)CHR$(208)"f"
90 rr=3.75
95 FOR n=0 TO 19
100 fr=0:GOSUB 1000
110 MOVE x,y
120 FOR fr=0 TO 2 STEP 0.03
130 GOSUB 1000:DRAW x,y
140 NEXT fr
150 rr=rr/1.22
160 NEXT n
170 REM
180 rr=3.75:fr=1:GOSUB 1000
190 FOR n=0 TO 59

```

```

200 MOVE x,y
210 GOSUB 1000: DRAW x,y
220 rr=rr/1.07
230 NEXT n
240 REM
250 WINDOW 25,40,1,11
260 LOCATE 25,1: INPUT "L="; l
270 LOCATE 25,2: INPUT "C="; c
280 c=c*10^-6: fo=1/(2*PI*SQR(1*c))
290 ro=2*SQR(1/c)
300 LOCATE 25,4: PRINT "Fo="fo
310 LOCATE 25,5: INPUT "f="; f
320 LOCATE 25,7: PRINT "Ro="ro
330 LOCATE 25,8: INPUT "r="; r
340 fr=f/fo: rr=r/ro
350 GOSUB 1000
360 LOCATE 25,10: PRINT "u="u
370 LOCATE 38,12: PRINT CHR$(243)
380 TAG: MOVE x-4,y+4: PRINT CHR$(203);
390 q=INKEY(1): IF q<>0 THEN 390
400 MOVE x-4,y+4: PRINT " "; TAGOFF
410 FOR a=25 TO 40
420 FOR b=1 TO 11
430 LOCATE a,b: PRINT " "
440 NEXT b: NEXT a
450 GOTO 260
470 REM
990 REM
1000 REM
1010 u=1/SQR(4*fr*fr*rr*rr+(1-fr*fr)^2)
1020 x=-20+(90-15*rr)+240*fr
1030 y=((60*u)+(-30*fr)+(90-15*rr))
1040 RETURN

```

$u$   
 $r$   $f$




Fig. 2


$$\begin{matrix} u \\ r \end{matrix} \begin{matrix} f \end{matrix}$$


Fig. 3



$$\begin{matrix} u \\ r \end{matrix} \begin{matrix} f \end{matrix}$$


Fig. 4

## 5. Filtrage RC Simple ou double alternance

A partir d'une tension sinusoïdale  $V_e$  issue par exemple du secondaire d'un transformateur dont le primaire est alimenté par le secteur, on désire obtenir en sortie une tension  $V_s$  unidirectionnelle, on dit "continue", si possible avec une faible ondulation.



La loi des nœuds en A donne

$$i = i_R + i_c \quad i = \frac{V_s}{R} + C \frac{dV_s}{dt}$$

1<sup>er</sup> cas : Si  $V_e > V_s$ , la diode conduit. Alors  $V_s = V_e$

2<sup>e</sup> cas : Si  $V_e < V_s$ , la diode est bloquée.  $i = 0$

$$\frac{V_s}{R} + C \frac{dV_s}{dt} = 0 \quad dV_s = - \frac{1}{RC} V_s dt$$

## Utilisation du programme

Entrer les données R,C,FQ. Tracer les courbes  $V_e$ ,  $V_s$ . Mettre ainsi en évidence le rôle des entrées sur la forme de  $V_s$ .

*Représentation graphique* : on veut représenter deux périodes de  $V_e$ .  $2T$  doit correspondre au 640<sup>e</sup> point. Si l'on donne à  $t$  des accroissements  $dt$ , chaque accroissement fera passer d'un point à un autre.

Donc  $dt \cdot 640 = 2T$  d'où  $dt = \frac{2T}{640}$  ou encore  $dt = \frac{2}{640 \cdot FQ}$  puisque  $FQ = \frac{1}{T}$ . Ce calcul est fait dans les premières instructions de la ligne 110.


## Déroulement du programme

lignes 10 à 70 entrées des données. A l'instant  $t = 0$  où l'on met sous tension le redresseur, la tension d'entrée  $V_e$  n'est pas nécessairement nulle. Autrement dit,  $V_e = V_s \sin(Wt + PH)$  avec PH : phase à l'origine.

Au moment où la diode cesse de conduire (instant  $t_1$ , point A) le condensateur se décharge dans R suivant une loi exponentielle et  $V_s$  est représentée par la portion AB. Au point A, la pente de la sinusoïde et la pente de l'exponentielle sont égales. Au contraire, dans la partie AB, la tangente à la sinusoïde a une pente plus grande que celle de l'exponentielle. On se servira de cette remarque dans le programme (lignes 200 à 230).

ligne 80

3 tableaux de 640 valeurs A, B, C



ligne 100

on forme 640 valeurs de  $t$  puisque  $t = 2 \cdot dt \cdot F$  et que  $F$  varie de 1 à 640. On calcule 640 valeurs de  $180 \cdot \sin(2 \cdot F \cdot t \cdot \text{PH})$  que l'on place dans le tableau A. On pourra ainsi les extraire du tableau pour dessiner la sinusoïde sur deux périodes.

ligne 140

initialisation de  $V_s$ ,  $dV_s$ ,  $i$  à zéro.

ligne 160

On calcule l'accroissement  $D = A(F+1) - A(F)$  qui représente  $dV_e$ .

ligne 170

$dV_s = V_s \frac{dt}{RC}$ . Le condensateur se décharge

dans R. La diminution de tension qui en résulte est telle que :

$$i = C \frac{(-dV_s)}{dt} \text{ ou } \frac{V_s}{R} = -C \frac{dV_s}{dt}$$

$$\text{d'où } dV_s = \frac{1}{RC} \cdot V_s dt$$

ligne 180 on suppose que la diode conduit d'où la condition  $i > 0$  et que l'augmentation de  $V_e$  est supérieure à la diminution de  $V_s$ . Alors  $V_s = A(F) = V_e$ . C'est la partie OA qui sera décrite, confondue avec  $V_e$ .

ligne 190 à partir de  $t_1$ , la tension aux bornes de C baisse et  $V_s$  (nouvelle valeur) =  $V_s$  (ancienne valeur) +  $dV_s$ .

ligne 200 on compare  $V_s$  nouvelle valeur à  $V_e$   
Si  $V_e > V_s$ ,  $V_s = A(F)$

lignes 210 on veut tracer  $i(t)$  dans la diode d'où le calcul de  

$$i = \frac{V_s}{R} + \frac{CdV_s}{dt}$$

ligne 230 si  $i > 0$  faire  $i = 0$

ligne 240 tableau de 640 valeurs de  $V_s = B(F)$   
tableau de 640 valeurs de  $i = C(F)$   
Le reste est le tracé proprement dit des courbes.  
On pourra essayer  $FQ = 1000$   $R = 500$   
 $C = 10E - 6$ ,  $PH = \frac{\pi}{3}$ , par exemple, ou d'autres valeurs. En particulier, faire varier uniquement C (ou R) en croissant. La courbe tend vers l'horizontale.

### Remarques :

Dans le cas du redressement monalternance, le lecteur pourra rentrer ligne 120 :  $A(u) = 180\sin(2\pi \cdot FQ \cdot t + PH)$ .

Dans le cas d'un redressement bialternance, réécrire la ligne 120 avec :  $A(u) = ABS(180\sin(2\pi \cdot FQ \cdot t + PH))$ .

Le lecteur pourra aussi faire apparaître les temps  $t_1$  et  $t_2$  (intervalle de temps correspondant à la décharge du condensateur dans la résistance R).

```

10 REM FILTRAGE RC
20 REM *****
30 REM
35 CLS
40 INPUT "VALEUR DE LA FREQUENCE FQ=";FQ
50 INPUT "VALEUR DE LA RESISTANCE R=";R
60 INPUT "VALEUR DE LA CAPACITE C=";C
70 INPUT "VALEUR DU DEPHASAGE PH=";PH



```






```

80 DIM A(640):DIM B(640):DIM C(640)
90 LOCATE 12,12:PRINT "CALCUL EN COURS"
100 FOR F=1 TO 640
110 DT=1/(FQ*640):T=2*DT*F
120 A(F)=ABS(180*SIN(2*PI*T*FQ+PH))
130 NEXT F
140 VS=0:DVS=0:I=0
150 FOR F=1 TO 640
160 D=A(F+1)-A(F):VE=A(F)
170 DVS=-VS*DT/(R*C)
180 IF ABS(D)>=DVS AND I>0 THEN VS=A(F):GOTO 200
190 VS=VS+DVS
200 IF VE>VS THEN VS=A(F)
210 I=(VS/R)+(D*C/DT)
220 IF I>M THEN M=I
230 IF I<0 OR VS>VE THEN I=0
240 B(F)=VS:C(F)=I
250 IF F=639 THEN GOTO 270
260 NEXT F
270 CLS
280 MODE 2
290 PLOT 0,200:DRAWR 640,0
300 PLOT 0,0:DRAWR 0,400
310 LOCATE 1,1:PRINT "VE,VS,I"
320 LOCATE 30,1:PRINT "FILTRAGE RC"
330 LOCATE 80,12:PRINT "T"
340 FOR F=1 TO 640
350 VE=A(F):PLOT F,A(F)+200
360 NEXT F
370 FOR F=1 TO 640
380 VS=B(F):PLOT F,VS+200
390 NEXT F
400 FOR F=1 TO 640
410 I=C(F):PLOT F,I/M*180
420 NEXT F
430 LOCATE 60,20:PRINT "FQ=";FQ;"HZ"
440 LOCATE 60,21:PRINT "R=";R;"OHM"
450 LOCATE 60,22:PRINT "C=";C;"FAR."
460 LOCATE 60,23:PRINT "PH=";PH;"RAD."

```







# 3

## Optique

### **1. Lentilles minces**

Les instruments d'optique classiques permettent d'obtenir des images par réfraction. Ils sont constitués de lentilles. Une lentille est un milieu tel que le verre, transparent à la lumière, limité par deux calottes sphériques. On définit son axe optique. Nous nous limiterons à l'étude des lentilles minces.

#### **Programme 1**

Il permet de générer l'image d'un objet, et de visualiser la construction géométrique qui y conduit. Il faut fournir au programme, par référence au bord gauche de l'écran

- la position horizontale de l'objet, de hauteur 70 pixels, appelée P ;
- la position horizontale de la lentille, de hauteur 140 pixels (x 2), appelée F ;
- la distance focale de la lentille, appelée DF.

Toutes ces grandeurs seront exprimées en pixels et pourront donc être imaginées en mètres, centimètres, etc.

Rappelons qu'il suffit de tracer deux rayons particuliers pour déterminer la position de l'image A' B' :

- l'un, partant du point B (voir dessin) et parallèle à l'axe optique, passe après traversée de la lentille par le foyer F' ;
- l'autre, partant de B également, traversant la lentille en son centre, n'est pas dévié.

Il a fallu tester la position de l'objet en ligne 230. En effet, suivant la valeur de celle-ci, l'image peut être devant ou derrière l'objet. Un "drapeau" vaut donc  $+1$  ou  $-1$  suivant le cas, ce qui détermine la direction dans laquelle on trace le rayon. Celui-ci est construit à l'aide de la tangente de l'angle qu'elle fait avec l'horizontale, soit  $70/DF$  à droite de la lentille,  $70(F - P)$  à gauche.

Plutôt que de donner une longue explication du programme, nous préférons donner un dessin sur lequel est porté, pour chaque segment de droite, le numéro de la ligne de programme qui le construit.

## Programme 2

Il réalise la même construction dans le cas de lentilles divergentes.

## Programme 3

Il ne génère pas de construction géométrique. Il se contente d'exploiter les relations mathématiques classiques sur les lentilles. On peut ainsi déterminer, connaissant la position et la grandeur de l'objet, la position et la grandeur de l'image.

Les exemples proposés traitent les différents cas les plus classiques.

### Programme 1 : Lentille convergente

- |        |  |
|--------|--|
| 1 et 2 | objet à plus de deux fois la distance focale $F$               |
| 3      | objet à une distance comprise entre $2F$ et $F$ de la lentille |
| 4      | objet à la distance $F$ de la lentille                         |
| 5 et 6 | objet à une distance inférieure à $F$ de la lentille           |

### Programme 2 : lentille divergente

- |   |   |
|---|---|
| 7 | objet à une distance supérieure de $F$ de la lentille |
| 8 | objet à une distance inférieure à $F$ de la lentille  |

### Programme 3 : cas général

9 et 10	identique à 1 et 2
11	identique à 3
12	identique à 5
13	identique à 6
14	identique à 7
15	identique à 8

#### Remarques sur la construction des rayons :

1° Un rayon parallèle à l'axe émerge de la lentille en direction du foyer. La tangente de l'angle qu'il fait avec l'horizontale est égale à  $\frac{\text{longueur de l'objet}}{\text{distance focale}}$

soit  $70/DF$  d'après nos notations. Un déplacement horizontal de 1 pixel est donc lié à un déplacement vertical de  $70/DF$  pixels (ligne 240, programme 1).


2° De même, pour un rayon passant par le centre optique, la tangente de l'angle qu'il fait avec l'horizontale est égale à

$$\frac{\text{longueur de l'objet}}{\text{distance objet/lentille}} = \frac{70}{F - P}$$

d'où la construction (ligne 240).

3° L'intersection des deux rayons est trouvée lorsque le second rayon tracé rencontre un point de couleur (ligne 330). L'image est alors tracée.

**Correspondance entre système électrique et système mécanique.**



- en gras : grandeurs géométriques,
- en maigre : numéros des lignes.




```



10 REM *****
20 REM *
30 REM * lentilles *
40 REM *
50 REM *****
60 REM
70 INPUT"position de la lentille";f
80 INPUT"distance focale";df
90 INPUT"position de l'objet";p
100 MODE 2:CLS
110 PLOT 0,180:DRAW 500,180
120 PLOT f,40:DRAW f-6,46

```



distance focale= 40





```


10 REM *****
20 REM *      *
30 REM *  lentilles *
40 REM * divergentes*
50 REM *      *
60 REM *****
70 REM
80 INPUT"position de la lentille";f
90 INPUT"distance focale";df
100 INPUT"position de l'objet";p
110 MODE 2:CLS
120 PLOT 0,180:DRAW 500,180
130 PLOT f,40:DRAW f-6,34
140 PLOT f,180:DRAW f,40
150 PLOT f,40 :DRAW f+6,34
160 PLOT f,180 :DRAW f,320


```

```

170 PLOT f,320:DRAW f-6,326
180 PLOT f,320 :DRAW f+6,326
190 x=f-df:y=180:GOSUB 360:GOSUB 400
200 x=f+df:y=180 :GOSUB 360
210 PLOT p,180 :DRAW p,180+70
220 PLOT p,180+70 :DRAW f,180+70
230 a=a+1/2:b=b-70/df/2
240 IF a-df+f>600 OR 180-b>350 THEN 280
250 IF a-df+f <0 OR 180-b <0 THEN 280
260 PLOT a-df+f,180-b : GOTO 230
270 REM
280 a=1:b=35/(f-p):PLOT p,180+70
290 a=a+1/2:b=b+70 /2/(f-p)
300 IF a+p>600 OR 180+70-b >350 THEN 330
310 IF TEST(a+p,180+70-b)=1 THEN PLOT a+p,180+ 70-
b:DRAW a+p,180
320 GOTO 290
330 PLOT p,180+70:DRAW a+p,180+70-b
340 STOP
350 STOP
360 REM croix
370 PLOT x-4,y-4:DRAW x+4,y+4
380 PLOT x+4,y-4:DRAW x-4,y+4
390 RETURN
400 LOCATE (f/8)-2,20:PRINT f
410 LOCATE (p/8)-2,20:PRINT p
420 LOCATE 1,22:PRINT"distance focale=";df
430 RETURN

```





```

10 REM *****
20 REM *           *
30 REM * lentilles *
40 REM *           *
50 REM *****
60 REM
70 CLS
80 INPUT "distance focale :";f
90 PRINT " distance de l'objet a la lentille :"
100 INPUT " ( - a gauche , + a droite )";p
110 INPUT " grandeur de l'objet :";o
120 PRINT " lentille convergente (1) , divergente
(2) "
130 INPUT a$
140 IF a$ ="1" THEN PRINT"lentille convergente ":
GOTO 170
150 IF a$ ="2" THEN PRINT"lentille divergente":f=
-f: GOTO 170
160 GOTO 140
170 p1=1/(1/f+1/p)
180 i=p1*o/p
190 PRINT"la grandeur de l'image est :";PRINT USI
NG "###.##";i
200 PRINT"distance de l'image a la lentille :";
210 PRINT USING "##.##";p1
220 STOP

```

distance focale : 40  
distance de l'objet a la lentille : distance de  
l'objet a la lentille :-100  
grandeur de l'objet : 70  
lentille convergente (1) , divergente (2)  
lentille convergente  
la grandeur de l'image est :-46.67  
distance de l'image a la lentille :66.67

distance focale : 40  
distance de l'objet a la lentille :-200  
grandeur de l'objet : 70  
lentille convergente (1) , divergente (2)  
lentille convergente  
la grandeur de l'image est :-17.50  
distance de l'image a la lentille :50.00

distance focale : 40  
distance de l'objet a la lentille :-60  
grandeur de l'objet : 70  
lentille convergente (1) , divergente (2)  
lentille convergente  
la grandeur de l'image est :%-140.00  
distance de l'image a la lentille :%120.00

distance focale : 70  
distance de l'objet a la lentille :-40  
grandeur de l'objet : 70  
lentille convergente (1) , divergente (2)  
lentille convergente  
la grandeur de l'image est :163.33  
distance de l'image a la lentille :%-93.33

distance focale : 70  
distance de l'objet a la lentille :-35  
grandeur de l'objet : 70  
lentille convergente (1) , divergente (2)  
lentille convergente  
la grandeur de l'image est :140.00  
distance de l'image a la lentille :%-70.00

```
distance focale : 70
distance de l'objet a la lentille :-100
grandeur de l'objet : 70
lentille convergente (1) , divergente (2)
lentille divergente
la grandeur de l'image est : 28.82
distance de l'image a la lentille :%-41.18
```

```
distance focale : 70
distance de l'objet a la lentille :-50
grandeur de l'objet : 70
lentille convergente (1) , divergente (2)
lentille divergente
la grandeur de l'image est : 40.83
distance de l'image a la lentille :%-29.17
```

## **2. Lentille**

Quel est le chemin parcouru par la lumière quand elle traverse une lentille ? Lentille d'objectif photographique ou d'un lecteur de disque laser ou simple loupe... Quelle image donne la lentille d'un objet ponctuel ? Ce programme vous montre la marche des rayons lumineux et vous montre dans quelles conditions on peut espérer une image. C'est l'utilisateur qui détermine les caractéristiques de la lentille et la position de l'objet.

### **Pourquoi ce programme ?**

L'ordinateur permet ici d'aborder un domaine de la physique de façon entièrement nouvelle : c'est un compagnon qui vous libère des calculs mais vous laisse la responsabilité de la décision finale.

Les cours de physique traitent en détail les lentilles minces qui sont peu utilisées. L'étude des lentilles épaisses pose en effet deux problèmes :

— la seule loi optique utilisée est la loi de réfraction de Descartes, mais les problèmes géométriques conduisent à des calculs complexes,

— l'image que la lentille fournit de l'objet n'est pas nette, elle est mal définie. Aussi, il est impossible de la localiser avec précision, donc impossible de donner une formule pour calculer sa position.

L'ordinateur vient aisément à bout de la première difficulté. Pour la seconde, il fournit une figure sur laquelle il vous laisse décider si l'image obtenue est suffisamment nette pour vos besoins et apprécier sa position.

## Utilisation du programme

La lentille est placée au centre de l'écran, dans un milieu d'indice 1, le diamètre de la lentille doit être égal ou supérieur à la hauteur de l'écran. L'objet peut se situer en dehors de l'écran mais il doit être réel. Si l'objet est à l'intérieur de l'écran, il est possible qu'il faille donner au verre un indice très élevé pour obtenir une image dans les limites de l'écran.

Les rayons réfléchis par les faces de la lentille ne sont pas tracés. Dans la réalité ils peuvent donner naissance à des images parasites.


Dans le cas de rayons extrêmement inclinés sur l'axe, l'ordinateur peut fournir des images fausses car il n'envisage pas deux rencontres successives du même rayon avec la même face de la lentille.

Pour faire avancer le programme, taper →

Pour sortir du programme, faire ESC ESC.

## Le cœur du programme

Rappelons tout d'abord la loi de Descartes (fig. 1) :





La figure 2 montre l'entrée dans la lentille et les notations utilisées dans le programme.

On a utilisé les relations

$$I = E + C \text{ (ligne 505)}$$

$$C1 = R - E \text{ (ligne 520)}$$

avec  $\tan C = c$ ,  $\tan E = te$ ,  $\tan I = ti$ ,  $\tan R = tr$ ,  $\sin R = sr$




Fig. 2. - Le rayon entre dans la lentille

La figure 3 montre la sortie de la lentille, on a utilisé les relations  $I = S - C1$  (ligne 610) et  $C2 = S - R$  (ligne 630).




Fig. 3. — Le rayon sort de la lentille

## Liste des variables

“e”	en deuxième lettre désigne l’entrée de la lentille
“s”	en deuxième lettre désigne la sortie de la lentille
v	indice du verre
e	épaisseur de la lentille
de,ds	type de dioptre
re,rs	rayon du dioptre
x,y	position du point sur le rayon lumineux
xo,yo	position de l’objet
a	angle variable pour le dessin de la lentille
pe, ps	position des faces de la lentille
xe,ye,xs,ys	position du point pour tracer la lentille
c	pente du rayon avant de pénétrer dans la lentille
ze di	distance du point au centre du dioptre d’entrée
m,n	position du point où le rayon entre dans la lentille
te,ti,i,sr,tr	voir figures 2 et 3
c1	pente du rayon pendant la traversée de la lentille
zs	distance du point au centre du dioptre de sortie
mm,nn	position du point où le rayon sort de la lentille
c2	pente du rayon après la sortie de la lentille

## Déroulement du programme

10	présentation
100	introduction des données
250	dessin du premier dioptre
330	dessin du second dioptre
420	départ du rayon
430	avant d’arriver à la lentille
450	est-on arrivé à la lentille ?
470	x avance d’une unité
490	entrée dans la lentille
500	calcul de c1
530	propagation à l’intérieur de la lentille
550	est-on arrivé à la sortie de la lentille ?
570	x avance d’une unité
590	sortie de la lentille
600	calcul de c2
640	après la lentille
650	arrivée en bout d’écran

660	x avance d'une unité
700	choix d'un nouveau rayon, retour à 430
710	le dernier rayon vient d'être tracé
800	conclusion

## Commentaire des illustrations

Fig. 4 : l'objet et l'image sont en dehors de l'écran.

Fig. 5 : obtenue avec dans l'ordre : 3, 200, 1, 400, 1, 400, 0, 0.  
L'indice du verre étant plus élevé que précédemment, on peut observer l'image et l'objet sur l'écran car la lentille est plus convergente. L'image n'est pas ponctuelle car les rayons à la sortie de la lentille ne se recoupent pas au même point.

Fig. 6 : même lentille, mais l'objet est en 0, 100.

Fig. 7 : avec 2, 100, 1, 600, -1, 240, 0, 100

La lentille est divergente.

Fig. 8 : avec 1,5, 60, -1, 700, -1, 700, 150, 100.

Dans les deux derniers cas, vous pouvez prolonger vers la gauche les rayons sortant de la lentille pour rechercher l'image virtuelle de l'objet.

```

10 CLS
20 LOCATE 14,8:PRINT"Lentilles"
30 LOCATE 14,10:PRINT"epaisses"
40 LOCATE 1,22:PRINT"pour faire avancer le program
me, appuyer sur"CHR$(243)
50 LOCATE 35,25:PRINT CHR$(243)
60 REM
70 REM
80 q=INKEY(1):IF q<>0 THEN 80
100 PRINT:INPUT"indice du verre=";v
110 PRINT:INPUT"epaisseur de la lentille";e
120 PRINT:PRINT"premier dioptre"
130 INPUT"concave:-1,convexe:1";de
140 INPUT"rayon (>200) =" ;re
150 PRINT:PRINT"deuxieme dioptre"
160 INPUT"concave:-1,convexe:1";ds
170 INPUT"rayon (>200) =" ;rs
180 PRINT:PRINT"position de l'objet"
190 INPUT"x=";xo:INPUT"y=";yo
200 LOCATE 39,25:PRINT CHR$(243)
210 q=INKEY(1):IF q<>0 THEN 210

```

```

220 REM
240 CLS
250 MOVE 0,200: DRAW 640,200
260 pe=320-e/2: ps=320+e/2
270 a=0
280 xe=pe+de*re*(1-COS(a))
290 ye=re*SIN(a)
300 IF ye>200 THEN GOTO 330
310 PLOT xe,200+ye: PLOT xe,200-ye
320 a=a+0.005: GOTO 280
325 REM
330 a=0
340 xs=ps-ds*rs*(1-COS(a))
350 ys=rs*SIN(a)
360 IF ys>200 THEN GOTO 390
370 PLOT xs,200+ys: PLOT xs,200-ys
380 a=a+0.005: GOTO 340
385 REM
390 IF xs<xe THEN CLS: GOTO 100
400 MOVE xe,399: DRAW xs,399 : MOVE xe,0: DRAW xs,0

410 REM
420 x=xo: c=(160-yo)/(pe-xo)
430 y=yo+c*(x-xo)
440 IF ABS(y)<200 THEN PLOT x,200+y ELSE GOTO 700

450 ze=de*((x-pe-de*re)^2+y^2)
460 IF ze<de*(re^2) THEN GOTO 490
470 x=x+1: GOTO 430
480 REM
490 m=x: n=y
500 te=n/(pe+de*re-m)
505 ti=(te+c)/(1-te*c) : i=ATN(ti)
510 sr=SIN(i)/v: tr=sr/SQR(1-sr^2)
520 c1=(tr-te)/(1+tr*te)
530 y=n+c1*(x-m)
540 IF ABS(y)<200 THEN PLOT x,200+y ELSE GOTO 700
550 zs=ds*((x-ps+ds*rs)^2+y^2)
560 IF zs>ds*(rs^2) THEN GOTO 590
570 x=x+1: GOTO 530
580 REM
590 mm=x: nn=y
600 ts=nn/(mm-ps+ds*rs)
610 ti=(ts-c1)/(1+ts*c1): i=ATN(ti)
620 sr=SIN(i)*v: IF ABS(sr)>1 THEN GOTO 700
625 tr=sr/SQR(1-sr^2)
630 c2=(ts-tr)/(1+tr*ts)
640 y=nn+c2*(x-mm)
650 IF ABS(y)<200 THEN PLOT x,200+y ELSE GOTO 700

```

```

660 x=x+1:IF x<640 THEN GOTO 640
700 c=c-0.2
710 IF c<(-160-yo)/(pe-xo) THEN GOTO 800
720 x=xo:GOTO 430
790 REM
800 TAG:MOVE 620,20:PRINT CHR$(243);:TAGOFF
810 q=INKEY(1):IF q<>0 THEN 810
820 CLS
830 PRINT:PRINT"meme lentille? "CHR$(243)",sinon E
SC ESC"
840 q=INKEY(1):IF q<>0 THEN 840
850 GOTO 180
860 END

```

**pour faire avancer le programme, appuyer sur→**

**indice du verre=? 1.5**  
**epaisseur de la lentille? 200**  
**premier dioptre**  
**concave:-1,convexe:1? 1**  
**rayon (>200)=? 400**  
**deuxieme dioptre**  
**concave:-1,convexe:1? 1**  
**rayon (>200)=? 400**  
**position de l'objet**  
**x=? -320**  
**y=? 0**




Fig. 4




Fig. 5




Fig. 6




Fig. 7




Fig. 8

### 3. Dioptre plan

Ce programme permet de visualiser le trajet d'un rayon lumineux lorsque celui-ci rencontre un dioptre plan. Nous ne donnerons pas la démonstration mathématique complexe qui conduit à l'algorithme employé. Seul compte le résultat !

Ce programme est largement modifiable et peut donc être adapté à la mise en évidence de phénomènes particuliers.

La position du dioptre plan est imposée ligne 110 ( $X = 200$ ) ; la position du point lumineux intervient ligne 150 ( $X = 100$ ) et ligne 160 (où le nombre 100 représente la distance entre ce point et le dioptre).  $T$  est l'angle de "départ" du rayon, en radians. Les valeurs prises par  $T$  sont imposées ligne 140 et 290.

```


10 REM *****
20 REM *
30 REM *      dioptre plan      *
40 REM *
50 REM *****
60 REM
70 MODE 2 : CLS
80 INPUT "n1=";n1
90 INPUT "n2=";n2

```


```

100 CLS:a0=n1/n2
110 PLOT 200,0:DRAW 200,399
120 LOCATE 1,1:PRINT"n1=";n1
130 LOCATE 42,1:PRINT"n2=";n2
140 FOR t=-PI/4 TO(PI/4+PI/32)STEP PI/16
150 PLOT 100,200:s=-COS(t)
160 DRAW 200,200+100*TAN(t)
170 c=n2^2 -n1^2 +(n1*s)^2
180 IF c<0 THEN a=-1:b=2*COS(t):GOTO 200
190 a=a0 :b=s*(a0-(c/(n2*s)^2)^0.5)
200 d=SIN(t)/(b-a*s)
210 x=200:y=200+100*TAN(t)
220 x=x+SGN(c):y=y+d
230 IF x >639 OR x<0 OR y>399 OR y<0 THEN 250
240 DRAW x,y:GOTO 220
250 IF c<0 THEN 290
260 x=x-1:y=y-d
270 IF x >639 OR x<0 OR y>399 OR y < 0 THEN 290
280 DRAW x,y : GOTO 260
290 NEXT t
300 STOP

```










# 4

## Chimie

### 1. Propriétés d'une solution chimique

Ce programme propose de calculer le pH d'une solution aqueuse, connaissant le pKa du couple acide-base correspondant, ainsi que sa molarité.

Un couple acide-base, représenté par la réaction :




est défini par son Ka :

$$K_a = \frac{(B) (H_3O^+)}{(A)}$$

on a aussi  $pK_a = -\log K_a$ .

Pour une solution d'acide ou de base, on peut écrire (pour plus de commodité, on pourra prendre l'exemple de l'acide acétique) :



On a alors (A) = (CH<sub>3</sub>CO<sub>2</sub>H) et (B) = (CH<sub>3</sub>CO<sub>2</sub><sup>-</sup>)

$$(B) + (OH^-) = (H_3O^+)$$

$$(A) + (B) = C \text{ (molarité initiale)}$$

$$(H_3O^+).(OH^-) = 10^{-14}$$

$$K_a = \frac{(B) (H_3O^+)}{(A)}$$

On en déduit donc :

$$(OH^-) = \frac{10^{-14}}{(H_3O^+)}$$

$$(B) = (H_3O^+) - (OH^-)$$

$$(B) = (H_3O^+) - \frac{10^{-14}}{(H_3O^+)}$$

$$(A) = C - (B) = C - (H_3O^+) - \frac{10^{-14}}{(H_3O^+)}$$

$$\text{d'où } K_a = \frac{\left[ (H_3O^+) - \frac{10^{-14}}{(H_3O^+)} \right] \cdot (H_3O^+)}{C - \left[ (H_3O^+) - \frac{10^{-14}}{(H_3O^+)} \right]}$$

Cette équation relie le  $pH = -\log (H_3O^+)$  à la molarité de la solution. Elle dépend de  $K_a$ .

il est donc possible de l'utiliser des quatre manières suivantes :

— Tracer des courbes universelles (graphique de Flood) donnant le pH en fonction de C, avec comme paramètre le  $pK_a$ .

— Calculer le pH d'une solution connaissant le  $pK_a$  et la molarité C (3).

— Calculer la molarité C, connaissant le  $pK_a$  et le pH de la solution (1).

— Calculer le  $pK_a$  du couple, connaissant le pH de la solution et sa concentration (2).

Ces trois dernières possibilités seront regroupées dans le programme.

## Liste des variables

P	pH
H	$(H_3O^+)$
K	$pK_a$
KA	$K_a$
C	concentration
PK	$pK_a$
PA	$pK_a$

## Détail du programme

Il permet d'étudier une solution particulière :

80 - 170 *Premier cas* :  $pK_a$  et pH connus. C est obtenue par application directe de la relation de base.

200 - 290 *Deuxième cas* : C et pH connus. Partant de  $pK_a = 0,1$ , toutes les valeurs de  $pK_a$  sont examinées jusqu'à ce que la formule de base donne une valeur de C presque identique à celle qui est donnée.

450 - 620 *Troisième cas* :  $pK_a$  et C connus. Partant de  $pH = 1/2 pK_a$ , toutes les valeurs de C données par la formule sont examinées. Le cas d'une base est traité comme celui d'un acide en utilisant son  $pK_b$ . La valeur finale du pH est alors  $14 - (pH \text{ trouvé})$ .

630 - 640 Calcul de la concentration  $K_e = 10^{-14}$ .

630 Formule de base, donnant la concentration C en fonction de  $(H^+)$  et de  $K_a$ .

```
10 REM *****
20 REM *
30 REM * SOLUTION CHIMIQUE *
40 REM *
50 REM *****
60 REM
70 KE=10^(-14)
80 CLS
90 PRINT
100 PRINT " faites votre choix "
110 PRINT " connaissant pKa et pH , vous voulez c
(1)
120 PRINT " connaissant c et pH , vous voulez pKa
(2)
130 PRINT " connaissant pKa et c, vous voulez pH (
3)
140 PRINT
150 PRINT"tapez 1,2 ou 3"
160 INPUT n
170 PRINT"choix numero:";n
180 IF n=1 OR n=2 OR n=3 THEN 190 ELSE 150
190 ON n GOSUB 200,300,460
200 PRINT:PRINT:PRINT
210 INPUT "ph=";p:h=10^(-p)
220 PRINT:PRINT:PRINT
230 INPUT"pka=";k:ka=10^(-k)
240 IF p>7 THEN h=ke/h:ka=ke/ka
250 GOSUB 630:PRINT:PRINT:c=f
260 PRINT" la concentration est c=";
```

```

270 PRINT USING"#.####";c;:PRINT" mole/litre"
280 z$=INKEY$:IF z$="" THEN GOTO 280
290 GOTO 10
300 PRINT:PRINT:PRINT
310 INPUT "concentration=";c
320 PRINT:INPUT"ph=";p:h=10^(-p)
330 IF p>7 THEN h=ke/h
340 pk=0.1
350 pk=pk+0.1:ka=10^(-pk):GOSUB 630
360 IF f>c THEN 370 ELSE 350
370 pk=pk-0.1
380 pk=pk+0.01:ka=10^(-pk):GOSUB 630
390 IF f>c THEN 400 ELSE 380
400 IF p >7 THEN ka=ke/ka
410 PRINT:PRINT"le pka est:";
420 PRINT INT(100*(-LOG(ka))/2.3)/100
430 z$=INKEY$:IF z$="" THEN GOTO 430
440 GOTO 10
450 PRINT:PRINT:PRINT
460 INPUT"pka=";pa:ka=10^(-pa)
470 PRINT:INPUT"concentration=";c
480 PRINT:PRINT"solution acide ou basique?"
490 INPUT a$
500 IF a$="a"THEN PRINT"solution acide"
510 IF a$="b"THEN ka=ke/ka:PRINT"solution basique"

520 p=0.5*pa:IF a$="b" THEN p=0.5*(14-pa)
530 p=p+0.1:h=10^(-p):GOSUB 630
540 IF f< c THEN 550 ELSE 530
550 p=p-0.1
560 p=p+0.01:h=10^(-p):GOSUB 630
570 IF f < c THEN 580 ELSE 560
580 IF a$="b" THEN p=14-p
590 PRINT:PRINT
600 PRINT" le pH est ";INT(p*100)/100
610 z$=INKEY$:IF z$="" THEN GOTO 430
620 GOTO 10
630 f=(h-(ke/h))*((h/ka)+1)
640 RETURN

```

faites votre choix

Connaissant pKa et pH , vous voulez c : (1)

Connaissant c et pH , vous voulez pKa : (2)

Connaissant pKa et c , vous voulez pH : (3)

tapez 1,2 ou 3

choix numero: 2

concentration= 0.01

ph= 11

le pka est: 10.05

faites votre choix

Connaissant pKa et pH , vous voulez c : (1)

Connaissant c et pH , vous voulez pKa : (2)

Connaissant pKa et c , vous voulez pH : (3)

tapez 1,2 ou 3

choix numero: 3

pka= 10

faites votre choix

Connaissant pKa et pH , vous voulez c : (1)

Connaissant c et pH , vous voulez pKa : (2)

Connaissant pKa et c , vous voulez pH : (3)

tapez 1,2 ou 3

choix numero: 3

pka= 10

solution acide ou basique ?

solution acide

le pH est 6.99

faites votre choix

Connaissant pKa et pH , vous voulez c : (1)

Connaissant c et pH , vous voulez pKa : (2)

Connaissant pKa et c , vous voulez pH : (3)

tapez 1,2 ou 3

choix numero: 3

pka= 8

solution acide ou basique ?

solution basique

le pH est 9.84

faites votre choix

Connaissant  $pK_a$  et  $pH$  , vous voulez  $c$  : (1)

Connaissant  $c$  et  $pH$  , vous voulez  $pK_a$  : (2)

Connaissant  $pK_a$  et  $c$  , vous voulez  $pH$  : (3)

tapez 1,2 ou 3

choix numero: 1

$pH = 4$

$pK_a = 6$

la concentration est  $c = 0.0101$  mol/litre

faites votre choix

Connaissant  $pK_a$  et  $pH$  , vous voulez  $c$  : (1)

Connaissant  $c$  et  $pH$  , vous voulez  $pK_a$  : (2)

Connaissant  $pK_a$  et  $c$  , vous voulez  $pH$  : (3)

tapez 1,2 ou 3

choix numero: 1

$pH = 10$

$pK_a = 7$

la concentration est  $c = 0.1001$  mol/litre

faites votre choix

Connaissant  $pK_a$  et  $pH$  , vous voulez  $c$  : (1)

Connaissant  $c$  et  $pH$  , vous voulez  $pK_a$  : (2)

Connaissant  $pK_a$  et  $c$  , vous voulez  $pH$  : (3)

tapez 1,2 ou 3

choix numero: 2

concentration= 0.01

$pH = 4$

le  $pK_a$  est: 6



## 2. Dosages volumétriques

### Introduction

Le dosage volumétrique d'une solution s'effectue en ajoutant m à un volume déterminé de cette solution un réactif convenable.

La fin de la réaction (équivalence) est le plus souvent marquée par le changement de coloration d'un indicateur coloré.

Le titre d'une solution est donné par sa *molarité* (nombre de moles présentes dans un litre de solution) ou par sa *normalité* (nombre de valences-grammes par litre de solution).


### Acidimétrie et alcalimétrie

— Une valence-gramme d'acide est équivalente à un ion-gramme  $\text{H}_3\text{O}^+$ .

— Une valence-gramme de base est équivalente à un ion-gramme  $\text{OH}^-$ .


— La normalité d'une solution acide est exprimée par le nombre d'ions-grammes  $\text{H}_3\text{O}^+$  présents dans un litre de solution.

Les valences-grammes des principaux acides sont :




— La normalité d'une solution basique est exprimée par le nombre d'ions-grammes  $\text{OH}^-$  présents dans un litre de solution.

Les valences-grammes des principales bases sont :



Soit  $v_1$  et  $v_2$ ,  $N_1$  et  $N_2$  les volumes et les normalités des solutions d'acide et de base qui ont réagi. A l'équivalence, c'est-à-dire pour le même nombre de valences-grammes d'acide et de base correspondant à la réaction complète.



## Equilibres ioniques

Les problèmes d'équilibres ioniques reviennent fréquemment à des calculs de concentrations ioniques. Le plus souvent, on calcule la concentration en ions  $\text{H}_3\text{O}^+$  qui se traduit par le PH ( $\text{PH} = -\log(\text{H}_3\text{O}^+)$ ).


Les difficultés du calcul résident dans l'établissement correct des relations dont on peut éventuellement disposer.

Intéressons-nous au cas des équilibres ioniques homogènes (la solution aqueuse est l'unique phase).

### 1° Equilibres présents en solution

Une solution aqueuse (solvant) contient une grande proportion de molécules d'eau dont on peut considérer la concentration comme constante.

Parmi les ions apportés par le ou les solutés figurent toujours des ions  $\text{H}_3\text{O}^+$  et  $\text{OH}^-$  en équilibre avec les molécules d'eau. Cet équilibre peut se traduire par l'équation suivante :




De plus, nous classerons les ions présents en solution en deux groupes : les ions "forts" et les ions "faibles".

— Un ion est dit "fort" lorsqu'il est sans action sur les autres ions ou molécules de la solution ; sa concentration est celle qui correspond à son introduction en solution.

— Un ion est dit "faible" s'il déplace l'équilibre précédent en créant un nouvel équilibre.

Ex. : L'ion acétate  $\text{CH}_3\text{COO}^-$  agit sur les ions  $\text{H}_3\text{O}^+$  présents en solution selon l'équilibre.



En présence de soude (NaOH) par exemple, les espèces présentes en solution sont :



### 2° Relations entre les concentrations ioniques et molaires

Ces relations utilisent les données : (volumes, concentrations, etc.). Elles peuvent être classées en deux groupes :


— *Relations linéaires :*

- Expression de la neutralité électrique.

On écrit que, dans un litre de solution, la somme des charges électriques (exprimées en électrons-grammes ou faradays) portées sur les cations est égale à celle portée par les anions. De cette manière, ce sont les concentrations ioniques qui interviennent dans la relation.


(*Attention :* certains ions peuvent éventuellement porter plusieurs charges).

Ex. : Pour une solution de sulfate ferrique ( $\text{Fe}_2(\text{SO}_4)_3$ ) :



- D'autres relations linéaires sont obtenues en exprimant que la solution contient ce qui a été introduit (Loi de conservation de la matière).

Ex. : Pour une solution d'acide acétique de concentration molaire  $C_1$




— *Relations non linéaires :*

Ce sont les relations obtenues par application de la loi-d'action de masse aux équilibres, y compris évidemment l'équilibre ionique de l'eau.



Ex. : Toujours pour l'acide acétique




On peut écrire :

$$K_a = \frac{(\text{CH}_3\text{COO}^-) (\text{H}_3\text{O}^+)}{(\text{CH}_3\text{COOH})} \quad \text{avec } K_a = 1.85 \cdot 10^{-5}$$

## Neutralisation d'un monoacide par une base

**1° Considérons l'acide HA de concentration molaire  $C_1$  en solution**

— Appliquons la loi d'action de masse à l'équilibre



*Remarque :* Pour simplifier l'écriture, nous écrirons dorénavant :



On peut écrire :

$$K_1 = \frac{(A^-)(H^+)}{(HA)} \text{ (si l'acide est fort (HA) } \rightarrow 0, \text{ donc } K_1 \rightarrow \infty)$$

La loi de conservation de la matière entraîne :

$$(HA) + (A^-) = C_1$$

$$\text{de plus } (H^+)(OH^-) = K_e \quad (K_e = 10^{-14})$$

Ecrivons l'électroneutralité de la solution acide :

$$(H^+) = (A^-) + (OH^-)$$


On montre facilement que :

$$C_1 = \left( (H^+) - \frac{K_e}{(H^+)} \right) \left( 1 + \frac{(H^+)}{K_1} \right)$$

$$\text{Remarque : } (H^+) = 10^{-PH}$$

## 2° Considérons la base BOH de concentration molaire $C_2$ en solution

— Appliquons la loi d'action de masse à l'équilibre



On peut écrire :

$$K_2 = \frac{(B^+)(OH^-)}{(BOH)} \text{ (si l'acide est fort (BOH) } \rightarrow 0 \text{ donc } K_2 \rightarrow \infty)$$

La loi de conservation de la matière entraîne

$$(BOH) + (B^+) = C_2$$

Ecrivons l'électroneutralité de la solution basique :

$$(B^+) + (H^+) = (OH^-)$$

On montre facilement que :

$$C_2 = \left( 1 + \frac{K_e}{K_2(H^+)} \right) \left( \frac{K_e}{(H^+)} - (H^+) \right)$$

## 3° Considérons le mélange BOH + HA

Ecrivons toujours les équations :

$$K_1 = \frac{(A^-)(H^+)}{(HA)} ; \quad K_2 = \frac{(B^+) \cdot (OH^-)}{(BOH)} ;$$

$$(H^+)(OH^-) = K_e \quad \text{et} \quad (H^+) + (B^+) = (OH^-) + (A^-)$$

Supposons que l'on soit parti d'un volume  $v_o$  de solution acide et que l'on ajoute un volume  $v$  de solution basique.

La loi de conservation de la matière conduit aux deux nouvelles équations

$$(HA) + (A^-) = \frac{C_1 v_o}{v_o + v} \quad \text{et} \quad (BOH) + (B^+) = \frac{C_2 v}{V_o + v}$$

On montre donc facilement que :

$$v = \frac{\frac{C_1}{1 + \frac{(H^+)}{K_1}} - \frac{K_e}{(H^+)} - (H^+)}{\frac{C_2}{1 + \frac{K}{K_2(H^+)}} - \frac{K_e}{(H^+)} + (H^+)}$$

## Liste des variables

K1	constante d'acidité
K2	constante de basicité
P1	PH de la solution acide de départ
P2	PH de la solution à la fin de la réaction
K	produit ionique de l'eau
VO	volume de la solution acide initiale
R	facteur d'échelle pris à 1 au départ
C1	concentration de la solution acide pur
C2	concentration de la solution de base pure
V1	volume de la solution correspondant à la neutralisation totale
V2	volume de solution basique ajoutée
B	PH du mélange

## Déroulement du programme

lignes 10 à 100      entrées des données

*Remarque :* le volume VO est pris égal à 50 cm<sup>3</sup>. Le facteur d'échelle (ligne 50) R est pris égal à 1 au départ (l'utilisateur peut éven-

	tuellement le changer selon le problème étudié).
lignes 110 à 150	on calcule les concentrations C1 et C2 de la solution acide et de la solution basique
lignes 160 à 210	tracé des axes de coordonnées
ligne 230	on fait varier le PH (B au clavier) de P1 jusqu'à P2
lignes 250 à 340	on calcule pour chaque valeur du PH le volume de base V ajouté correspondant
ligne 270	la courbe $PH = f(V)$ est tracée en continu
ligne 400	l'ordinateur imprime le volume V2 correspondant à la neutralisation totale ( $V2 = V1$ )
ligne 410	nous trouvons un point d'arrêt ; l'ordinateur interroge l'utilisateur : "Avez-vous observé la neutralisation totale ? »
ligne 460	entrée d'une variable chaîne A\$ correspondant à "OUI" ou "NON"
	si A\$ = "OUI", l'ordinateur imprime : "Expérience terminée"
	si A\$ = "NON", l'ordinateur imprime "Nouveau facteur d'échelle ?" L'utilisateur peut rentrer un nouveau facteur d'échelle R et l'on recommence.

```

10 REM COURBE PH=F(V)
20 REM *****
30 REM
40 REM INTRODUCTION DES DONNEES
50 REM *****
55 CLS
60 INPUT "CONSTANTE D'ACIDITE K1=";K1
70 INPUT "CONSTANTE DE BASICITE K2=";K2
80 INPUT "PH DE DEPART P1=";P1
90 INPUT "PH D'ARRIVEE P2=";P2
100 CLS
110 K=1E-14:V0=50:R=1
120 REM C1=CONCENT.DE L'ACIDE
130 C1=(10^(-P1)-K/10^(-P1))*(1+10^(-P1)/K1)
140 REM C2=CONCENT.DE LA BASE
150 C2=(1+K/(K2*10^(-P2)))*(K/10^(-P2)-10^(-P2))
160 V1=C1*V0/C2
170 MODE 2
180 PLOT 0,0:DRAW 0,400
190 PLOT 0,0:DRAW 640,0

```

```


200 LOCATE 2,3:PRINT "PH"
210 LOCATE 78,24:PRINT "V"
220 T=0:G=0
230 FOR B=P1 TO P2 STEP 0.05
240 H=10^(-B)
250 V=V0*(C1/(1+10^(-B)/K1)+K/10^(-B)-10^(-B))/(C2
/(1+K/(K2*10^(-B))))-K/10^(-B)+10^(-B))
260 IF G=1 THEN GOTO 300
270 IF V>=V1 THEN G=1:LOCATE 50,10:PRINT "TL NTR":
LOCATE 50,11:PRINT "V2=";V1:LOCATE 50,12:PRINT "PH
=";B
300 IF V/R>=300 THEN GOTO 400
310 IF T=0 THEN PLOT V/R,24*B:T=1:GOTO 330
320 DRAWR V/R-AV,24*B-AB
330 AV=INT(V/R+0.5):AB=INT(24*B+0.5)
340 NEXT B
400 LOCATE 1,1:PRINT "AVEZ VOUS OBSERVE LA NEUTRA
LISATION TOTALE?"
410 INPUT "A$=";A$
420 IF A$="OUI" THEN GOTO 500
430 IF A$="NON" THEN INPUT "NOUVEAU FACTEUR D'ECHE
LLE R=";R:GOTO 170
500 LOCATE 50,18:PRINT "VOLUME D'ACIDE V0="V0
510 LOCATE 50,19:PRINT "K1=";K1
520 LOCATE 50,20:PRINT "K2=";K2
530 LOCATE 50,21:PRINT "PH DE DEPART="P1
540 LOCATE 50,22:PRINT "PH ARRIVEE=";P2
550 LOCATE 50,23:PRINT "FACTEUR D'ECHELLE=";R

```

AVEZ VOUS OBSERVE LA NEUTRALISATION TOTALE?

A\$= OUI

PH




TL NTR  
V2= 58  
PH= 7.88888884

VOLUME D'ACIDE V0= 58  
K1= 100000000  
K2= 100000000  
PH DE DEPART= 3  
PH ARRIVEE= 11  
FACTEUR D'ECHELLE= 1

AVEZ VOUS OBSERVE LA NEUTRALISATION TOTALE?

AS=? OUI

PH



TL MTR  
V2= 99.7631163  
PH= 7.00000004


VOLUME D'ACIDE V0= 50  
K1= 100000000  
K2= 100000000  
PH DE DEPART= 2.7  
PH ARRIVEE= 11  
FACTEUR D'ECHELLE= 1

u

AVEZ VOUS OBSERVE LA NEUTRALISATION TOTALE?

AS=? OUI

PH



TL MTR  
V2= 28.6881631  
PH= 8.15000006


VOLUME D'ACIDE V0= 50  
K1= 0.0000185  
K2= 100000000  
PH DE DEPART= 3.5  
PH ARRIVEE= 12  
FACTEUR D'ECHELLE= 1

u

AVEZ VOUS OBSERVE LA NEUTRALISATION TOTALE?

AS=? OUI

PH



TL MTR  
V2= 44.8254042  
PH= 8.25000006

VOLUME D'ACIDE V0= 50  
K1= 0.0000185  
K2= 100000000  
PH DE DEPART= 3.4  
PH ARRIVEE= 12  
FACTEUR D'ECHELLE= 1

u



### 3. Dissociation d'une solution chimique

Considérons un acide AH, susceptible de se dissocier en solution :



Si l'on appelle :

H la concentration en ions  $\text{H}_3\text{O}^+$

KA la constante de dissociation

A le degré de dissociation

A0 la concentration de la solution

On démontre les relations suivantes :

$$(1) \quad A = \frac{KA}{KA + H}$$

$$(2) \quad AO = \left( H - \frac{K}{H} \right) / A \quad (K = 10^{-14})$$

Pour une solution donnée, si H et KA sont connus (le plus souvent par l'intermédiaire de pH et pKa), on peut grâce à (1) en déduire A, puis avec (2) en déduire A0. La solution étant alors parfaitement définie, il est possible d'en déduire la courbe  $A = f(\text{pH})$ . Le programme exploite donc directement ces deux relations.

Dans un premier temps, le programme évalue A0 à partir du pH de la solution ainsi que de son pKa.

Après appui sur une touche, le programme trace la courbe  $A = f(\text{pH})$  pour la solution donnée.

```

10 REM *****
20 REM *
30 REM * degre de dissociation *
40 REM * d'une solution *
50 REM *
60 REM *****
70 REM
80 CLS : MODE 2
90 k=1E-14
100 INPUT " ph de la solution:";ph
110 INPUT "pka de la solution:";pk
120 ka=10^(-pk):h=10^(-ph)
130 a=( ka/(ka+h))
140 a0=(h-k/h)/a


```

```


150 PRINT"la concentration est:";
160 PRINT USING"#.### " ;a0;:PRINT"    mol/1."
170 a$=INKEY$;IF a$="" THEN 170
180 PLOT 30,30:DRAW 600,0: LOCATE 75,22:PRINT"ph"
190 PLOT 30,30:DRAW 0,310: LOCATE 6,5: PRINT"a0"
200 LOCATE 39,22:PRINT"7"
210 FOR n=1 TO 2
220 PLOT 28,150*n+30: DRAW 4,0
230 NEXT n
240 LOCATE 6,14:PRINT"1/2"
250 FOR n=1 TO 13
260 PLOT 30+40*n,28:DRAW 0,4
270 NEXT n
280 FOR ph=1 TO 13 STEP 0.1
290 h=10^(-ph)
300 a=ka/(ka+h)
310 PLOT 30+40 *ph,30+300*a
320 NEXT ph

```


ph de la solution: 2  
 pka de la solution: 3  
 la concentration est: 0.110 mol/l.



ph de la solution: ? 4  
 pka de la solution: ? 7  
 la concentration est: 0.100 mol/l.



ph de la solution: ? 6.5  
 pka de la solution: ? 10  
 la concentration est: 0.001 mol/l.



## 4. Couches électroniques

Le cauchemar des étudiants débutant en chimie ! En effet, de manière générale, si la répartition en “couches” des électrons est simple (il y a sept couches possibles, distinguées par les lettres K à Q),

l'ordre de remplissage est plutôt complexe. Ce programme propose de visualiser la manière dont les couches se remplissent pour un atome quelconque. Cet ordre est en fait imposé par la ligne des DATA (ligne 210) qui contient des groupes de trois données étant respectivement le numéro de la couche (1 à 7), le type de la sous-couche (s, p, d, en f), et le nombre maximal d'électrons qu'elle peut contenir.

Le nombre d'électrons (soit le numéro atomique) de l'atome est appelé S. On cherche d'abord si la sous-couche en cours sera complètement remplie ou non (lignes 140 - 160). Dans un cas (ligne 140), on retourne en 110 pour chercher à remplir d'autres sous-couches. Dans l'autre cas, il s'agit de la dernière, qui sera alors complète. Si enfin la dernière sous-couche ne doit être qu'incomplètement remplie, on recherche le nombre d'électrons à placer.

L'intérêt pédagogique du programme est certain : l'étudiant "voit" les couches électroniques se remplir et admet ainsi qu'il y a un ordre à respecter. Nous présentons les résultats obtenus pour l'hélium (N=2), le carbone (N=6), le fluor (N=9), l'aluminium (N=13), le fer (N=26), l'argent (N=47), le plomb (N=82), l'uranium (N=92).

Un second programme, qualifié de variante, permet simplement de générer les couches électroniques d'un atome.

```

10 REM *****
20 REM *                                     *
30 REM * couches electroniques *
40 REM *                                     *
50 REM *****
60 REM
70 u=0
80 INPUT "nombre d'electron:";s
90 CLS: LOCATE 2,2:PRINT "n=";s
100 FOR z=1 TO 7:x=256:y=192:r=24*z:GOSUB 280:NEXT
    z
110 READ a$,b$,c$
120 u=u+VAL(c$)
130 IF u<s THEN GOSUB 180:GOTO 110
140 IF u=s THEN GOSUB 180
150 IF u>s THEN b=VAL(c$)-(u-s)-1:c$=STR$(b):GOSUB
    210
160 STOP
170 DATA 1,s,2,2,s,2,2,p,6,3,s,2,3,p,6,4,s,2,3,d,1
    0,4,p,6,5,s,2,4,d,10,5,p,6,6,s,2,4,f,14,5,d,10,6,p
    ,6,6,d,10,7,s,2


```

```


180 IF c$="2" THEN c$="1"
190 IF c$="6" THEN c$="5"
200 IF c$="10" THEN c$="9"
210 IF a$="1" THEN a=k:b=k+VAL(c$):c=2:GOSUB 290:k
=k+VAL(c$)+1:RETURN
220 IF a$="2" THEN a=1:b=1+VAL(c$):c=8:GOSUB 290:1
=1+VAL(c$)+1:RETURN
230 IF a$="3" THEN a=m:b=m+VAL(c$):c=18:GOSUB 290:
m=m+VAL(c$)+1:RETURN
240 IF a$="4" THEN a=n:b=n+VAL(c$):c=32:GOSUB 290:
n=n+VAL(c$)+1:RETURN
250 IF a$="5" THEN a=o:b=o+VAL(c$):c=32:GOSUB 290:
o=o+VAL(c$)+1:RETURN
260 IF a$="6" THEN a=p:b=p+VAL(c$):c=32:GOSUB 290:
p=p+VAL(c$)+1:RETURN
270 IF a$="7" THEN a=q:b=q+VAL(c$):c=32:GOSUB 290:
q=q+VAL(c$)+1:RETURN
280 FOR t=0 TO 2*PI STEP 2*PI/120:PLOT x+r*COS(T),
y+r*SIN(t):NEXT t:RETURN
290 FOR z=a TO b:x=256-VAL(a$)*24*COS(2*PI*z/c):y=
192-VAL(a$)*24*SIN(2*PI*z/c):r=8:GOSUB 280:NEXT z:
RETURN




```

**n= 2**




**n= 6**





 $n = 26$ 

 $n = 13$ 

 $n = 9$


**n= 47**



**n= 82**



**n= 92**



```

10 REM *****
20 REM *
30 REM * couches electroniques *
40 REM *
50 REM *****
60 REM
70 INPUT "nom de l'element :";z$: PRINT #8 , z$
80 INPUT "nombre d'electron:";s
110 READ a$,b$,c$
120 d$=a$+b$+c$
130 u=u+VAL(c$)
140 IF u<s THEN PRINT d$ : PRINT #8 , d$ : GOTO 1
150 IF u=s THEN PRINT d$ : PRINT #8 , d$ :RUN
160 IF u>s THEN b=VAL(c$)-(u-s)
190 d$=a$+b$+RIGHT$(STR$(B),LEN(STR$(b))-1)
200 PRINT d$ : PRINT #8 , d$
210 DATA 1,s,2,2,s,2,2,p,6,3,s,2,3,p,6,4,s,2,3,d,1
0,4,p,6,5,s,2,4,d,10,5,p,6,6,s,2,4,f,14,5,d,10,6,p
,6,6,d,10,7,s,2
220 RUN

```

```

helium
1s2
carbone
1s2
2s2
2p2
fluor
1s2
2s2
2p5
aluminium
1s2
2s2
2p6
3s2
3p1
fer
1s2
2s2
2p6
3s2
3p6
4s2
3d6

```



argent

1s2

2s2

2p6

3s2

3p6

4s2

3d10

4p6

5s2

4d9

plomb

1s2

2s2

2p6

3s2

3p6

4s2

3d10

4p6

5s2

4d10

5p6

6s2

4f14

5d10

6p2

uranium

1s2

2s2

2p6

3s2

3p6

4s2

3d10

4p6

5s2

4d10

5p6

6s2

4f14

5d10

6p6

6d6



# 5

## Programmes généraux

### 1. Equation de Van der Waals

Pour traduire le comportement des gaz, Van der Waals introduit en 1873, l'équation (relative à une môle de gaz).

$$\left(p + \frac{a}{V^2}\right) (V - b) = RT$$

où la pression est corrigée par le terme  $a/V^2$  dû aux forces d'attraction entre les molécules, et où le volume  $V$  est corrigé par le volume propre des molécules (covolume)  $b$ .

Les constantes  $a$  et  $b$  dépendent de la nature du gaz.

En calculant les dérivées première et seconde de  $P$  par rapport à  $V$ , le long d'une isotherme, et en les annulant, soit :

$$\left(\frac{\partial P}{\partial V}\right)_T = - \frac{RT}{(V-b)^2} + \frac{2a}{V^3} = 0$$

$$\left(\frac{\partial^2 P}{\partial V^2}\right)_T = + \frac{2RT}{(V-b)^3} - \frac{6a}{V^4} = 0$$

On obtient les relations suivantes :

$$V_c = 3b \quad ; \quad T_c = 8a/27Rb \quad \text{et} \quad P_c = a/27b^2$$

qui permettent, les constantes  $a$  et  $b$  étant obtenues expérimentalement, le calcul des valeurs critiques  $V_c$ ,  $T_c$  et  $P_c$  et vice versa.

#### Remarques :

— La température critique  $T_c$  est celle au-dessous de laquelle il faut amener un gaz pour qu'il soit liquéfiable par compression.

—  $\frac{P_c V_c}{RT_c} = \frac{3}{8}$  alors que pour le gaz parfait on aurait un rapport égal à l'unité.

— Si on cherche les constantes a et b à partir des valeurs critiques, on trouve :

$$a = \frac{27}{64} \frac{R^2 T_c^2}{P_c} \quad \text{et} \quad b = \frac{RT_c}{8P_c}$$

— Pour l'hydrogène  $a = 0,244$  et  $b = 0,266 \text{ l.atm.mole}^{-1}$  avec ces unités la constante du gaz parfait R vaut  $0,082 \text{ l.atm.k}^{-1}\text{mole}^{-1}$ .

## Parabole de Mariotte

Reprenons l'équation de Van de Waals et posons  $Pv = y$ , il vient :

$$y + \frac{aP}{y} - bP - \frac{abP^2}{y^2} = RT$$

C'est l'équation implicite  $f(y,P)$  de l'isotherme. Pour que la tangente soit horizontale, il faut  $\left(\frac{df}{dP}\right)_y = 0$

$$\text{c'est-à-dire : } -by^2 + ay - 2abP = 0.$$

C'est une parabole d'axe horizontal passant par l'origine et par  $y = a/b$  pour  $P = 0$ .

Le long de cette parabole la fonction  $Pv$  est constante et on peut écrire, pour de petites variations de la pression, que

$$Pv = \text{Cte}$$

On l'appelle donc la "parabole de Mariotte".

Bien que le gaz suive la loi de Mariotte, le produit  $Pv$  est différent de  $RT$ , ce n'est pas un gaz parfait, sauf au point  $P=0$  où  $Pv = a/b$ .

La "température de Mariotte" est celle de l'isotherme au-dessus de laquelle il n'y a plus de minimum ( $T_m = a/bR$ ).

Le programme 1 correspond à la représentation de Clapeyron ( $P = f(V)$ ) pour l'hydrogène.

Le programme 2 met en évidence la parabole de Mariotte.

## Liste des variables

A, B	constantes a et b de l'équation de Van der Waals pour l'hydrogène
R	constante du gaz parfait
P, V et T	pression, volume et température
Pc, Vc et Tc	valeurs critiques


## Déroulement du programme 1

lignes 10 à 70	introduction des données
lignes 80 à 130	tracé des axes
lignes 140 à 220	tracés des différents isothermes

## Déroulement du programme 2

lignes 10 à 70	introduction des données
lignes 70 à 130	tracé des axes
lignes 130 à 340	tracés des différents isothermes
lignes 350 à 500	tracé de la parabole de Mariotte

```
10 REM EQUATION DE VAN DER WAALS
20 REM *****
30 REM
40 REM INTRODUCTION DES CONSTANTES
50 REM *****
60 A=0.244:B=0.0266:R=0.082
70 VC=3*B:TC=8*A/(27*R*B):PC=A/(27*B*B)
80 CLS
90 MODE 2
100 PLOT 0,0:DRAWR 640,0
110 PLOT 0,0:DRAWR 0,400
120 LOCATE 2,3:PRINT "P"
130 LOCATE 78,24:PRINT "V"
140 FOR T=100 TO 20 STEP -10
150 FOR V=0.5 TO 0.005 STEP -0.001
160 P=(R*T)/(V-B)-A/(V^2)
170 IF P<=0 OR 20*P>=400 THEN GOTO 200
180 PLOT 500*V/0.5,20*P
190 NEXT V
200 IF T=TC THEN STOP
210 NEXT T
220 T=TC :GOTO 150
```



```


10 REM EQUATION DE VAN DER WAALS
20 REM *****
30 REM
40 REM ENTREE DES CONSTANTES
50 REM *****
60 A=0.244:B=0.0266:R=0.082
70 VC=3*B:TC=8*A/(27*R*B):PC=A/(27*B*B)
80 CLS
90 MODE 2
100 PLOT 0,0:DRAWR 640,0
110 PLOT 0,0:DRAWR 0,400
120 LOCATE 2,3:PRINT "PV"
130 LOCATE 78,24:PRINT "P"
140 FOR T=100 TO 20 STEP -10
150 AP=0:AV=0
160 FOR V=0.001 TO 7 STEP 0.02
170 P=(R*T)/(V-B)-A/(V^2)
180 IF P<0 OR AP<0 THEN GOTO 300
190 IF P*3>=600 OR AP*3>=600 THEN GOTO 300
200 IF P*V*40>=400 OR AP*AV*40>=400 THEN GOTO 300
210 PLOT P*3,P*V*40:DRAWR AP*3-P*3,AP*AV*40-P*V*40
300 AV=V:AP=P
310 NEXT V
320 IF T=TC THEN GOTO 400
330 NEXT T
340 IF T=TC THEN GOTO 160
350 REM PARABOLE DE MARIOTTE
360 REM *****
400 FOR P=0 TO 50
410 K=A^2-B*(B^2)*A*P
420 IF K<0 THEN STOP

```

```

430 X1=(A+SQR(K))/(2*B):X2=(A-SQR(K))/(2*B)
440 IF X1*40>=400 OR X2*40>=400 THEN GOTO 500
450 PLOT P*3,X1*40:PLOT P*3,X2*40
500 NEXT P

```



## 2. Méthode des moindres carrés Applications aux équations non linéaires

En physique, les graphiques expérimentaux présentent en général des irrégularités. Si les points représentés se répartissent suivant une configuration assez régulière, le physicien ajustera la ou les grandeurs étudiées ; c'est-à-dire qu'il substituera aux grandeurs observées des grandeurs calculées.

La première étape de l'ajustement porte sur le choix de la courbe d'ajustement, la seconde étape porte sur la détermination de l'équation de la courbe d'ajustement. Nous nous intéresserons aux équations de la forme :  $Y = f(X)$ .

### Méthode des moindres carrés


C'est une méthode analytique qui permet de calculer les différents paramètres  $W_i$  d'une équation de la forme  $Y = f(X)$ .

Représentons graphiquement les points expérimentaux de coordonnées :

$(X_1, Y_1)$  ,  $(X_2, Y_2)$  ...  $(X_J, Y_J)$

et les points de coordonnées qui se trouvent sur la courbe d'ajustement, soit :

$(X_1, Y'_1)$  ,  $(X_2, Y'_2)$  ...  $(X_J, Y'_J)$



La fonction  $Y = f(X)$  dépendant de plusieurs paramètres  $W_i$  des valeurs différentes de ces paramètres conduisent à des fonctions  $Y = f(X)$  différentes.

La méthode des moindres carrés consiste à retenir la fonction  $Y = f(X)$  qui rend la plus faible possible la somme des carrés des écarts :  $Y_i - Y'_i$ .

Il est hors de question pour nous de donner, dans cet ouvrage, la démonstration mathématique rigoureuse du calcul des différents paramètres  $W_i$  de la fonction  $Y = f(X)$ .

Nous utiliserons la méthode de Deming (1), reprise par Wentworth (2), le lecteur intéressé pourra éventuellement consulter les articles des auteurs.

Supposons que la fonction d'ajustement contienne la variable  $X$  et un certain nombre de paramètres :  $W_1, W_2 \dots W_J$  : soit  $Y' (X, W_1 \dots W_J)$ .

Il s'agit de trouver à l'aide d'un calcul itératif les valeurs optimisées des différents paramètres  $W_i$  (somme des carrés des écarts  $Y_i - Y'_i$  minimale).





Z(J,J) tableau pour les différents éléments de la matrice principale  
 COR coefficient de corrélation

## Déroulement du programme

lignes 1 à 20 entrée des données (N et J)  
 lignes 40 à 130 entrée de chaque couple de points expérimentaux ( $X_i$ ,  $Y_i$ )  
**Remarque** : à la ligne 70, on rentre les différentes valeurs des coefficients  $W_i$  de départ (choix de l'utilisateur)  
 lignes 140 et 150 on repère les valeurs maximales en valeur absolue des  $X_i$  et des  $Y_i$   
 lignes 200 à 250 initialisation des différents éléments de la matrice  
 ligne 340 on pose  $V(1) = Y(l) - Y'$   
 ligne 370 on pose  $DEF\ FNZ(W) = Y(l) - Y'$   
 lignes 390 à 410 les différentes dérivées  $F(W_i) = V(L+1)$  sont calculées numériquement  
 ligne 412 on calcule le coefficient de corrélation COR après chaque itération  
 ligne 500 saut au sous-programme d'inversion de matrice

Les calculs se déroulent sous vos yeux jusqu'à la convergence du coefficient de corrélation (Noter les différents  $W_i$  après avoir arrêté le programme en appuyant 2 fois sur ESC à la convergence).

lignes 1000 à 1200 l'ordinateur efface l'écran. Il demande à nouveau les différents  $W_i$  calculés à la convergence.

ligne 1025 on trace les axes, les points expérimentaux à l'échelle et la courbe d'ajustement.

**Remarque** : en ligne 1080, n'oubliez pas de rentrer la fonction d'ajustement  $Y'$ .

Nous donnons 5 exemples d'utilisation de ce programme.

**Exemple 1** : la fonction d'ajustement est une droite.

**Exemple 2** : la fonction d'ajustement est une fonction du troisième degré.

Pour les 3 derniers exemples, les fonctions d'ajustement sont des fonctions d'exponentielle (exemples les plus difficiles à traiter).

En effet, Leydet (3) a montré que les fonctions de mélange en thermodynamique pouvaient être représentées par des fonctions d'exponentielle. Nous donnons donc 3 exemples de mélange.

**Remarque :** ces fonctions s'annulent pour  $x = 0$  et  $x = 1$ .

## Bibliographie

1. DEMING W.E., *Statistical Adjustment of Data*, John Wiley and Sons, Inc., New York, 1943.
2. WENTWORTH W.E., *Journal of Chemical Education*, volume 4'2, Number 2, February 1965.
3. LEYDET P., *Thermochimica Acta*, 55 (1982) 193-200.

```

1 CLS
5 REM MOINDRES CARRES
6 REM *****
7 REM
10 INPUT "NB DE POINTS EXP.N=";N
20 INPUT "NB DE COEFFTS.J=";J
35 DIM X(N):DIM Y(N):DIM Z(J,J):DIM D(J)
40 M=0:P=0:DIM W(J+1):DIM R(J)
42 K=J+1:DIM V(K)
44 DIM U(J,J):DIM E(J):DIM F(J)
50 DW=0.001:PR=0.00000001:MI=1E-30
55 MODE 2
60 FOR I=1 TO J
70 INPUT "W=";W(I):PRINT "W";I;"=";W(I)
80 NEXT I
90 CLS
100 FOR I=1 TO N
110 PRINT "I=";I
120 INPUT "X=";X(I):INPUT "Y=";Y(I)
130 PRINT "X";I;"=";X(I);": ";Y(I);": ";Y(I):PRINT
140 IF ABS(Y(I))>P THEN P=ABS(Y(I))
150 IF ABS(X(I))>M THEN M=ABS(X(I))
160 NEXT I
190 S=0
200 FOR C=1 TO J
210 FOR L=1 TO J
220 Z(L,C)=0
230 NEXT L
240 NEXT C

```

```

250 FOR L=1 TO J:D(L)=0:NEXT L
300 REM CALCUL ITERATIF
310 REM *****
316 FOR I=1 TO N
320 X=X(I)
340 V(1)=Y(I)-(W(1)*(1-X)-W(1)*(1-X)*EXP(-W(2)*X/(
1-X))+W(3)*X-W(3)*X*EXP(-W(4)*(1-X)/X))
341 D=V(1)*V(1):S=S+D
350 FOR L=1 TO J
360 W=W(L):A=W(L)
370 DEF FNZ(W)=Y(I)-(W(1)*(1-X)-W(1)*(1-X)*EXP(-W(
2)*X/(1-X))+W(3)*X-W(3)*X*EXP(-W(4)*(1-X)/X))
390 W(L)=A+DW:Z1=FN Z(W)
400 W(L)=A-DW:Z2=FN Z(W)
410 V(L+1)=(Z1-Z2)/(2*DW)
411 B=V(L+1)*V(1)
412 D(L)=D(L)+B
415 W(L)=A
420 NEXT L
440 FOR C=1 TO J
445 FOR L=1 TO J
450 H=V(L+1)*V(C+1)
455 Z(L,C)=Z(L,C)+H
460 NEXT L
470 NEXT C
490 NEXT I
491 CR=S/N
492 PRINT "COR=";SQR(CR)
493 PRINT
500 GOSUB 2021
600 FOR L=1 TO J:R(L)=0
610 FOR C=1 TO J
620 R(L)=R(L)+Z(L,C)*D(C)
630 NEXT C
640 NEXT L
641 FOR F=1 TO J
642 W(F)=W(F)-R(F):PRINT W(F)
643 NEXT F
660 GOTO 190
1000 CLS
1010 PLOT 0,200:DRAWR 620,0
1020 PLOT 0,0:DRAWR 0,400
1025 FOR F=1 TO J:INPUT W(F):NEXT F
1030 FOR I=1 TO N
1040 PLOT X(I)/M*600,Y(I)/P*180+200:DRAWR 2,0
1041 PLOT X(I)/M*600,Y(I)/P*180+200:DRAWR 0,2
1042 PLOT X(I)/M*600,Y(I)/P*180+200:DRAWR -2,0
1043 PLOT X(I)/M*600,Y(I)/P*180+200:DRAWR 0,-2
1050 NEXT I
1060 FOR K=1 TO 600

```

```

1070 X=(M/600)*K
1080 Y=W(1)*(1-X)-W(1)*(1-X)*EXP(-W(2)*X/(1-X))+W(
3)*X-W(3)*X*EXP(-W(4)*(1-X)/X)
1085 IF ABS(Y/P*180)>200 THEN GOTO 1100
1090 PLOT K,Y/P*180+200
1100 NEXT K
1200 STOP
2000 REM SP.INVERSION DE MATRICE
2010 REM *****
2021 FOR C=1 TO J
2022 FOR L=1 TO J
2023 U(L,C)=Z(L,C)
2025 NEXT L
2026 NEXT C
2030 DET=1
2031 FOR G=1 TO J
2035 MA=MI
2036 FOR L=G TO J
2037 FOR C=G TO J
2040 A=Z(L,C)
2045 IF ABS(A)<=MA THEN GOTO 2050
2046 IM=L:JM=C:MA=ABS(A)
2050 NEXT C
2060 NEXT L
2070 IF MA<>MI THEN GOTO 2100
2080 STOP
2100 PT=Z(IM,JM):DET=DET*PT
2110 F(G)=IM:E(G)=JM
2120 IF IM=G THEN GOTO 2160
2130 FOR C=1 TO J
2140 A=Z(IM,C):Z(IM,C)=Z(G,C):Z(G,C)=A
2150 NEXT C
2160 IF JM=G THEN GOTO 2200
2170 FOR L=1 TO J
2180 A=Z(L,G):Z(L,G)=Z(L,JM):Z(L,JM)=A
2190 NEXT L
2200 FOR L=1 TO J
2210 IF L=G THEN GOTO 2300
2220 R=Z(L,G)/PT
2230 FOR C=1 TO J
2240 IF C=G THEN GOTO 2290
2250 DD=Z(L,C)-R*Z(G,C)
2260 IF ABS(DD)<ABS(Z(L,C))*PR THEN DD=0
2270 Z(L,C)=DD
2290 NEXT C
2300 NEXT L
2400 FOR Z=1 TO J
2410 Z(Z,G)=Z(Z,G)/PT
2420 Z(G,Z)=-Z(G,Z)/PT
2430 NEXT Z

```

```

2500 Z(G,G)=1/PT
2510 NEXT G
2520 FOR G=1 TO J
2530 Z=J-G+1
2540 PL=F(Z)
2550 IF PL=Z THEN GOTO 2600
2560 FOR L=1 TO J
2570 A=Z(L,Z):Z(L,Z)=Z(L,PL):Z(L,PL)=A
2580 NEXT L
2600 PC=E(Z)
2610 IF PC=Z THEN GOTO 3000
2620 FOR C=1 TO J
2630 A=Z(Z,C):Z(Z,C)=Z(PC,C):Z(PC,C)=A
2640 NEXT C
3000 NEXT G
3010 RETURN

```

### EXEMPLE 1

```

1.8
3.8
COR= 0.048528138

```

```

1.8
3.8
COR= 0.048528138

```

```

1.8
3.8
COR= 0.048528138

```

```

1.8
3.8
COR= 0.048528138

```

```

1.8
3.8
COR= 0.048528138

```

```

1.8
3.8
Break in 643
Ready
CALL AAB00

```

On part de  $W1 = 1$  ;  $W2 = 1$

X	Y
2	7
3	10
4	10
5	14
6	14




Fig. 1

EXEMPLE 2

```

-15.734123
18.9999963
COR= 0.636289815

-0.277775826
3.98807447
-15.7340627
18.9999473
COR= 0.636289811


-0.27777538
3.98805238
-15.7341184
18.9999325
COR= 0.636289886

-0.27777541
3.98809261
-15.7341185
18.9999326
COR= 0.636289811

Break in 2290
Ready
CALL 40000
  
```

On part de W1 = 1 ; W2 = 1 ; W3 = 1 et W4 = 1

X	Y
1	7
2	1
3	1
4	1
5	6
6	8



### EXEMPLE 3

Tableau :

**MELANGE : 1 : Tri-n-dodecylamine, C36NH75**  
**2 : Chlorobenzene, C6C1H5**

x(C36NH75)	Hm(J.mol <sup>-1</sup> )
0.1046	565.2
0.1242	670.8
0.1491	736.3
0.1822	800.9
0.2607	883.7
0.2932	895.0
0.3444	901.6
0.3965	884.6
0.4834	808.7
0.5789	650.2
0.7064	467.9
0.8279	281.4



COR= 18.2847852

1540.15393  
4.54356881  
COR= 18.2771273

1540.22793  
4.54381345  
COR= 18.2771143

1540.22758  
4.54381641  
COR= 18.2771144

1540.22758  
4.54380687  
COR= 18.2771144

1540.22777  
4.54380479  
COR= 18.2771143

Break in 2630  
Ready  
CALL A0000




Fig. 3

$$Y' = W1.(1 - x) - W1.(1 - x).Exp (- W2.x/(1 - x) )$$

On part de W1 = 1000

W2 = 4

#### EXEMPLE 4

Tableau :

**MELANGE : 1 : Benzene, C<sub>6</sub>H<sub>6</sub>**  
**2 : Cyclohexane, C<sub>6</sub>H<sub>12</sub>**

x(C <sub>6</sub> H <sub>6</sub> )	H <sub>m</sub> (J.mol <sup>-1</sup> )
0.02055	62.71
0.04847	143.94
0.10063	283.30
0.14769	394.02
0.18974	471.94
0.22787	552.61
0.26264	608.74
0.29442	654.19
0.32360	690.16
0.35041	718.98
0.37522	741.81
0.41956	773.21
0.45795	790.71
0.49163	798.72
0.51860	800.12
0.57886	786.04
0.60473	773.39
0.63650	751.99
0.67183	720.49
0.71129	675.57
0.75592	611.98
0.78027	571.43
0.80625	523.32
0.83401	466.66
0.86376	399.34
0.89570	319.37
0.93046	223.37
0.96744	109.53
0.98720	44.10

```

765.297645
3.35234634
COR= 0.432040786

```

```

934.883765
3.50982326
767.317127
3.35265383
COR= 0.225633226

```

```

934.956387
3.50984111
767.220793
3.35257044
COR= 0.225631447

```

```

934.958043
3.50984133
767.254542
3.35285759
COR= 0.225630854

```

```

Break in 2630
Ready
CALL 40000

```

```

? 934.958043
? 3.50984133
? 767.254542
? 3.35285759

```




Fig. 4

$$Y' = W1.(1 - x) - W1(1 - x).Exp(- W2.x/(1 - x) ) \\ + W3.x - W3.x. Exp ( - W4.(1 - x)/x)$$

On part de W1 = 1000

W2 = 4

W3 = 1000

W4 = 4

## EXEMPLE 5

Tableau :

MELANGE : 1 : Propanol  
2 : Eau

x	Hm
0.0059	- 12.9
0.0177	- 36.9
0.0291	- 56.5
0.0403	- 70.5
0.051	- 78.4
0.071	- 81.3
0.105	- 73.4
0.149	- 59.5
0.189	- 48.8
0.203	- 46.6
0.229	- 39.8
0.280	- 22.6
0.337	- 4.1
0.384	+ 9.6
0.478	+ 32.6
0.592	+ 49.7
0.714	+ 51.9
0.833	+ 35.7
0.909	+ 18.9

```

222.983277
1.81918182
CON= 2.69842536

```

```

-114.888325
28.6588338
223.24018
1.81637476
CON= 2.69838542

```

```

-114.888184
28.6542562
223.213591
1.81668126
CON= 2.69838492

```

```

-114.888929
28.6538295
223.218808
1.8166649
CON= 2.69838493

```

```

Break in 2845
Ready
CALL 40000

```

```

? -114.888929
? 28.6538295
? 223.218808
? 1.8166649

```




Fig. 5

$$Y' = W1.(1 - x) - W1(1 - x).Exp(-W2.x/(1 - x)) + W3.x - W3.x.Exp(-W4.(1 - x)/x)$$

On part de W1 = -100

W2 = 30

W3 = 300

W4 = 3

*Imprimé en France.* — JOUVE, 18, rue Saint-Denis, 75001 PARIS  
N° 16127. Dépôt légal : Septembre 1986  
N° d'Éditeur : 4534

Photocomposition : COMPO SPRINT - 28240 La Loupe









Vous n'avez pas encore eu le coup de foudre pour la physique ?

Votre ordinateur et ses programmes, en vous libérant des longs calculs, des formules « bombardées », de l'impatience du professeur, vous offrent une nouvelle chance.

De plus, ces programmes sont soigneusement commentés afin que cette nouvelle approche de la physique vous paraisse transparente et que vous vous sentiez « maître ». Ils vous expliqueront également par l'exemple les possibilités graphiques de votre ordinateur.

La mécanique, l'électricité, la thermodynamique, l'optique et la chimie sont abordées dans cet ouvrage au niveau des classes de lycée.

Les auteurs mettent ici à votre service leur expérience de professeurs de physique. Ils ont déjà écrit de nombreux ouvrages d'électronique et d'informatique.

Les programmes qui suivent correspondent à l'enseignement des Sciences Physiques en classes de lycée. Ils ont été écrits pour l'ordinateur AMSTRAD. Les copies d'écran pour l'illustration de ce livre ont été obtenues au moyen de l'imprimante AMSTRAD DMP 2000.

Nous avons quelquefois sacrifié quelque peu la présentation sur l'écran pour que le programme soit moins long à recopier, nous n'avons pas recherché systématiquement les astuces de programmation, afin que la construction du programme vous apparaisse plus clairement. Nous espérons que vous apprécierez ces compromis.



EVEROLLES  
PROGRESS  
DRAWN  
DEPTHS  
HYS  
JUL  
STRAITS  
TRADE  
P. BEAL  
M. LAMARC  
Y. MUGLIA

P. BEAUJOUR  
M. LAMARC  
Y. MUGGIAT




Document **numérisé**  
avec amour par :

**AMSTRAD**

CPC 

MÉMOIRE ÉCRITE



<https://acpc.me/>